


+ EXTRA
COMPUTERCURSUS
MORSE

50 PROGRAMMA'S VOOR

MSX

COMPUTERS

M. B. IMMERZEEL - 50 PROGRAMMA'S VOOR MSX COMPUTERS



```
1610 CLS: PRINT"Galgje is een spel dat";  
1620 PRINT" u een woord laat raden. ";  
1630 PRINT"Een aantal stippen geven ";  
1640 PRINT"de leng-te van het woord. ";  
1650 PRINT"U kunt een letter in-";  
1660 PRINT"voeren en als deze in het ";  
1670 PRINT"woord voorkomtwordt hij ";  
1680 PRINT"op de juiste plaats ge";  
1690 PRINT"toond. Door steeds ";  
1700 PRINT"opnieuw een letter in te";  
1710 PRINT"voeren moet u het woord ";  
1720 PRINT"raden. U kunt";  
1730 PRINT"een max. aantal letters ";  
1740 PRINT"invvoeren. Heeft u het dan ";  
1750 PRINT"nog niet gevonden dan ";  
1760 PRINT"hangt"; PRINT"de man.";  
1770 PRINT"; PRINT"Gebbruik om naar het ";  
1780 PRINT"programma terug te";  
1790 PRINT"keren een willekeurige ";  
1800 PRINT"toets."
```

omreken tabellen
rekenprogramma's
renteberekeningen
kalenders
spelletjes

**50 PROGRAMMA'S
VOOR
MSX COMPUTERS**

© 1985 De Muiderkring b.v. Bussum - Nederland

Niets uit deze uitgave mag worden verveelvoudigd en/of openbaar gemaakt door middel van druk, fotocopie, microfilm of op welke andere wijze ook, zonder voorafgaande schriftelijke toestemming van de uitgever.

ISBN 90 6082 273 0

M. B. IMMERZEEL

50 PROGRAMMA'S VOOR

**MSX
COMPUTERS**

+ EXTRA COMPUTERCURSUS MORSE



DE MUIDERKRING B.V. - BUSSUM

UITGEVERIJ VAN TECHNISCHE BOEKEN EN TIJDSCHRIFTEN

Inhoud

Voorwoord	7
1. Het leren rekenen met de computer	8
2. Het berekenen van een term in een reeks	12
3. Conversie	13
3.1. Conversie van inches naar millimeter v.v.	14
3.2. Conversie van foot naar meter v.v.	14
3.3. Conversie van mile naar kilometer v.v.	15
3.4. Conversie van gram naar ounce v.v.	16
3.5. Conversie van pound naar kilogram v.v.	16
3.6. Conversie van gallon naar liter v.v.	17
3.7. Conversie van graden naar radialen v.v.	18
3.8. Conversie van Arabische naar Romeinse cijfers v.v. ...	18
4. Rekenprogramma's	20
4.1. Het berekenen van priemgetallen	20
4.2. Het berekenen van het kgv en ggd	20
4.3. Bewerkingen met breuken	21
4.4. Berekening van priemdelers	23
4.5. Het omzetten van tiendelige naar gewone breuken v.v. .	23
4.6. Functies (het berekenen van grafieken)	24
4.7. Het berekenen van pythagorische driehoeken	27
4.8. Het oplossen van een vierkantsvergelijking	28
4.9. Omzetting van het polaire naar het rechthoekcoördinatenstelsel	29
4.10. Het omzetten van radialen naar graden	30
4.11. Berekening kegel en bol	31
4.12. Berekening van een kogelbaan	31
4.13. Vallende voorwerpen	32
4.14. Statistiek	33
5. Renteberekeningen	34
5.1. Berekening van het percentage	34
5.2. Sparen	34
5.3. De persoonlijke lening	35
6. Complexe getallen	37
6.1. Complex rekenen	37
6.2. Notatie complexe getallen	38
7. Datum	39
7.1. Eeuwigdurende kalender	39
7.2. De kalender	40
7.3. Feestdagen	41
7.4. Aantal dagen	41

8.	Spelletjes	43
8.1.	Het meten van de reactiesnelheid	43
8.2.	Stuivertje verwisselen	44
8.3.	Verplaatsen van een getallenrij	45
8.4.	Neem	47
8.5.	Omkeren	49
8.6.	Schuiven	51
8.7.	Getal raden	53
8.8.	Woord raden	54
8.9.	Colormind	55
8.10.	Galgje.....	57
8.11.	Ruimtecapsule.....	59
8.12.	Paren zoeken	60
9.	Gokken	63
9.1.	Eenentwintig	63
9.2.	Het lottoformulier	65
10.	Morse cursus.....	66
10.1.	Morse leren	66
10.2.	Morse oefenen	68

Voorwoord

Dit boek biedt een aantal programma's. U kunt kiezen uit verschillende - hopelijk voor u nuttige - onderwerpen, verdeeld over 10 hoofdstukken.

Zo vindt u onder andere: rekenen - waarvoor de computer eigenlijk is gemaakt - conversie, spelle-tjes, morse. Ik heb getracht de programma's handelbaar te maken, dat wil zeggen dat de gegevens die voor het verloop van het programma nodig zijn op een zo eenvoudig mogelijke wijze kunnen worden ingevoerd, terwijl het programma duidelijk laat merken wanneer er gegevens worden verlangd en zo ja, welke. Bij elk programma is dan ook een beschrijving gegeven van de manier waarop het moet worden gebruikt.

Aan de meeste programma's zijn een aantal REM-statements toegevoegd. Die zijn bedoeld voor hen die het verloop van het programma willen volgen

en eventueel verbeteringen of veranderingen willen aanbrengen. Bij het intoetsen van het programma geven deze regels wel extra werk, maar als u ze niet nodig denkt te hebben dan laat u ze gewoon weg. Dat is ook het geval met de aanwijzingen die bij sommige programma's staan. Deze zijn te beschouwen als korte gebruiksaanwijzingen die u ten dienste staan, als u het programma na verloop van tijd weer eens van stal haalt. U hoeft dan de beschrijving niet meer door te lezen.

Ten behoeve van de leesbaarheid zijn aan de programmaregels een aantal spaties toegevoegd. Ook al ten gunste van de leesbaarheid is de lengte van de regels beperkt tot veertig karakters, inclusief de spaties. Dit kan zo hier en daar wel eens tot een wat vreemde constructie hebben geleid.

Ede, voorjaar 1985

M. B. Immerzeel

Attentie

Bij een aantal programma's kunt u een werkwijze kiezen door een bepaalde letter in te toetsen. U dient daarbij het lettertype (hoofdletter of kleine letter) te gebruiken dat ook in de tekst van dit boekje of op het scherm hiervoor is gebruikt.

1. Het leren rekenen met de computer

Dat de computer, mits op de juiste de wijze toegepast, heel goed bij het onderwijs op school kan worden gebruikt, zal zo langzamerhand wel bekend zijn. Eén van de toepassingen is het oefenen in rekenen, het cijferen, waarbij bewerkingen met meer of minder moeilijke getallen moeten worden uitgevoerd.

Er volgen nu drie programma's waarmee u uw kinderen kunt laten oefenen in het cijferen zodat ze daarin een grotere vaardigheid kunnen krijgen. Er zijn vijf moeilijkheidsgraden voor de opgaven waarvan de eerste twee zeker geschikt zijn voor het oefenen in hoofdrekenen. Deze moeilijkheidsgraden kunnen worden gekozen met de toetsen 1 tot en met 5 en het laagste cijfer geeft ook de laagste moeilijkheidsgraad.

Er wordt een serie van twintig opgaven op het scherm geprojecteerd, steeds één opgave tegelijk. Deze opgaven worden steeds opnieuw door de computer verzonden zodat, indien in een serie met een bepaalde moeilijkheidsgraad wordt herhaald, voor elke serie steeds andere opgaven kunnen worden verwacht. Overigens kan het bij de laagste moeilijkheidsgraad ook wel eens gebeuren dat binnen een serie meerdere gelijke opgaven voorkomen. Wordt op een opgave een foutief antwoord gegeven dan wordt dit gemeld en wordt nog een keer de gelegenheid gegeven het goede antwoord in te voeren. Na elke serie wordt het aantal goede antwoorden vermeld: zijn dat er minder dan twaalf in rode, bij twaalf of meer in groene tekens. Zijn er twaalf of meer goede antwoorden gegeven dan volgt automatisch de volgende serie met de naast hogere moeilijkheidsgraad. Bij minder dan twaalf goede antwoorden volgt een serie met dezelfde moeilijkheidsgraad.

Het eerste programma betreft het optellen en het aftrekken. Na het invoeren en het starten van het programma leest u op het scherm:

```
OPTELLEN OF AFTREKKEN.  
KIES VOOR MOEILIKHEIDS-  
GRAAD 1-5.  
?
```

Nadat de betreffende toets is ingedrukt volgt:

KIES NU VOOR + OF -?

Direct na uw keuze verschijnt de eerste opgave. Hebt u de tweede moeilijkheidsgraad en optellen gekozen dan kan dat zijn:

1 56 + 8 = ?

na het invoeren van het getal 64 leest u eerst:

GOED, VOLGENDE OPGAVE:

en na enige tijd verschijnt opgave twee op de plaats van die van de eerste. Het geheel herhaalt zich tot en met de twintigste opgave.

```
10 'leren optellen en aftrekken  
20 'voor een echt random getal  
30 CLS: PRINT"druk op toets SELECT"  
40 A$=INKEY$: A=RND(1)  
50 IF A$(<>CHR$(24)) THEN 40  
60 KEY OFF  
70 'voor geel veld, groen kader  
80 'en witte tekst  
90 SCREEN 1: COLOR 15,10,3: WIDTH(32)  
100 'voor zwarte cijfers  
110 VPOKE 8197,26  
120 VPOKE 8198,26: VPOKE 8199,26  
130 PRINT SPC(5) "Optellen en aftrekken."  
140 PRINT: PRINT"kies voor moeilijk";  
150 PRINT"heidsgraad 1-5"  
160 'invoer v.d. moeilijkheidsgraad  
170 INPUT A  
180 'alleen getallen van 1 t.e.m. 5  
190 IF A<1 OR A>5 THEN 170  
200 'kies voor optellen of aftrekken.  
210 INPUT"Kies nu voor + of voor -"; A$  
220 'alleen + of -  
230 IF A$(<>"-") AND A$(<>"+") THEN 210  
240 'voor 20 opgaven  
250 FOR T=1 TO 20  
260 'B is getal 1; C is getal 2  
270 B=RND(1): C= RND(1): S=0: H=1  
280 'geen getallen kleiner dan .1  
290 IF B<.1 THEN B=B*10: GOTO 290  
300 IF C<.1 THEN C=C*10: GOTO 300  
310 'selecteer naar moeilijkheidsgraad  
320 ON A GOTO 340,360,390,420,450
```

```

330 'getal van 1-10
340 D=INT(10*B+1):E=INT(10*C+1):GOTO 480
350 '1e getal van 10-100; 2e van 1-10
360 D=INT(100*B+10)
370 E=INT(10*C+1): GOTO 480
380 '1e getal 100-1000; 2e van 10-100
390 D=INT(1000*B+100)
400 E=INT(100*C+10): GOTO 480
410 'getallen met 1 cijfer na de komma
420 H=10: D=INT(1000*B+100)/H
430 E=INT(100*C+10)/H: GOTO 480
440 'getallen met 2 cijfers na de komma
450 H=100: D=INT(10000*B+1000)/H
460 E=INT(1000*C+100)/H
470 'grootste getal eerst
480 IF D<E THEN SWAP D,E
490 'print de opgave
500 CLS: PRINT " ";D;A$;E;"=";
510 'voer het resultaat in
520 INPUT G
530 'selecteer naar opt. of aftr.
540 IF A$="+" THEN 600
550 'bepaal of het antwoord goed is
560 IF ABS(D-E-G)<.1/H THEN 720
570 'antwoord niet goed
580 PRINT: PRINT"fout, ";: GOTO 650
590 'bepaal of het antwoord goed is
600 IF ABS(D+E-G)<.1/H THEN 720
610 'antwoord niet goed
620 PRINT: PRINT"fout, ";
630 'bij eerste poging, nog een keer
640 'bij tweede poging, geef juiste antw.
650 BEEP: IF S=1 THEN S=0: GOTO 750
660 S=S+1
670 PRINT"probeer nog een keer"
680 'vertraag
690 FOR I=1 TO 2000: NEXT: GOTO 500
700 'antwoord goed
710 'R is aantal goede antwoorden
720 R=R+1: PRINT:PRINT"goed ";
730 'geef muziek
740 PLAY"sim4661t90L8o5gg16g16g16g16abo6c"
750 IF A$="-" THEN F=D-E ELSE F=D+E
760 PRINT F
770 'bepaal of serie is voltooid
780 IF T=20 THEN 810
790 PRINT"Volgende opgave:"
800 'vertraag, geef volgende opgave
810 FOR I=0 TO 2000: NEXT I: NEXT T
820 'de serie is voltooid.
830 'bij goed result. groene tekst
840 'bij slecht result. rode tekst
850 IF R<12 THEN COLOR 6 ELSE COLOR 12
860 PRINT: PRINT: PRINT"je hebt ";
870 PRINT R "sommen goed"
880 PRINT: PRINT: PRINT
890 'bij slecht result geen hogere graad
900 IF R<12 THEN 930
910 'hogere moeilijkheidsgraad
920 A=A+1: IF A=6 THEN 1000
930 PRINT"Volgende moeilijkheidsgraad";
940 PRINT A: R=0
950 'vertraag, daarna volgende serie
960 FOR I=0 TO 5000: NEXT I
970 COLOR 15: VPOKE 8197,26
980 VPOKE 8198,26
990 VPOKE 8199,26: CLS: GOTO 250
1000 PRINT"Nog moeilijker kan ";
1010 PRINT"ik het niet!": END

```

Het tweede programma betreft het vermenigvuldigen en het delen.

De bewerkingstekens die zijn toegepast zijn gelijk aan de symbolen die door de computer worden gebruikt, dus respectievelijk * en /.

De werkwijze is verder gelijk aan die van het eerste programma. De delingen komen in alle moeilijkheidsgraden steeds uit, dus zonder rest.

```

10 'leren vermenigvuldigen en delen
20 'voor een echt random getal
30 CLS: PRINT"druk op toets SELECT"
40 A$=INKEY$: A=RND(1)
50 IF A$<>CHR$(24) THEN 40
60 KEY OFF
70 'voor geel veld, groen kader
80 'en witte tekst
90 SCREEN 1: COLOR 15,10,3: WIDTH(32)
100 'voor zwarte cijfers
110 VPOKE 8197,26
120 VPOKE 8198,26: VPOKE 8199,26
130 PRINT SPC(3) "Vermenigvuldigen ";
140 PRINT"of delen"
150 PRINT: PRINT"kies voor moeilijk";
160 PRINT"heidsgraad 1-5"
170 'invoer v.d. moeilijkheidsgraad
180 INPUT A
190 'alleen getallen van 1 t.e.m. 5
200 IF A<1 OR A>5 THEN 180
210 'kies voor vermenigvuld. of delen
220 INPUT"Kies nu voor * of voor /"; A$
230 'alleen * of /
240 IF A$<>"*" AND A$<>"/" THEN 220
250 'voor 20 opgaven
260 FOR T=1 TO 20
270 'B is getal 1; C is getal 2
280 B=RND(1): C= RND(1): S=0: H=1
290 'geen getallen kleiner dan .1
300 IF B<.1 THEN B=B*10: GOTO 300
310 IF C<.1 THEN C=C*10: GOTO 310
320 'selecteer naar moeilijkheidsgraad
330 ON A GOTO 350,370,400,430,460
340 'getal van 1-10
350 D=INT(10*B+1):E=INT(10*C+1):GOTO 490
360 '1e getal van 10-100; 2e van 1-10
370 D=INT(100*B+10)
380 E=INT(10*C+1): GOTO 490
390 '1e getal 100-1000; 2e van 10-100
400 D=INT(1000*B+100)
410 E=INT(100*C+10): GOTO 490
420 'getallen met 1 cijfer na de komma
430 H=10: D=INT(1000*B+100)/H
440 E=INT(100*C+10)/H: GOTO 490
450 'getallen met 2 cijfers na de komma
460 H=100: D=INT(10000*B+1000)/H
470 E=INT(1000*C+100)/H
480 'grootste getal eerst
490 IF D<E THEN SWAP D,E
500 IF A$="/" THEN D=D*E
510 'print de opgave
520 CLS: PRINT " ";D;A$;E;"=";
530 'voer het resultaat in
540 INPUT G
550 'selecteer naar verm. of delen
560 IF A$="*" THEN 620

```

```

570 'bepaal of het antwoord goed is
580 IF ABS(D/E-G)<.1/H THEN 740
590 'antwoord niet goed
600 PRINT: PRINT"fout, "; GOTO 670
610 'bepaal of het antwoord goed is
620 IF ABS(D\E-G)<.1/H THEN 740
630 'antwoord niet goed
640 PRINT: PRINT"fout, ";
650 'bij eerste poging, nog een keer
660 'bij tweede poging, geef juiste antw.
670 BEEP: IF S=1 THEN S=0: GOTO 770
680 S=S+1
690 PRINT"probeer nog een keer"
700 'vertraag
710 FOR I=1 TO 2000: NEXT: GOTO 520
720 'antwoord goed
730 'R is aantal goede antwoorden
740 R=R+1: PRINT:PRINT"goed ";
750 'geef muziek
760 PLAY"slm4661t90L8o5gg16g16g16g16abo6c"
770 IF A$="/" THEN F=D/E ELSE F=D\E
780 PRINT F
790 'bepaal of serie is voltooid
800 IF T=20 THEN 830
810 PRINT"Volgende opgave:"
820 'vertraag, geef volgende opgave
830 FOR I=0 TO 2000: NEXT I: NEXT T
840 'de serie is voltooid.
850 'bij goed result. groene tekst
860 'bij slecht result. rode tekst
870 IF R<12 THEN COLOR 6 ELSE COLOR 12
880 PRINT: PRINT: PRINT"je hebt ";
890 PRINT R "sommen goed"
900 PRINT: PRINT: PRINT
910 'bij slecht result geen hogere graad
920 IF R<12 THEN 950
930 'hogere moeilijkheidsgraad
940 A=A+1: IF A=6 THEN 1020
950 PRINT"Volgende moeilijkheidsgraad";
960 PRINT A: R=0
970 'vertraag, daarna volgende serie
980 FOR I=0 TO 5000: NEXT I
990 COLOR 15: VPOKE 8197,26
1000 VPOKE 8198,26
1010 VPOKE 8199,26: CLS: GOTO 260
1020 PRINT"Nog moeilijker kan ";
1030 PRINT"ik het niet!": END

```

Het derde programma betreft het machtsverheffen en het (vierkants)worteltrekken. Ook hierbij zijn de bewerkingstekens toegepast die ook door de computer worden gebruikt: 2 en SQR(.). De werkwijze is weer gelijk aan die van het eerste programma en de worteltrekkingen komen in alle moeilijkheidsgraden steeds uit, zonder rest. Bij de laagste moeilijkheidsgraad zal echter meerdere keren dezelfde opgave voorkomen.

```

10 'leren machtsverh. en worteltr.
20 CLS: PRINT"druk op toets SELECT"
30 'voor een echt random getal
40 A$=INKEY$: A=RND(1)
50 IF A$<>CHR$(24) THEN 30

```

```

60 KEY OFF
70 'voor geel veld, groen kader
80 'en witte tekst
90 SCREEN 1: COLOR 15,10,3: WIDTH(32)
100 'voor zwarte cijfers
110 VPOKE 8197,26
120 VPOKE 8198,26: VPOKE 8199,26
130 PRINT "machtsverheffen ";
140 PRINT"of worteltrekken"
150 PRINT: PRINT"kies voor moeilijk";
160 PRINT"heidsgraad 1-5"
170 'invoer v.d. moeilijkheidsgraad
180 INPUT A
190 'alleen getallen van 1 t.e.m. 5
200 IF A<1 OR A>5 THEN 180
210 'kies machtsverh. of worteltr.
220 INPUT"Kies voor ^2 of SQR"; A$
230 'alleen ^2 of SQR
240 IF A$<>"^2" AND A$<>"SQR" THEN 220
250 'voor 20 opgaven
260 FOR T=1 TO 20
270 'B is getal 1
280 B=RND(1): S=0: H=1
290 'geen getallen kleiner dan .1
300 IF B<.1 THEN B=B*10: GOTO 300
310 'selecteer naar moeilijkheidsgraad
320 ON A GOTO 340,360,380,400,420
330 'getal van 1-10
340 D=INT(10*B): GOTO 430
350 'getal van 10-100
360 D=INT(100*B+1): GOTO 430
370 'getal van 100-1000
380 D=INT(1000*B+10): GOTO 430
390 'getal met 1 cijfer na de komma
400 H=10: D=INT(1000*B+10)/H:GOTO 430
410 'getal met 2 cijfers na de komma
420 H=100: D=INT(10000*B+100)/H
430 IF A$="SQR" THEN D=D*D
440 'print de opgave
450 CLS: PRINT T; " ";
460 IF A$="^2" THEN PRINT D;A$="=";
470 IF A$="SQR" THEN PRINT"SQR("D")=";
480 'voer het resultaat in
490 INPUT G
500 'selecteer naar werkwijze
510 IF A$="^2" THEN 570
520 'bepaal of het antwoord goed is
530 IF ABS(SQR(D)-G)<.1/H THEN 690
540 'antwoord niet goed
550 PRINT: PRINT"fout, "; GOTO 620
560 'bepaal of het antwoord goed is
570 IF ABS(D^2-G)<.1/H THEN 690
580 'antwoord niet goed
590 PRINT: PRINT"fout, ";
600 'bij eerste poging, nog een keer
610 'bij tweede poging, geef juiste antw.
620 BEEP: IF S=1 THEN S=0: GOTO 720
630 S=S+1
640 PRINT"probeer nog een keer"
650 'vertraag
660 FOR I=1 TO 2000: NEXT: GOTO 450
670 'antwoord goed
680 'R is aantal goede antwoorden
690 R=R+1: PRINT:PRINT"goed ";
700 'geef muziek
710 PLAY"slm4661t90L8o5gg16g16g16g16abo6c"
720 IF A$="SQR" THEN F=SQR(D) ELSE F=D^2
730 PRINT F
740 'bepaal of serie is voltooid

```

```
750 IF T=20 THEN 780
760 PRINT"Volgende opgave:"
770 'vertraag, geef volgende opgave
780 FOR I=0 TO 2000: NEXT I: NEXT T
790 'de serie is voltooid.
800 'bij goed result. groene tekst
810 'bij slecht result. rode tekst
820 IF R<12 THEN COLOR 6 ELSE COLOR 12
830 PRINT: PRINT: PRINT"je hebt ";
840 PRINT R "sommen goed"
850 PRINT: PRINT: PRINT
860 'bij slecht result geen hogere graad
```

```
870 IF R<12 THEN 900
880 'hogere moeilijkheidsgraad
890 A=A+1: IF A=6 THEN 970
900 PRINT"Volgende moeilijkheidsgraad";
910 PRINT A: R=0
920 'vertraag, daarna volgende serie
930 FOR I=0 TO 5000: NEXT I
940 COLOR 15: VPOKE 8197,26
950 VPOKE 8198,26
960 VPOKE 8199,26: CLS: GOTO 260
970 PRINT"Nog moeilijker kan ";
980 PRINT"ik het niet!": END
```

2. Het berekenen van een term in een reeks

Het volgende programma laat u steeds één opgave oplossen. De opgave zal soms zeer eenvoudig zijn maar in andere gevallen bijzonder moeilijk. Omdat dit puur van het toeval afhangt, is er maar van afgezien om een soort puntenwaardering te geven. Het programma ontwikkelt een reeks van vijf getallen die in een onderling verband staan. Van deze vijf getallen worden de eerste vier op het scherm getoond. Het vijfde getal moet u berekenen en invoeren.

Heeft u de goede oplossing gegeven dan wordt dit vermeld en krijgt u een nieuwe opgave. Heeft u een foute oplossing gegeven dan wordt dit ook vermeld en krijgt u nog een keer de gelegenheid de goede oplossing in te voeren. Wordt opnieuw een foute oplossing gegeven dan krijgt u in beeld welk getal u had moeten invoeren. Ook dan wordt een nieuwe opgave getoond.

De reeks kan oplopend zijn, dat wil zeggen dat elke volgende term groter is dan zijn voorgaande. De reeks kan ook aflopend zijn. Elke volgende term is dan kleiner dan zijn voorgaande. Een derde mogelijkheid is dat de reeks intermitterend is. De termen zijn dan om en om groter en kleiner. Ook kunnen negatieve termen voorkomen. U ziet, er zijn mogelijkheden genoeg om steeds weer voor een verrassing te worden geplaatst. Soms wordt de volgende term gevonden door bij de voorgaande een getal op te tellen. Aan u de taak uit te vinden welk getal bij de vierde term moet worden opgeteld om de vijfde term te vinden.

In andere gevallen is elke volgende term te vinden door de voorgaande met een bepaald getal te vermenigvuldigen.

Na het invoeren en het starten van het programma kunt u met ja of nee antwoorden of u een aanwijzing nodig hebt. Antwoordt u met ja dan krijgt u de aanwijzing waarna u door de spatiebalk in te drukken weer terugkomt in het normale programma. Antwoordt u met nee dan wordt direct met het programma doorgedaan, hetgeen betekent dat u een reeks van vier getallen ziet. Deze reeks kan bijvoorbeeld zijn:

2 4 8 14

In dit geval is bij de eerste term $4 - 2 = 2$ opgeteld, bij de tweede term $8 - 4 = 4$ en bij de derde term $14 - 8 = 6$. Hieruit volgt dat u bij de vierde term 8 moet optellen om de vijfde term te vinden. Uw antwoord is:

$$14 + 8 = 22$$

De getoonde reeks had ook kunnen zijn:

1 3 18 216

Nu is de eerste term met 3 vermenigvuldigd: $1 * 3 = 3$. De tweede term is met 6 vermenigvuldigd: $3 * 6 = 18$ en de derde met 12: $18 * 12 = 216$. De gevolgtrekking hieruit is dat u de vierde term met 24 moet vermenigvuldigen om het juiste antwoord te vinden:

$$216 * 24 = 5184$$

Deze voorbeelden waren beslist nog niet de moeilijkste van de opgaven die mogelijk zijn. U zult zien dat dit programma echt niet alleen voor de wat oudere kinderen nuttig kan zijn!

```
10 'termen van reeksen.
20 'voor een echt random getal
30 PRINT"gebruik toets SELECT"
40 A=RND(1): A#=INKEY#
50 IF A#<>CHR$(24) THEN 40
60 INPUT"wil je uitleg, ja of nee";A#
70 IF A#="ja" THEN 570
80 IF A#<>"nee" THEN 60
90 'voor blauw veld en kader
100 'en witte tekst
110 SCREEN 1: COLOR 15,4,4: WIDTH(32)
120 'voor groene cijfers
130 VPOKE 8197,52: VPOKE 8198,52
140 VPOKE 8199,52
150 'def rnd getallen van 1 tot X
160 DEF FNI(X)=INT(RND(1)*X)+1
170 'rnd getallen voor A(0), B en C
180 A(0)=FNI(3): B=FNI(3): C=FNI(2): E=0
190 'bij T=-1 altenerende reeks
200 T=1: IF FNI(2)<2 THEN T=-1
210 'het berekenen v.d. termen A(1)-A(4)
220 FOR N=1 TO 4
230 IF C=1 THEN A(N)=ABS(A(N-1))+T^(N)*B*N
240 IF C=2 THEN A(N)=ABS(A(N-1))*T^(N)*B*N
250 NEXT N: CLS
```

```

260 'voor een aflopende reeks
270 IF FNI(2)<2 THEN 540
280 'het afbeelden van de reeks
290 FOR N=0 TO 3: PRINT A(N);: NEXT N
300 'invoeren van het antwoord
310 PRINT: PRINT
320 PRINT"Wat is het eerst volgende"
330 INPUT"getal van de reeks";G
340 'bepaal of het antwoord juist is
350 IF G<>A(4) THEN 440
360 'antwoord is goed
370 PRINT: PRINT "juist, ";
380 PRINT "de volgende reeks:"
390 'vertraag
400 FOR N=0 TO 3000: NEXT N
410 'voor een volgende opgave
420 GOTO 180
430 'antwoord is fout
440 PRINT: PRINT "fout, ";
450 'na tweede poging geef antwoord
460 IF E=1 THEN 520
470 E=E+1: PRINT"probeer nog een keer"
480 'vertraag
490 FOR N=0 TO 3000: NEXT N

```

```

500 'voor de tweede poging
510 CLS: GOTO 290
520 PRINT"de volgende term is"A(4)
530 FOR N=0 TO 7000: NEXT N: GOTO 180
540 FOR N=0 TO 4: D(N)=A(N): NEXT N
550 FOR N=0 TO 4: A(4-N)=D(N): NEXT N
560 GOTO 290
570 SCREEN 0: WIDTH(40): COLOR 10,4,4
580 PRINT "Op het scherm wordt ";
590 PRINT "een reeks van getal-len ";
600 PRINT "getoond die in een bepaald ";
610 PRINT "verband tot elkaar staan."
620 PRINT "Je moet nu het vijfde ";
630 PRINT "getal van de reeksgeven."
640 PRINT "Is de reeks bijvoorbeeld"
650 PRINT: PRINT "1 3 5 7"
660 PRINT: PRINT "dan is het vijfde ";
670 PRINT "cijfer 9."
680 PRINT: PRINT "Gebruik de ";
690 PRINT "spatiebalk voor het ";
700 PRINT "vervolgen van het ";
710 PRINT "programma."
720 A$=INKEY$: IF A$<>CHR$(32) THEN 720
730 GOTO 110

```

3. Conversie

Nu volgen een aantal programma's waarmee u getallen of eenheden van een bepaald stelsel kunt omzetten in een ander stelsel. De programma's hebben ongeveer dezelfde opzet en geven ook dezelfde indeling op uw scherm.

3.1. Conversie van inches naar millimeter v.v.

Op het vasteland van Europa wordt het metrieke maatstelsel gehanteerd. Helaas is dat niet over de gehele wereld het geval en vinden we daarnaast ook nog het Engelse maatstelsel. Om ons van een opgegeven Engelse maat een voorstelling te kunnen maken, is dan ook omrekening naar het metrieke stelsel nodig. Het volgende programma geeft u de mogelijkheid een lengtemaat, opgegeven in inches, om te rekenen in millimeter. Ook het omgekeerde is mogelijk, omrekening van een lengtemaat in millimeter naar inches. Dit kunt u kiezen door I/M of M/I in te toetsen voor omzetting van respectievelijk van inches naar millimeter of van millimeter naar inches. Daarna kunt u de grootte van het getal invoeren.

Heeft u gekozen voor conversie van inches naar millimeter en heeft u daarna het getal 4 ingevoerd dan geeft uw scherm het volgende weer:

CONVERSIE VAN EENHEDEN.

WELK GETAL? 4

4 INCHES KOMT OVEREEN MET
101.6 MILLIMETER

GEBRUIK VOOR EEN VOLGENDE
CONVERSIE DE F1-TOETS.

Bij de conversie is ervan uitgegaan dat 1 inch overeenkomt met 25.4 millimeter.

```
10 'voor interrupt met f1 toets
20 ON KEY GOSUB 440
30 'conversie millimeter-Inches
40 'voor blauw veld en kader
50 'en witte letters.
60 SCREEN 1: WIDTH(32): COLOR 15,4,4
70 'voor zwarte cijfers
80 VPOKE 8198,20: VPOKE 8199,20
90 'voor de titel van het programma
100 GOSUB 450
110 'keuze van de conversie
120 PRINT: PRINT: PRINT
130 PRINT "Wilt u van Inches naar ";
140 PRINT "millimeter (I/M),";
150 PRINT " of van millimeter"
160 PRINT "naar Inches (M/I)";
170 INPUT K$
180 'opnieuw bij verkeerde invoer
190 IF K$(">"I/M" AND K$(">"M/I" THEN 60
200 'print opnieuw de titel
210 GOSUB 450
220 PRINT: PRINT: PRINT
230 'invoeren van het getal
240 INPUT "Welk getal";G: PRINT
250 'selectie naar conversiekeuze
260 IF K$="I/M" THEN 320
270 'omrekening mm/Inches
280 R=.0393700787##G
290 M$="millimeter": N$=" Inches"
300 GOTO 350
310 'omrekening Inches/mm
320 R=25.4*G
330 M$="Inches": N$=" millimeter"
340 'het printen van het resultaat
350 PRINT G;M$
360 PRINT"komt overeen met"
370 PRINT R;N$: PRINT: PRINT: PRINT
380 PRINT"Gebruik voor een volgende"
390 PRINT"conversie toets f1."
400 'schakel f1 toets in
410 KEY(1) ON
420 'wacht op interrupt
430 GOTO 430
440 KEY(1) OFF: RETURN 60
450 CLS
460 PRINTTAB(5)"conversie van eenheden"
470 RETURN
```

3.2. Conversie van foot naar meter v.v.

Het volgende programma maakt het u mogelijk een lengtemaat die in Engelse voet (foot) is uitgedrukt, om te zetten in meters. Ook het omgekeerde is weer mogelijk. Voor de omzetting foot-meter moet u F/M intoetsen, voor het omzetten van

meter naar foot M/F. Heeft u gekozen voor F/M en het getal 3 ingevoerd dan is het antwoord:

CONVERSIE VAN EENHEDEN.

WELK GETAL? 3

3 FEET KOMT OVEREEN MET
.9144 METER

GEBRUIK VOOR EEN VOLGENDE CONVERSIE DE F1-TOETS.

Bij de conversie is ervan uitgegaan dat 1 foot overeenkomt met 0.3048 meter.

```
10 'conversie foot-meter
20 'voor interrupt met f1 toets
30 ON KEY GOSUB 440
40 'voor blauw veld en kader
50 'en witte letters.
60 SCREEN 1: WIDTH(32): COLOR 15,4,4
70 'voor zwarte cijfers
80 VPOKE 8198,20: VPOKE 8199,20
90 'voor de titel van het programma
100 GOSUB 450
110 'keuze van de conversie
120 PRINT: PRINT: PRINT
130 PRINT "Wilt u van foot naar ";
140 PRINT "meter (F/M)"
150 PRINT "of van meter ";
160 PRINT "naar foot (M/F)"
170 INPUT K$
180 'opnieuw bij verkeerde invoer
190 IF K$<>"F/M" AND K$<>"M/F" THEN 60
200 'print opnieuw de titel
210 GOSUB 450
220 PRINT: PRINT: PRINT
230 'invoeren van het getal
240 INPUT "Welk getal";G: PRINT
250 'selectie naar conversiekeuze
260 IF K$="F/M" THEN 320
270 'omrekening meter/foot
280 R=3.2808399#*G
290 M$="meter": N$=" foot"
300 GOTO 350
310 'omrekening foot/meter
320 R=.3048#G
330 M$="foot": N$=" meter"
340 'het printen van het resultaat
350 PRINT G;M$;
360 PRINT " komt overeen met"
370 PRINT R;N$: PRINT: PRINT: PRINT
380 PRINT"Gebruik voor een volgende"
390 PRINT"conversie toets f1."
```

```
400 'schakel f1 toets in
410 KEY(1) ON
420 'wacht op interrupt
430 GOTO 430
440 KEY(1) OFF: RETURN 60
450 CLS
460 PRINTTAB(5)"conversie van eenheden"
470 RETURN
```

3.3. Conversie van mile naar kilometer v.v.

Dit programma komt overeen met het voorgaande. Nu is het de conversie tussen mile en kilometer. Voor omzetting van mile naar kilometer moet u M/KM invoeren en voor omzetting van kilometer naar mile KM/M. Heeft u M/KM en het getal 2 ingevoerd dan is het resultaat:

CONVERSIE VAN EENHEDEN.

WELK GETAL? 2

2 MILES KOMT OVEREEN MET
3.218688 KILOMETER

GEBRUIK VOOR EEN VOLGENDE CONVERSIE DE F1-TOETS.

Voor de conversie is ervan uitgegaan dat 1 mile overeenkomt met 1.609344 kilometer.

```
10 'conversie mile-kilometer
20 'voor interrupt met f1 toets
30 ON KEY GOSUB 440
40 'voor blauw veld en kader
50 'en witte letters.
60 SCREEN 1: WIDTH(32): COLOR 15,4,4
70 'voor zwarte cijfers
80 VPOKE 8198,20: VPOKE 8199,20
90 'voor de titel van het programma
100 GOSUB 450
110 'keuze van de conversie
120 PRINT: PRINT: PRINT
130 PRINT "Wilt u van mile naar ";
140 PRINT "kilometer (M/KM)";
150 PRINT " of van kilometer naar"
160 PRINT "mile (KM/M)";
170 INPUT K$
180 'opnieuw bij verkeerde invoer
190 IF K$<>"M/KM" AND K$<>"KM/M" THEN 60
```

```

200 'print opnieuw de titel
210 GOSUB 450
220 PRINT: PRINT: PRINT
230 'invoeren van het getal
240 INPUT "Welk getal";G: PRINT
250 'selectie naar conversiekeuze
260 IF K$="M/KM" THEN 320
270 'omrekening kilometer/mile
280 R=.621371193##*G
290 M$="kilometer": N$=" miles"
300 GOTO 350
310 'omrekening mile/kilometer
320 R=1.609344##*G
330 M$="miles": N$=" kilometer"
340 'het printen van het resultaat
350 PRINT G;M$
360 PRINT"komt overeen met"
370 PRINT R;N$: PRINT: PRINT: PRINT
380 PRINT"Gebbruik voor een volgende"
390 PRINT"conversie toets f1."
400 'schakel f1 toets in
410 KEY(1) ON
420 'wacht op interrupt
430 GOTO 430
440 KEY(1) OFF: RETURN 60
450 CLS
460 PRINTTAB(5)"conversie van eenheden"
470 RETURN

```

3.4. Conversie van gram naar ounce v.v.

Ook bij het meten van massa's worden in het Engelse maatstelsel en het metrieke stelsel verschillende eenheden gebruikt. Het volgende programma is voor de omzetting van gram naar ounce. Dit omzetten is ook weer in twee richtingen mogelijk.

Voor de omzetting van gram naar ounce moet GR/O worden ingevoerd, voor de omzetting van ounce naar gram moet O/GR worden ingevoerd. Heeft u gekozen voor O/GR en heeft u het getal 7 ingevoerd dan vertoont het scherm:

CONVERSIE VAN EENHEDEN.

WELK GETAL? 7

7 OUNCES KOMT OVEREEN MET
198.446662 GRAM

GEBRUIK VOOR EEN VOLGENDE
CONVERSIE DE F1-TOETS.

Bij de conversie is ervan uitgegaan dat 1 ounce overeenkomt met 28.3495231 gram.

```

10 'conversie ounce-gram
20 'voor interrupt met f1 toets
30 ON KEY GOSUB 440
40 'voor blauw veld en kader
50 'en witte letters,
60 SCREEN 1: WIDTH(32): COLOR 15,4,4
70 'voor zwarte cijfers
80 VPOKE 8198,20: VPOKE 8199,20
90 'voor de titel van het programma
100 GOSUB 450
110 'keuze van de conversie
120 PRINT: PRINT: PRINT
130 PRINT "Wilt u van ounce naar ";
140 PRINT "gram": PRINT"(O/GR)";
150 PRINT " of van gram naar ";
160 PRINT "ounce"
170 INPUT"(GR/O)";K$
180 'opnieuw bij verkeerde invoer
190 IF K$<>"O/GR" AND K$<>"GR/O" THEN 60
200 'print opnieuw de titel
210 GOSUB 450
220 PRINT: PRINT: PRINT
230 'invoeren van het getal
240 INPUT "Welk getal";G: PRINT
250 'selectie naar conversiekeuze
260 IF K$="O/GR" THEN 320
270 'omrekening gram/ounce
280 R=.035273962##*G
290 M$="gram": N$=" ounces"
300 GOTO 350
310 'omrekening ounce/gram
320 R=28.3495231##*G
330 M$="ounces": N$=" gram"
340 'het printen van het resultaat
350 PRINT G;M$
360 PRINT"komt overeen met"
370 PRINT R;N$: PRINT: PRINT: PRINT
380 PRINT"Gebbruik voor een volgende"
390 PRINT"conversie toets f1."
400 'schakel f1 toets in
410 KEY(1) ON
420 'wacht op interrupt
430 GOTO 430
440 KEY(1) OFF: RETURN 60
450 CLS
460 PRINTTAB(5)"conversie van eenheden"
470 RETURN

```

3.5. Conversie van pound naar kilogram v.v.

Dit programma verloopt overeenkomstig het voorgaande. Wilt u een omzetting van pound (libra, afkorting lb) naar kilogram dan dient u LB/KG in te voeren. Voor een omzetting van kilogram naar pound moet KG/LB worden ingevoerd. Hebt u gekozen voor LB/KG en heeft u het getal 4 ingevoerd dan is het resultaat:

CONVERSIE VAN EENHEDEN.

WELK GETAL? 4

4 POUNDS KOMT OVEREEN MET
1.81436948 KILOGRAM

GEBRUIK VOOR EEN VOLGENDE
CONVERSIE DE F1-TOETS.

Voor de conversie is ervan uitgegaan dat 1 pound overeenkomt met 0.45359237 kilogram.

```
10 'conversie pound-kilogram
20 'voor interrupt met f1 toets
30 ON KEY GOSUB 440
40 'voor blauw veld en kader
50 'en witte letters.
60 SCREEN 1: WIDTH(32): COLOR 15,4,4
70 'voor zwarte cijfers
80 VPOKE 8198,20: VPOKE 8199,20
90 'voor de titel van het programma
100 GOSUB 450
110 'keuze van de conversie
120 PRINT: PRINT: PRINT
130 PRINT "Wilt u van pound naar ";
140 PRINT "kilogram": PRINT "(LB/KG)";
150 PRINT " of van kilogram naar "
160 PRINT "pound (KG/LB)";
170 INPUT K$
180 'opnieuw bij verkeerde invoer
190 IF K$("<"LB/KG" AND K$("<"KG/LB" THEN 60
200 'print opnieuw de titel
210 GOSUB 450
220 PRINT: PRINT: PRINT
230 'invoeren van het getal
240 INPUT "Welk getal";G: PRINT
250 'selectie naar conversiekeuze
260 IF K$="LB/KG" THEN 320
270 'omrekening kilogram/pound
280 R=2.20462262##G
290 M$="kilogram": N$="pounds"
300 GOTO 350
310 'omrekening pound/kilogram
320 R=.45359237##G
330 M$="pounds": N$="kilogram"
340 'het printen van het resultaat
350 PRINT G;M$
360 PRINT"komt overeen met"
370 PRINT R;N$: PRINT: PRINT: PRINT
380 PRINT"Gebruik voor een volgende"
390 PRINT"conversie toets f1."
400 'schakel f1 toets in
410 KEY(1) ON
420 'wacht op interrupt
430 GOTO 430
440 KEY(1) OFF: RETURN 60
450 CLS
460 PRINTTAB(5)"conversie van eenheden"
470 RETURN
```

3.6. Conversie van gallon naar liter v.v.

De gallon is een inhoudsmaat van het Engelse maatstelsel. Het volgende programma geeft de omzetting tussen de gallon en de liter. Wilt u een omzetting van de gallon naar de liter dan dient GAL/L te worden ingevoerd. Voor de omzetting van liter naar gallon: L/GAL. Heeft u voor GAL/L gekozen en heeft u het getal 3 ingevoerd dan vertoont het scherm:

CONVERSIE VAN EENHEDEN.

WELK GETAL? 3

3 GALLONS KOMT OVEREEN MET
11.3562353 LITER

GEBRUIK VOOR EEN VOLGENDE
CONVERSIE DE F1-TOETS.

Voor de conversie is ervan uitgegaan dat 1 gallon overeenkomt met 3.78541178 liter.

```
10 'conversie gallon-liter
20 'voor interrupt met f1 toets
30 ON KEY GOSUB 440
40 'voor blauw veld en kader
50 'en witte letters.
60 SCREEN 1: WIDTH(32): COLOR 15,4,4
70 'voor zwarte cijfers
80 VPOKE 8198,20: VPOKE 8199,20
90 'voor de titel van het programma
100 GOSUB 450
110 'keuze van de conversie
120 PRINT: PRINT: PRINT
130 PRINT "Wilt u van gallon naar ";
140 PRINT "liter": PRINT "(GAL/L)";
150 PRINT " of van liter naar "
160 PRINT "gallon (L/GAL)";
170 INPUT K$
180 'opnieuw bij verkeerde invoer
190 IF K$("<"GAL/L" AND K$("<"L/GAL" THEN 60
200 'print opnieuw de titel
210 GOSUB 450
220 PRINT: PRINT: PRINT
230 'invoeren van het getal
240 INPUT "Welk getal";G: PRINT
250 'selectie naar conversiekeuze
260 IF K$="GAL/L" THEN 320
270 'omrekening liter/gallon
```

```

280 R=.264172052##G
290 M$="liter": N$="gallons"
300 GOTO 350
310 'omrekening gallon/liter
320 R=3.78541178##G
330 M$="gallons": N$="liter"
340 'het printen van het resultaat
350 PRINT G;M$
360 PRINT"komt overeen met"
370 PRINT R;N$: PRINT: PRINT: PRINT
380 PRINT"Gebruik voor een volgende"
390 PRINT"conversie toets f1."
400 'schakel f1 toets in
410 KEY(1) ON
420 'wacht op interrupt
430 GOTO 430
440 KEY(1) OFF: RETURN 60
450 CLS
460 PRINTTAB(5)"conversie van eenheden"
470 RETURN

```

3.7. Conversie van graden naar radialen v.v.

De grootte van een hoek is zowel in graden als in radialen uit te drukken. De omrekening tussen radialen en graden verzorgt het volgende programma. Wilt u een omrekening van graden naar radialen dan moet GR/RD worden ingevoerd. Voor een omrekening van radialen naar graden voert u RD/GR in. Heeft u gekozen voor RD/GR en is het ingevoerde getal 3 dan is het resultaat:

CONVERSIE VAN EENHEDEN.

WELK GETAL? 3

3 RADIALEN KOMT OVEREEN MET 171.887339 GRADEN

GEBRUIK VOOR EEN VOLGENDE CONVERSIE DE F1-TOETS.

Voor de conversie is ervan uitgegaan dat 1 radiaal overeenkomt met 57.2957795 graden.

```

10 'conversie graden-radialen
20 'voor interrupt met f1 toets
30 ON KEY GOSUB 440
40 'voor blauw veld en kader

```

```

50 'en witte letters.
60 SCREEN 1: WIDTH(32): COLOR 15,4,4
70 'voor zwarte cijfers
80 VPOKE 8198,20: VPOKE 8199,20
90 'voor de titel van het programma
100 GOSUB 450
110 'keuze van de conversie
120 PRINT: PRINT: PRINT
130 PRINT "Wilt u van graden naar ";
140 PRINT "radialen": PRINT"(GR/RD)";
150 PRINT " of van radialen naar "
160 PRINT "graden (RD/GR)";
170 INPUT K$
180 'opnieuw bij verkeerde invoer
190 IF K$<>"GR/RD" AND K$<>"RD/GR" THEN 60
200 'print opnieuw de titel
210 GOSUB 450
220 PRINT: PRINT: PRINT
230 'invoeren van het getal
240 INPUT "Welk getal";G: PRINT
250 'selectie naar conversiekeuze
260 IF K$="RD/GR" THEN 320
270 'omrekening graden/radialen
280 R=.01745329252##G
290 M$="graden": N$="radialen"
300 GOTO 350
310 'omrekening radialen/graden
320 R=57.2957795##G
330 M$="radialen": N$="graden"
340 'het printen van het resultaat
350 PRINT G;M$
360 PRINT"komt overeen met"
370 PRINT R;N$: PRINT: PRINT: PRINT
380 PRINT"Gebruik voor een volgende"
390 PRINT"conversie toets f1."
400 'schakel f1 toets in
410 KEY(1) ON
420 'wacht op interrupt
430 GOTO 430
440 KEY(1) OFF: RETURN 60
450 CLS
460 PRINTTAB(5)"conversie van eenheden"
470 RETURN

```

3.8. Conversie van Arabische naar Romeinse cijfers v.v.

De tekens die wij normaal gebruiken voor het schrijven van getallen zijn Arabische cijfertekens. Op sommige wijzerplaten van klokken worden de uren met Romeinse cijfertekens aangegeven. Ook zien we deze tekens wel eens gebruikt op gebouwen om het bouwjaar aan te geven. Met het volgende programma is het mogelijk de getallen van de beide stelsels om te zetten. Wilt u van Arabische cijfers naar Romeinse cijfers dan voert u A/R in. Voor omzetting van getallen in Romeinse cijfers naar Arabische cijfers gebruikt u R/A. Heeft u A/R gekozen en is het ingevoerde getal 1984 dan is het resultaat:

CONVERSIE VAN GETALLEN.

WELK GETAL? 1984

HET GETAL 1984 WORDT IN ROMEINSE TEKENS: MCMLXXXIV

GEBRUIK VOOR EEN VOLGENDE CONVERSIE DE F1-TOETS.

```
10 'conversie Romeins-Arabisch
20 DIM E(16)
30 'voor interrupt met f1 toets
40 ON KEY GOSUB 720
50 'voor blauw veld en kader
60 'en witte letters.
70 SCREEN 1: WIDTH(32): COLOR 15,4,4
80 'voor zwarte cijfers
90 VPOKE 8198,20: VPOKE 8199,20
100 'voor de titel van het programma
110 GOSUB 730
120 'keuze van de conversie
130 PRINT: PRINT: PRINT
140 PRINT "Wilt u van Romeins naar ";
150 PRINT "Arabisch": PRINT"(R/A)";
160 PRINT " of van Arabisch naar "
170 PRINT: PRINT "Romeins (A/R)";
180 INPUT K$
190 'opnieuw bij verkeerde invoer
200 IF K$<>"R/A" AND K$<>"A/R" THEN 70
210 'print opnieuw de titel
220 GOSUB 730
230 PRINT: PRINT: PRINT
240 'invoeren van het getal
250 INPUT "Welk getal";G$: PRINT
260 'selectie naar conversiekeuze
270 IF K$="R/A" THEN 480
280 'omrekening Arabisch/Romeins
290 P$(1)="I": Q$(1)="V": R$(1)="X"
300 P$(2)="X": Q$(2)="L": R$(2)="C"
```

```
310 P$(3)="C": Q$(3)="D": R$(3)="M"
320 P$(4)="M": E$="": L=LEN(G$)
330 'geen getallen groter dan 3999
340 IF VAL(G$)>3999 THEN 180
350 T=L+1: FOR N=L TO 1 STEP -1
360 C(N)=VAL(MID$(G$,T-N,1))
370 IF C(N)=0 THEN 450
380 A$(1)=P$(N): A$(2)=A$(1)+P$(N)
390 A$(3)=A$(2)+P$(N): A$(4)=P$(N)+Q$(N)
400 IF N=4 THEN 440
410 A$(5)=Q$(N): A$(6)=A$(5)+P$(N)
420 A$(7)=A$(5)+A$(2): A$(8)=A$(5)+A$(3)
430 A$(9)=P$(N)+R$(N)
440 E$=E$+A$(C(N))
450 NEXT N: S$=" Romeinse"
460 GOTO 620
470 'conversie Romeins/arabisch
480 L=LEN(G$): T=L+1: E=0
490 FOR N=1 TO L
500 C$=MID$(G$,T-N,1)
510 IF C$="I" THEN E(N)=1
520 IF C$="V" THEN E(N)=5
530 IF C$="X" THEN E(N)=10
540 IF C$="L" THEN E(N)=50
550 IF C$="C" THEN E(N)=100
560 IF C$="D" THEN E(N)=500
570 IF C$="M" THEN E(N)=1000
580 IF E(N)<E(N-1) THEN E(N)=-E(N)
590 E=E+E(N): NEXT N: E$=STR$(E)
600 S$=" Arabische"
610 'het printen van het resultaat
620 PRINT: PRINT "Het getal "G$" wordt"
630 PRINT "in "S$;
640 PRINT " tekens: ": PRINT E$
650 PRINT: PRINT: PRINT
660 PRINT"Gebruik voor een volgende"
670 PRINT"conversie toets f1."
680 'schakel f1 toets in
690 KEY(1) ON
700 'wacht op interrupt
710 GOTO 710
720 KEY(1) OFF: RETURN 70
730 CLS
740 PRINTTAB(5)"conversie van eenheden"
750 RETURN
```

4. Rekenprogramma's

Onder dit hoofdstuk wordt een aantal programma's gegeven waarbij de computer uitsluitend als rekenmachine wordt gebruikt. Soms zijn overeenkomstige programma's opgenomen in de zogenoemde modulen van programmeerbare rekenmachines.

4.1. Het berekenen van priemgetallen

Een priemgetal is een getal dat slechts deelbaar is door 1 en door zichzelf. Dit programma berekent alle priemgetallen die zich bevinden tussen de twee grenzen die door uzelf kunnen worden gekozen (de getallen die de grenzen aangeven inbegrepen). Na het starten van het programma kunt u eerst de onderste grens invoeren en daarna de bovenste grens. Direct daarop worden de gevraagde priemgetallen op het scherm geprojecteerd. Betreffen het grote getallen dan kan het resultaat wel enige tijd op zich laten wachten. Wilt u de priemgetallen weten vanaf 17 tot en met 40 dan voert u deze getallen als grenzen in waardoor het volgende op uw scherm verschijnt:

```
WAT IS DE ONDERGRENS? 17
WAT IS DE BOVENGRENS? 40
```

```
17 19 23 29 31 37
```

```
10 'priemgetallen
20 'voor interrupt met f1 toets
30 ON KEY GOSUB 440
40 'voor blauw veld en kader
50 'en voor witte letters
60 SCREEN 0: WIDTH(40): COLOR 15,4,4
70 'het printen van de titel
80 PRINT TAB(13)"Priemgetallen"
90 PRINT: PRINT: PRINT: B=1
100 'het invoeren van de grenzen
110 INPUT"Wat is de ondergrens";OG
120 INPUT"Wat is de bovengrens";BG
130 PRINT: PRINT
140 A=INT((OG+1)/6): IF A<1 THEN A=1
150 IF OG<4 THEN P=OG-1
160 IF OG>3 THEN 200
170 'print P als P<5
180 P=P+1: GOTO 340
190 'kies een getal P
200 B=-B: P=6*A+B: IF B>0 THEN A=A+1
```

```
210 'P steeds groter dan de ondergrens
220 IF P<OG THEN 200
230 'P steeds kleiner dan de bovengrens
240 D=1: C=1: IF P>BG THEN 370
250 'kies Q als deler van P
260 D=-D: Q=6*C+D: IF D>0 THEN C=C+1
270 'Q hoeft niet groter dan SQR(P)
280 IF INT(Q^2)>P THEN 340
290 'bepaal of Q en deler is van P
300 E=INT(P/Q)*Q: IF E<P THEN 260
310 'zo ja, kies en nieuwe waarde voor P
320 IF E=P THEN 200
330 'print P als P een priemgetal is
340 PRINT P;: IF P<3 THEN 180
350 'voor een volgend priemgetal
360 GOTO 200
370 'schakel f1 toets in
380 KEY(1) ON
390 PRINT: PRINT: PRINT: PRINT: PRINT
400 PRINT"gebruik voor een volgende ";
410 PRINT"serie toets f1"
420 'wacht op interrupt
430 GOTO 430
440 KEY(1) OFF: RETURN 60
```

4.2. Het berekenen van het kleinste gemene veelvoud en de grootste gemene deler

Met dit programma kunt u de grootste gemene deler (ggd) en het kleinste gemene veelvoud (kgv) berekenen van een willekeurig aantal getallen, mits dit aantal maar kleiner is dan tien. Na het invoeren en het starten van het programma moet u eerst het aantal van de getallen invoeren en daarna de getallen zelf. Deze mogen in een willekeurige volgorde worden ingevoerd, dus niet noodzakelijk in de volgorde van grootte. Als u de ggd en het kgv van de getallen 4284, 20196 en 302328 wilt weten dan voert u eerst het cijfer 3 in en daarna de getallen. Het schermbeeld is dan als volgt:

```
HOEVEEL GETALLEN (MAX. 10)? 3
```

```
GETAL 1? 4284
GETAL 2? 302328
GETAL 3? 20196
```

```
HET KLEINSTE GEMENE VEELVOUD
IS 69837768
```

DE GROOTSTE GEMENE DELER IS 612

GEBRUIK VOOR DE VOLGENDE SERIE GETALLEN DE F1-TOETS

Zie dat de volgorde van het invoeren van de getallen hier willekeurig is geweest!

```
10 'ggd en kgv
20 'voor interrupt met f1 toets
30 ON KEY GOSUB 850
40 'voor blauw veld en kader
50 'en voor witte letters
60 SCREEN 0: WIDTH(40): COLOR 15,4,4
70 'het printen van de titel
80 PRINT TAB(15)"ggd en kgv"
90 PRINT: PRINT: PRINT
100 'invoer aantal getallen
110 PRINT"Hoeveel getallen (max. 10)";
120 'Q is een priemdelers
130 INPUT I: PRINT: Q=2
140 'niet meer dan 10 getallen
150 IF I>10 THEN 110
160 'invoer van de getallen
170 FOR N=1 TO I: PRINT"Getal";N;
180 INPUT D(N): NEXT N:KV=1:GD=1:A=1:B=1
190 'getallen in volgorde van grootte
200 FOR T=1 TO I-1: FOR N=1 TO I-T
210 IF D(N)>D(N+1) THEN SWAP D(N),D(N+1)
220 NEXT N: NEXT T
230 'bepaal Q^T voor elk getal
240 FOR N=1 TO I: GOSUB 660
250 'bepaal Q^T met nieuwe Q
260 'en bewaar de laatste waarde van Q
270 IF QA>Q THEN QB=QA: QA=0: GOTO 240
280 NEXT N
290 'per priemdelers wordt de macht
300 'van Q bepaald voor ggd en kgv
310 K=T(1): G=K: FOR N=2 TO I
320 IF T(N)<G THEN G=T(N)
330 IF T(N)>K THEN K=T(N)
340 NEXT N
350 'per priemdelers wordt
360 'ggd en kgv berekend
370 KV=KV*INT(Q^K): GD=GD*INT(Q^G)
380 'als elk getal D(N) is opgedeeld
390 'door de priemdelers dan
400 'is hun rest 1
410 DK=1: FOR N=1 TO I
420 DK=DK*D(N): NEXT N
430 'als DK=1 dan is resultaat bekend
440 IF ABS(DK-1)<.1 THEN 530
450 'eventueel opnieuw met de
460 'vorige priemdelers
470 IF QB>0 THEN Q=QB: QB=0: GOTO 240
480 IF Q=2 THEN Q=3: GOTO 240
490 'bepaal een nieuwe priemdelers
500 B=-B: Q=6*A+B: IF B>0 THEN A=A+1
510 GOTO 240
520 'het printen van het resultaat
```

```
530 PRINT: PRINT: PRINT
540 PRINT"Het kleinste gemene ";
550 PRINT"veelvoud is": PRINT KV
560 PRINT: PRINT"De grootste gemene";
570 PRINT" deler is": PRINT GD
580 PRINT: PRINT: PRINT
590 PRINT"Gebruik voor een volgende ";
600 PRINT"serie toets f1"
610 'schakel toets f1 in
620 KEY(1) ON
630 'wacht op interrupt
640 GOTO 640
650 'subr. bepaalt Q^T
660 T=-1: D=D(N)
670 'getal is reeds opgedeeld
680 IF D(N)=1 THEN T(N)=0: RETURN
690 'voor de laatste priemdelers
700 IF (INT(Q^2)>D) AND (QB=0) THEN 830
710 'voor het aantal delingen
720 T=T+1: D=INT(D)
730 'getal is opgedeeld
740 IF D=1 THEN DT=1: GOTO 800
750 'opnieuw als de deling opgaat
760 DT=D/Q
770 DE=INT(DT)
780 IF ABS(DE*Q-D)<.1 THEN D=DT:GOTO 720
790 'de laatste deling is uitgevoerd
800 T(N)=T: D(N)=D: RETURN
810 'de priemdelers wordt gelijk D(N)
820 'QA wordt als de vorige priemdelers
830 QA=Q: Q=D: RETURN
840 'RUN voor resetten v.d. variabelen
850 KEY(1) OFF: RUN
```

4.3. Bewerkingen met breuken

Het nu volgende programma is voor het rekenen met breuken. De volgende bewerkingen zijn hierbij mogelijk: optellen, aftrekken, vermenigvuldigen en delen: de vier basisfuncties van elke calculator. Na het starten van het programma worden eerst de teller en de noemer van de eerste breuk ingevoerd, dan het bewerkingsteken en daarna de teller en de noemer van de tweede breuk:

BEWERKINGEN MET BREUKEN

```
TELLER? 3
NOEMER? 4
BEWERKING? +
TELLER? 7
NOEMER? 8
```

Nu wordt de berekening $3/4 + 7/8$ uitgevoerd:

BEWERKINGEN MET BREUKEN

$$(3/4) + (7/8) = 13/8$$

BEWERKING? =

U kunt meerdere berekeningen na elkaar maken, vandaar dat u nu opnieuw een bewerkingsteken kunt invoeren (ook weer +, -, * of /) met daarna de teller en de noemer van de volgende breuk. Met deze breuk en met het resultaat van de vorige berekening wordt dan de nieuwe bewerking uitgevoerd. In dit geval is het = teken ingevoerd. De computer geeft daarna het definitieve resultaat:

BEWERKINGEN MET BREUKEN

$$1 + 5/8$$

GEBRUIK VOOR EEN VOLGENDE BEWERKING TOETS F1

Met dit programma is dus een samengestelde berekening mogelijk met een aantal breuken (het aantal is onbeperkt) en na elke berekening kan een willekeurige bewerking worden gekozen. Nadat het = teken is gebruikt, wordt het definitieve resultaat bekend gemaakt. Is een volgende serie berekeningen nodig, dan moet de f1-toets worden gebruikt.

```

10 'bewerkingen met breuken
20 'voor interrupt met de f1 toets
30 ON KEY GOSUB 1020
40 'voor blauw veld en kader
50 'en witte letters.
60 SCREEN 0: WIDTH(40): COLOR 15,4,4
70 PRINT"Wilt u een aanwijzing";
80 PRINT" ja of nee"; INPUT A$
90 IF A$="ja" THEN 790
100 IF A$<>"nee" THEN 60
110 'voor het printen van de titel
120 GOSUB 610
130 S$="+-*/"
140 'het invoeren van de eerste breuk
150 INPUT"teller";RT: INPUT"noemer";RN
160 'variabele I geeft de bewerking aan
170 INPUT"bewerking";B$: FOR I=1 TO 4
180 IF B$=MID$(S$,I,1) THEN 230
190 NEXT: IFB$="" THEN 460
200 'opnieuw bij verkeerde invoer

```

```

210 GOTO 170
220 'het invoeren van de volgende breuk
230 INPUT"teller";T: INPUT"noemer";N
240 'selectie naar bewerking
250 ON I GOTO 270,380,420,440
260 'voor het bepalen van het kgv
270 A=RN: B=N: GOSUB 650
280 'voer de bewerking + uit.
290 TT=KV/RN*RT+KV/N*T: TN=KV
300 'voor het berekenen van het ggd
310 A=TT: B=TN: GOSUB 650
320 'vereenvoudig de breuk
330 TT=TT/GD: TN=TN/GD
340 'voor het printen van het resultaat
350 'terug voor de volgende bewerking
360 GOSUB 740: GOTO 170
370 'bepaal het kgv
380 A=RN: B=N: GOSUB 650
390 'voer de bewerking - uit
400 TT=KV/RN*RT-KV/N*T: TN=KV: GOTO 310
410 'voer de bewerking * uit
420 TT=RT*T: TN=RN*N: GOTO 310
430 'inverse voor de bewerking /
440 SWAP T,N: GOTO 420
450 'bereken het eindresultaat
460 H=0
470 IF TT<TN THEN 500
480 H=H+1: TT=TT-TN: GOTO 470
490 'het printen van het resultaat
500 GOSUB 610: IF H=0 THEN 530
510 PRINT H; IF TT=0 THEN 540
520 PRINT"+";
530 PRINT TT/"TN
540 PRINT: PRINT: PRINT
550 PRINT"Gebruik voor een volgende ";
560 PRINT"bewerking": PRINT"toets f1."
570 'schakel toets f1 in
580 KEY(1) ON
590 'wacht op interrupt
600 GOTO 600
610 CLS: PRINT TAB(5);
620 PRINT"Bewerkingen met breuken."
630 PRINT: PRINT: RETURN
640 'subroutine berekent ggd en kgv
650 A=ABS(A): B=ABS(B)
660 IF A>B THEN 680
670 SWAP A,B
680 PR=A*B
690 Q=A-INT(A/B)*B: IF Q=0 THEN 710
700 A=B: B=Q: GOTO 690
710 GD=B: KV=PR/B: RETURN
720 'subroutine print het resultaat
730 'van de bewerking
740 GOSUB 610: PRINT("RT"/"RN);
750 IF B$="/" THEN SWAP T,N
760 PRINT") "B$ ("T"/"N") ="
770 PRINT: PRINT TT/"TN
780 PRINT: RT=TT: RN=TN: RETURN
790 CLS: COLOR 10
800 PRINT"Voer eerst de teller van ";
810 PRINT"de eerste breukin."
820 PRINT"Toets daarna de noemer ";
830 PRINT"van de eerste"
840 PRINT"breuk in."
850 PRINT"Type nu het bewerking";
860 PRINT"teken +, -, * of / in. ";
870 PRINT"Voer nu de tweede breuk ";
880 PRINT"in. Nu wordt";
890 PRINT"het antwoord van de ";

```



```

900 PRINT"eerste bewerking"
910 PRINT"getoond en kan de volgende ";
920 PRINT"breuk worden"
930 PRINT"ingevoerd. Een = ";
940 PRINT"teken geeft";
950 PRINT" het defini-tieve antwoord."
960 PRINT: PRINT: PRINT"Gebruik een ";
970 PRINT"willekeurige toets om "
980 PRINT"terug te keren naar het";
990 PRINT" programma."
1000 A$=INKEY$: IF A$="" THEN 1000
1010 CLS: COLOR 15: GOTO 110
1020 KEY(1) OFF: GOTO 110

```

4.4. Berekening van priemdelers

Priemdelers zijn de delers van een getal die tevens priemgetallen zijn. Zo zijn van het getal 30 de getallen 1, 2, 3 en 5 de priemdelers. Het getal 10 is wel een deler van 30 maar is geen priemdeler. Het volgende programma geeft van een bepaald ingevoerd getal de priemdelers. Elk geheel getal kan worden ingevoerd. Alleen kan bij een groot getal de uitslag wel even op zich laten wachten. Voert u het getal 69837768 in dan geeft het scherm het volgende weer:

PRIEMDELERS

WELK GETAL? 69837768

DE PRIEMDELERS VAN 69837768
ZIJN:

1 2 3 7 11 13 17 19

GEBRUIK VOOR HET VOLGENDE
GETAL DE F1-TOETS

Steeds zal na het indrukken van de f1-toets een nieuw getal kunnen worden ingevoerd, waarvan dan weer de priemdelers worden gegeven.

```

10 'berekening van priemdelers
20 'voor interrupt met de f1 toets
30 ON KEY GOSUB 460

```

```

40 'voor blauw veld en kader
50 'en witte letters.
60 SCREEN 0: WIDTH(40): COLOR 15,4,4
70 'het printen van de titel
80 CLS: PRINT TAB(6);
90 PRINT"Berekening priemdelers."
100 PRINT: PRINT: PRINT
110 'invoer van het getal
120 INPUT"Welk getal";G: Q=2: A=1: B=1
130 PRINT: PRINT: PRINT
140 PRINT"De priemdelers van"G"zijn:"
150 PRINT: PRINT"1 ";
160 'ga na of G een priemgetal is
170 T=0: IF INT(Q^2)>G THEN 350
180 'bepaal of Q een deler is
190 D=INT(G/Q): E=INT(D*Q)
200 IF E=G THEN 280
210 'Q was geen deler, nieuwe Q
220 IF Q<3 THEN Q=3: GOTO 170
230 B=-B: Q=6*A+B
240 IF B>0 THEN A=A+1
250 'voor de volgende deling
260 GOTO 170
270 'laatste priemdeler gevonden
280 G=D: IF G=1 THEN 360
290 'Q is een deler, probeer of Q
300 'nogmaals op G te delen is
310 IF T>0 THEN 190
320 'voor de volgende deler
330 PRINT Q;: T=1: GOTO 190
340 'G is zelf een priemgetal
350 PRINT G
360 PRINT: PRINT: PRINT
370 PRINT"Gebruik voor het ";
380 PRINT"volgende getal de"
390 PRINT"f1 toets."
400 'schakel f1 toets in
410 KEY(1) ON
420 'wacht op interrupt
430 GOTO 430
440 'RUN voor het resetten van de
450 'variabelen
460 KEY(1) OFF: RUN

```

4.5. Het omzetten van tiendelige naar gewone breuken v.v.

Bij elk programma voor het omzetten van getallen en dergelijke is steeds omzetting in twee richtingen mogelijk geweest. Bij dit programma is dat ook zo, hoewel de omzetting van gewone breuken naar tiendelige breuken erg eenvoudig is en feitelijk geen programma behoeft. Na het starten van het programma kunt u met T/G kiezen voor omzetting van tiendelige naar gewone breuken en met G/T kiezen voor de omzetting van gewone naar tiendelige breuken. Hebt u voor G/T gekozen dan kunt u de teller en de noemer van de breuk invoeren waarna het resultaat op het scherm verschijnt. Hebt u voor T/G gekozen dan kunt u het getal van de tien-

delige breuk invoeren. Is dat getal .125 dan is het resultaat:

HET OMZETTEN VAN BREUKEN

$$.125 = 1 / 8$$

GEBRUIK VOOR DE VOLGENDE OMZETTING DE F1-TOETS

Gebruikt u hierna de f1-toets dan kunt u in dit geval kiezen welke omzetting hierna zal plaatsvinden.

```
10 'omzetten van breuken
20 'voor interrupt met de f1 toets
30 ON KEY GOSUB 530
40 'voor blauw veld en kader
50 'en witte letters
60 SCREEN 0: WIDTH(40): COLOR 15,4,4
70 'voor het printen van de titel
80 GOSUB 490
90 PRINT"Wilt u van tiendelige naar";
100 PRINT" gewone breu- ken (T/G) of";
110 PRINT" van gewone naar tiendelige"
120 PRINT"breuken (G/T)";
130 INPUT K$: IF K$="T/G" THEN 240
140 'voor verkeerde invoer
150 IF K$<>"G/T" THEN 60
160 'voor het printen van de titel
170 GOSUB 490
180 'invoer van de gewone breuk
190 INPUT"teller";T: INPUT"noemer";N
200 'bereken de tiendelige breuk
210 G=T/N: GOSUB 490
220 'print het resultaat
230 PRINT T"/"N"="G: GOTO 400
240 'invoer v.d. tiendelige breuk
250 INPUT"getal";G: I=0: T=G
260 'voor het bepalen v.d. teller
270 'en de noemer
280 I=I+1: T=10*T: N=INT(10^I): D=INT(T)
290 IF ABS(D-N*G)<.1 THEN 320
300 GOTO 280
310 'voor het bepalen van de ggd
320 A=N: B=INT(T): PR=A*B
330 Q=A-INT(A/B)*B: IF Q=0 THEN 360
340 A=B: B=Q: GOTO 330
350 'het vereenvoudigen van de breuk
360 T=T/B: N=N/B
370 GOSUB 490
380 'het printen van het resultaat
390 PRINT G"="T"/"N
400 PRINT: PRINT: PRINT
410 PRINT"gebruik voor de ";
420 PRINT"volgende omzetting"
430 PRINT"de f1 toets."
440 'schakel f1 toets in
```

```
450 KEY(1) ON
460 'wacht op interrupt
470 GOTO 470
480 'subr. print de titel
490 CLS: PRINT TAB(9);
500 PRINT"Omzetten van breuken"
510 PRINT:PRINT:PRINT: RETURN
520 'start het programma opnieuw
530 KEY(1) OFF: RETURN 60
```

4.6. Functies (het berekenen van grafieken)

Het zal wel bekend zijn dat door twee punten precies een rechte lijn gaat. Ook bepaalde kromme lijnen liggen vast als twee punten daarvan gegeven zijn. Het volgende programma bepaalt de vergelijkingen van drie soorten functies als daarvan twee punten bekend zijn. Deze functies zijn:

$Y = a * X + b$, $Y = a * X^2 + b * X + c$ (met voorwaarden) en $Y = a * e^{(b * X)}$. Hierin is e de grondtal van het natuurlijke logaritme stelsel. Het is niet alleen de bedoeling dat de vergelijkingen worden bepaald aan de hand van twee punten, maar ook dat een vergelijking wordt berekend met behulp van een aantal punten die door metingen zijn verkregen. De gemeten punten zullen dan niet steeds met de lijn samenvallen omdat er in het algemeen wel meetfouten optreden. Van deze punten wordt daarom een gemiddelde bepaald en door middel van dat gemiddelde wordt de vergelijking berekend.

a. Na het starten van het programma kunt u kiezen uit de drie mogelijkheden met de toetsen 1, 2 of 3. Hier gaat het om de rechte lijn: $Y = a * X + b$. Daarvoor dient u dus toets 1 te gebruiken. Hierna komt de vraag hoeveel punten u wilt invoeren. Als het niet gaat om een karakteristiek die moet worden geconstrueerd aan de hand van een aantal metingen dan is het invoeren van twee punten genoeg. Het getal dat u moet invoeren is dus 2. Daarna kunt u de respectievelijke X en Y waarden van de punten invoeren. Na het invoeren van de laatste waarden wordt de vergelijking van de lijn getoond die door die punten gaat. Wilt u de punten P1(1,5) en P2(2,7) invoeren dan is het verloop als volgt: Na het starten van het programma wordt u gevraagd of u een aanwijzing wilt. Als u dit allemaal gelezen heeft is dat niet meer nodig en dus antwoordt u met nee. Nu toont het scherm:

KIES VOOR $Y = A * X + B$ TOETS 1,
VOOR $Y = A * X^2 + B * X + C$ TOETS 2,
EN VOOR $Y = A * E^X$ TOETS 3

U drukt nu toets 1 in en het vervolg is:

HOEVEEL PUNTEN?


Na het invoeren van het getal 2 en van de X- en Y-waarden van die twee punten vertoont het scherm:

FUNCTIE:

$$Y = 2 * X + 3$$


GEbruik voor berekening van een punt op de lijn bij gegeven X-waarde toets 1 of bij gegeven Y-waarde toets 2.

Het programma maakt het mogelijk de Y-waarde van een punt op de karakteristiek te berekenen als de X-waarde van dat punt wordt ingevoerd. Hiertoe moet u de toets 1 indrukken waarna de betreffende X-waarde kan worden ingevoerd. Voert u $X=5$ in dan is het antwoord $Y=13$. Na het indrukken van de toets 2 kunt u de Y-waarde van een punt op de karakteristiek invoeren waarna de X-waarde van dat punt wordt getoond. Voert u $Y=15$ in dan is het antwoord $X=6$. Wilt u een aantal opgemeten punten invoeren dan




Afb. 1

dient u na de vraag om het aantal punten het getal van dit aantal in te toetsen. Hierna kunt u de betreffende X- en Y-waarden invoeren waarna de vergelijking van een lijn wordt getoond die ongeveer het gemiddelde van de ingevoerde punten volgt. Er wordt steeds de vergelijking van een rechte lijn verkregen! Afbeelding 1 geeft de karakteristiek van een lijn door de punten P1 en P2. ($Y = X + 2$).



Afb. 2



Afb. 3

b. dit programma maakt het ook mogelijk om berekeningen uit te voeren aan bepaalde kwadratische functies. De vergelijking van zo'n functie is $Y = a * X^2 + b * X + c$. Omdat hier drie constanten zijn: a, b en c moeten in principe drie punten worden ingevoerd. Er zijn daarom bepaalde beperkingen. In afbeelding 2 is de karakteristiek voor $Y = (X^2)/2$ weergegeven. Door de Y-waarden niet lineair af te beelden maar hiervoor een kwadratische schaal te nemen ontstaat voor dezelfde karakteristiek een rechte lijn. Dit is te zien in afbeelding 3. Dit programma brengt de functie dan ook terug tot een rechte lijn en veroorzaakt daarmee bepaalde voorwaarden voor de functie zodat slechts twee punten behoeven te worden ingevoerd. De berekende functie voldoet daarom steeds aan de voorwaarde

$$Y = (a * X + b)^2 = a^2 * X^2 + 2 * a * b * X + b^2.$$

Hiermee liggen de drie constanten van de vierkantsvergelijking vast. Verder blijven alle mogelijkheden die we voor de rechte lijn kennen ook voor deze kwadratische functie gelden. Voor de selectie moet nu echter toets 2 worden gebruikt. Voert u daarna $X = 2$, $Y = 1$ en $X = 3$, $Y = 9$ in dan wordt de functie $Y = 4 * X^2 - 12 * X + 9$ berekend. Deze voldoet aan de voorwaarde $Y = (2 * X - 3)^2$. Nu een voorbeeld van het invoeren van een aantal gemeten punten. Stel dat een voorwerp zonder beginsnelheid een constante versnelling ondervindt. Op verschillende tijdstippen meten we de afgelegde afstand zo nauwkeurig mogelijk. De volgende tabel geeft de gemeten waarden:

tijd						
in sec. (t)	1	2	3	4	5	6
afstand						
in m. (s)	2.01	7.99	17.9	32.1	50.2	71.8

De afgelegde afstand is uit te drukken in $s = 1/2 * a * t^2$. Stellen we $s = Y$ en $t = X$ dan wordt de functie: $Y = 1/2 * a * X^2$ waarvan de karakteristiek in afbeelding 2 is getekend. Hiervoor voeren we de volgende punten in:

X	1	2	3	4	5	6
Y	2.01	7.99	17.9	32.1	50.2	71.8

Het resultaat van de berekening is:

$$Y = 1.99932936 * X^2 + 2.878750E - 03 * X + 1.03624817E - 06$$

Deze vergelijking is een benadering waarbij we de tweede en de derde term moeten laten vervallen (hebben maar kleine waarde). Uit de eerste term stellen we vast dat $1/2 * a$ gelijk moet zijn aan 1.99932936 zodat de versnelling a gelijk is aan 3.99865872 m/(s²).

Ook nu kan door het invoeren van een X- of een Y-waarde de bijbehorende respectievelijke Y- of X-waarde worden gevonden.

c. Ook voor de functie $Y = a * e^{(b * X)}$ is de handeling gelijk aan die van de voorgaande functies. De werkwijze kan worden gekozen door het indrukken van de toets 3. Nu is het invoeren van twee punten weer genoeg voor het berekenen van de vergelijking. Voert u de punten $X = 1$, $Y = 100$ en $X = 4$, $Y = 1$ in dan is het resultaat:

$$Y = 464.158883 * e^{(-1.53505673 * X)}$$

Bij dit voorbeeld is a dus gelijk aan 464.158883 en b gelijk aan -1.53505673. Ook nu kunnen een aantal gemeten punten worden ingevoerd.

```

10 'berekening van functies
20 'voor interrupt met de f1 toets
30 ON KEY GOSUB 1150
40 'voor blauw veld en kader
50 'en witte letters
60 SCREEN 0: WIDTH(40): COLOR 15,4,4
70 PRINT"Wilt u een aanwijzing, ";
80 INPUT"ja of nee";G$
90 IF G$="ja" THEN 1160
100 IF G$(">"nee" THEN 60
110 'voor het printen van de titel
120 GOSUB 1110
130 'keuze van de functie
140 PRINT"Kies voor Y=a*X+b toets 1,"
150 PRINT: PRINT"voor Y=a*X^2+b*X+c";
160 PRINT" toets 2,": PRINT
170 PRINT"en voor Y=e^X toets 3."
180 PRINT: INPUT Z: IF Z>3 THEN 120
190 'invoer van het aantal punten
200 INPUT"Hoeveel punten";I: PRINT
210 DIM X(I), Y(I), A(I)
220 'invoer van de punten
230 FOR N=1 TO I
240 XX$=STR$(N): X$="X"+XX$
250 Y$="Y"+XX$
260 PRINT X$;: INPUT X(N)
270 PRINT Y$;: INPUT Y(N): NEXT N
280 'selectie naar functie
290 ON Z GOTO 330,540,720
300 'voor het berekenen van de
310 'gemiddelde waarde van a en b
320 VOOR DE FUNCTIE Y=A*X+B
330 XS=0: YS=0: XY=0: X2=0
340 FOR N=1 TO I
350 XS=XS+X(N): YS=YS+Y(N)
360 XY=XY+X(N)*Y(N)

```

```

370 X2=X2+X(N)^2: NEXT N: Q=I*X2-XS^2
380 A=(I*XY-XS*YS)/Q
390 B=(YS*X2-XS*XY)/Q
400 'voor selectie naar functie
410 ON Z GOTO 430,560,740
420 'voor het printen van het resultaat
430 GOSUB 880
440 'selectie voor invoeren X- of Y-get.
450 G#=INKEY$: IF G#=CHR$(50) THEN 510
460 IF G#(<)CHR$(49) THEN 450
470 GOSUB 1060: PRINT: INPUT "X";X
480 'print Y als functie van X
490 Y=A*X+B: PRINT "Y=Y"
500 GOSUB 890: GOTO 450
510 GOSUB 1060: PRINT: INPUT "Y";Y
520 'print X als functie van Y
530 X=(Y-B)/A: PRINT "X=X": GOTO 500
540 FOR N=1 TO I: Y(N)=SQR(Y(N))
550 NEXT: GOTO 330
560 C=A^2: D=2*A*B: E=B^2
570 GOSUB 960: GOSUB 890
580 G#=INKEY$: IF G#=CHR$(50) THEN 630
590 IF G#(<)CHR$(49) THEN 580
600 GOSUB 960: PRINT: INPUT "X";X
610 Y=C*X^2+D*X+E: PRINT "Y=Y"
620 GOSUB 890: GOTO 580
630 GOSUB 960: PRINT: INPUT "Y";Y
640 P=D^2-4*C*(E-Y): IF P<0 THEN 680
650 P=SQR(P): X1=(-D+P)/2/C
660 X2=(-D-P)/2/C: PRINT "X1=X1
670 PRINT" X2=X2": GOSUB 890: GOTO 580
680 Q=(4*C*E-D^2)/4/C
690 PRINT"Y moet groter of gelijk";
700 PRINT" zijn aan"Q: FOR N=0 TO 9000
710 NEXT N: GOTO 630
720 FOR N=1 TO I: IF Y(N)=<0 THEN 840
730 Y(N)=LOG(Y(N)): NEXT: GOTO 330
740 D=A: C=EXP(B): GOSUB 1030: GOSUB 890
750 G#=INKEY$: IF G#=CHR$(50) THEN 800
760 IF G#(<)CHR$(49) THEN 750
770 GOSUB 1030: PRINT: INPUT "X";X
780 Y=C*EXP(D*X): PRINT "Y=Y";Y
790 GOSUB 890: GOTO 750
800 GOSUB 1030: PRINT: INPUT "Y";Y
810 IF Y=<0 THEN GOSUB 850: GOTO 800
820 X=(LOG(Y)-B)/D: PRINT "X=X"
830 GOSUB 890: GOTO 750
840 GOSUB 850: GOTO 230
850 PRINT"Y steeds groter dan 0."
860 FOR N=0 TO 3000: NEXT N: RETURN
870 'subr. geeft aanwijzing toetsen
880 GOSUB 1060
890 PRINT: PRINT: PRINT
900 PRINT"Gebruik voor berekening van";
910 PRINT" een punt op de lijn bij ";
920 PRINT"gegeven X-waarde toets 1"
930 PRINT"of bij gegeven Y-waarde ";
940 PRINT"toets 2.": RETURN
950 'subr. print resultaat parabool
960 GOSUB 1110: PRINT"Functie:"
970 D1=D: E1=E
980 D#="+": IF D<0 THEN D#="-": D1=-D
990 E#="+": IF E<0 THEN E#="-": E1=-E
1000 PRINT"Y ="C* X^2 "D#D1"* X "E#E1
1010 RETURN
1020 'subr. print resultaat exp. functie
1030 GOSUB 1110: PRINT"Functie:"
1040 PRINT"Y ="C* e^("D"* X)": RETURN
1050 'subr. print het resultaat rechte

```

```

1060 GOSUB 1110: PRINT"Functie:": PRINT
1070 B1=B: B#="+
1080 IF B<0 THEN B#="-": B1=-B
1090 PRINT"Y ="A* X "B#B1: RETURN
1100 'subr. print de titel
1110 CLS: PRINT TAB(10);
1120 PRINT"Berekening functie."
1130 PRINT:PRINT:PRINT: RETURN
1140 'start het programma opnieuw
1150 KEY(1) OFF: RETURN 60
1160 CLS: PRINT"Als u een aantal punten ";
1170 PRINT"voor een": PRINT"karacteri";
1180 PRINT"stiek hebt opgemeten dient u"
1190 PRINT"eerst het getal van het ";
1200 PRINT"aantal punten inte voeren ";
1210 PRINT"en daarna de X en de Y ";
1220 PRINT"waardenvan elk punt ";
1230 PRINT"afzonderlijk."
1240 PRINT: PRINT"Na het berekenen ";
1250 PRINT"van de vergelijking "
1260 PRINT"door de computer kunt u ";
1270 PRINT"Y=f(X) of X=f(Y)laten ";
1280 PRINT"bepalen.": PRINT
1290 PRINT"Gebruik voor het vervolgen";
1300 PRINT" van het pro- gramma een ";
1310 PRINT"willekeurige toets."
1320 A#=INKEY$: IF A#=""THEN 1320
1330 CLS: GOTO 120

```

4.7. Het berekenen van pythagorische driehoeken

Het volgende programma berekent voor u de pythagorische driehoeken, dat wil zeggen: driehoeken met een rechte hoek waarvan de lengte van alle drie de zijden in een geheel getal zijn uit te drukken.

De bekendste hiervan is wel de 3 4 5 driehoek. Er zijn er echter nog veel meer. Het aantal driehoeken dat wordt berekend, is beperkt. De grenzen kunt u zelf bepalen door de kleinste lengte (ondergrens) en de grootste lengte (bovengrens) van de korte rechthoekszijde in te voeren. Meervouden zullen niet voorkomen. Een driehoek waarvan de zijden zijn uitgedrukt in 12 16 en 20 als meervoud van de 3 4 5 driehoek zal dus niet worden gegeven. Heeft u als ondergrens 3 en als bovengrens 15 ingevoerd dan is het resultaat:

3	4	5
5	12	13
7	24	25
8	15	17
9	40	41
11	60	61
12	35	37
13	84	85
15	112	113

```

10 'pythagorische driehoeken
20 'voor blauw veld en kader
30 'en witte tekens
40 SCREEN 0: WIDTH(40): COLOR 15,4,4
50 PRINT"Wilt u een aanwijzing, ";
60 INPUT"ja of nee";G$
70 IF G$="ja" THEN 280
80 IF G$<>"nee" THEN 40
90 CLS: PRINT TAB(7)"Pythagorische ";
100 PRINT"driehoeken.": PRINT: PRINT
110 'invoer van de grenzen
120 INPUT"Ondergrens";A
130 INPUT"Bovengrens";B: PRINT
140 FOR I=A TO B
150 'voor oneven getallen
160 IF (I AND 1)=1 THEN 250
170 IF I=4 THEN 270
180 'voor het voorkomen van
190 'veelvouden van pyth. driehoeken
200 IF ((I/2) AND 1)=1 THEN 270
210 'voor het berekenen van de tweede
220 'rechthoekszijde en de
230 'schuine zijde
240 J%=(I^2-4)/4: K%=(I^2+4)/4: GOTO 260
250 J%=(I^2-1)/2: K%=(I^2+1)/2
260 PRINT I; J%; K%
270 NEXT I: END
280 CLS
290 PRINT"Voor het berekenen van de";
300 PRINT" driehoeken wordt ";
310 PRINT"uitgegaan van de korte ";
320 PRINT"rechthoeks-zijde. ";
330 PRINT"Hiervan moet de laagst ";
340 PRINT"gewenste en de hoogst";
350 PRINT"gewenste waarde worden "
360 PRINT"ingevoerd.": PRINT
370 PRINT"Gebruik voor het vervolgen";
380 PRINT" van het pro- gramma ";
390 PRINT"een willekeurige toets."
400 G$=INKEY$: IF G$="" THEN 400
410 GOTO 90

```

4.8. Het oplossen van een vierkantsvergelijking

Dit programma berekent voor u de wortels van een vierkantsvergelijking. De volledige vierkantsvergelijking is $a \cdot X^2 + b \cdot X + c = 0$. Andere vormen zijn: $a \cdot X^2 + c = 0$ en $a \cdot X^2 + b \cdot X = 0$. Steeds wordt echter gevraagd om drie gegevens in te voeren: a, b en c. Voor de vergelijking $a \cdot X^2 + c = 0$ moeten we daarom voor b 0 invoeren en voor de vergelijking $a \cdot X^2 + b \cdot X = 0$ moeten we voor c 0 invoeren. Voor de vergelijking $2 \cdot X^2 - 15 \cdot X + 18 = 0$ moeten de volgende waarden worden ingetoetst: a=2, b=-15 en c=18. Het invoeren hiervan kan direct na het starten van het programma gebeuren. Het resultaat volgt direct daarna.

DE VERGELIJKING IS:

$$A \cdot X^2 + B \cdot X + C = 0$$

GETAL VOOR A? 2

GETAL VOOR B? -15

GETAL VOOR C? 18

DE WORTELS ZIJN:

X1 = 6 X2 = 1.5

GEBRUIK VOOR EEN VOLGENDE
OPGAVE DE F1-TOETS

Het kan zijn dat bepaalde vierkantsvergelijkingen uitkomen op een stel complexe wortels. In die gevallen wordt de uitkomst ook in de complexe getallen gegeven. Probeer maar eens: $X^2 - 5 \cdot X + 7 = 0$ (a=1).

Na het oplossen van een vergelijking kunnen de gegevens voor een volgende vergelijking worden ingevoerd als eerst toets f1 is ingedrukt.

```

10 'vierkantsvergelijkingen
20 'voor interrupt met de f1 toets
30 ON KEY GOSUB 420
40 'voor blauw veld en kader
50 'en witte letters
60 SCREEN 0: WIDTH(40): COLOR 15,4,4
70 'het printen van de titel
80 PRINT TAB(9)"Vierkantsvergelijking."
90 PRINT: PRINT: PRINT
100 PRINT"De vergelijking is"
110 PRINT: PRINT"aX^2+bXX+c=0": PRINT
120 'invoeren van de gegevens
130 INPUT"Getal voor a";A: PRINT
140 INPUT"Getal voor b";B: PRINT
150 INPUT"Getal voor c";C: PRINT
160 'berekening van de discriminant D
170 D=B^2-4*A*C
180 'als D=neg dan complexe wortels
190 IF D<0 THEN 340
200 'berekening van de wortels
210 D=SQR(D): X1=(-B+D)/2/A: X2=(-B-D)/2/A
220 'het printen van de wortels
230 PRINT: PRINT
240 PRINT"De wortels zijn:": PRINT
250 PRINT"X1="X1: PRINT: PRINT"X2="X2
260 PRINT: PRINT: PRINT"Gebruik voor ";
270 PRINT"een volgende opgave"
280 PRINT"toets f1"
290 'schakel f1 toets in
300 KEY(1) ON
310 'wacht op interrupt
320 GOTO 320
330 'berekening complexe wortels
340 P=-B/2/A: D=-D: Q=SQR(D)/2/A

```

```

350 'print de complexe wortels
360 PRINT: PRINT
370 PRINT"De wortels zijn:": PRINT
380 PRINT"Z1="P"+ i"Q": PRINT
390 PRINT"Z2="P"- i"Q"(complex)"
400 GOTO 260
410 'start het programma opnieuw
420 KEY(1) OFF: RUN

```

4.9. Omzetting van het polaire naar het rechthoekig coördinatenstelsel

Een berekening die nogal eens nodig is, betreft het omzetten van de coördinaten van een punt in het rechthoekig coördinatenstelsel naar het polaire coördinatenstelsel. Hiervoor dient het volgende programma, waarmee omzetting in beide richtingen mogelijk is. In het rechthoekig coördinatenstelsel wordt de plaats van een punt aangegeven met een X- en een Y-waarde (tweedimensionaal) en in het polaire coördinatenstelsel met de lengte van de lijn R, dit is de rechte afstand van het punt tot aan de oorsprong van de assen en met de hoek die deze lijn maakt met de X-as (afbeelding 4). Na het starten van het programma kunt u kiezen voor de omzetting van rechthoekige coördinaten naar polaire coördinaten (kiezen voor R/P) of voor de omzetting van polaire coördinaten naar rechthoekige coördinaten (kiezen voor P/R). Wilt u omzetten van rechthoekige coördinaten naar polaire coördinaten dan dient u de waarden X en Y in te voeren. Het programma geeft u dan R en de hoek. Heeft u X = 10 en Y = 20 ingevoerd dan is het resultaat:


CO-ORDINATENSTELSEL

WILT U VAN RECHTHOEK CO-ORDINATEN HAAR POLAIRE CO-ORDINATEN (R/P) OF ANDERSOM (R/P)? R/P

X? 10
Y? 20

R = 22.3606798
HOEK = 1.10714872 RAD

GEBRUIK VOOR EEN VOLGENDE OMZETTING DE F1-TOETS



Zoals u ziet wordt de grootte van de hoek uitgedrukt in radialen. Zo moet u voor de omzetting in tegenovergestelde richting, die van polaire naar rechthoekige coördinaten, de grootte van de hoek ook uitdrukken in radialen. Heeft u voor P/R gekozen en wordt door R 25 en voor de hoek 1.04719755 rad. ingevoerd, dan is het resultaat:

CO-ORDINATENSTELSEL

WILT U VAN RECHTHOEK CO-ORDINATEN NAAR POLAIRE CO-ORDINATEN (R/P) OF ANDERSOM (P/R)? P/R

R? 25
HOEK? 1.04719755

X = 12.5 Y = 21.6506351

GEBRUIK VOOR EEN VOLGENDE OMZETTING DE F1-TOETS

```

10 'van rechthoek naar polair vv.
20 'voor interrupt met de f1 toets
30 ON KEY GOSUB 520
40 'voor blauw veld en kader
50 'en witte tekens
60 SCREEN 0: WIDTH(40): COLOR 15,4,4
70 'het printen van de titel
80 PRINT TAB(10)"Co-ordinatenstelsel."
90 PRINT: PRINT: PRINT
100 'keuze van stelsel

```

```

110 PRINT"Wilt u van rechthoekscor";
120 PRINT"dinaten naar polaire co-or";
130 PRINT"dinaten (R/P) of andersom";
140 PRINT" (P/R)";
150 INPUT A$: IF A$="R/P" THEN 320
160 IF A$<>"P/R" THEN 60
170 'invoer van de gegevens
180 PRINT: INPUT"r";R: INPUT"Hoek";H
190 'conversie polair-rechthoek
200 X=R*COS(H): Y=R*SIN(H)
210 'print het resultaat
220 PRINT: PRINT"x="X: PRINT"y=";Y
230 PRINT: PRINT: PRINT
240 PRINT"Gebbruik voor een ";
250 PRINT"volgende omzetting"
260 PRINT"toets f1."
270 'schakel f1 toets in
280 KEY(1) ON
290 'wacht op interrupt
300 GOTO 300
310 'invoer gegevens
320 PRINT: INPUT"x";X: INPUT"y";Y
330 'niet delen door 0
340 IF X=0 THEN 470
350 'berekening hoek
360 H=ATN(Y/X)
370 'correctie betreffende kwadrant
380 IF Y<0 THEN H=H+4*ATN(1)
390 IF SGN(X)=-SGN(Y) THEN H=H+4*ATN(1)
400 IF (Y=0) AND (X<0) THEN H=4*ATN(1)
410 'berekening straal
420 R=SQR(X^2+Y^2)
430 'print het resultaat
440 PRINT: PRINT"r="R
450 PRINT"Hoek="H"rad.": GOTO 230
460 'berekening van de hoek voor x=0
470 IF Y>0 THEN H=2*ATN(1)
480 IF Y<0 THEN H=6*ATN(1)
490 'berekening van de straal voor x=0
500 R=Y: GOTO 440
510 'start het programma opnieuw
520 KEY(1) OFF: RUN

```

4.10. Het omzetten van radialen naar graden

Omdat de computer nu eenmaal de grootte van de hoeken uitdrukt in radialen is een programma voor het omzetten van graden naar radialen wel zo handig. Ook dit programma laat omzetten in twee richtingen toe: van radialen naar graden (kiezen voor R/G) en van graden naar radialen (kiezen voor G/R). Heeft u gekozen voor G/R en heeft u de hoek 60 graden ingevoerd dan is het resultaat:

CONVERSIE GRADEN-RADIALEN

60 GRADEN KOMT OVEREEN MET

1.04719755 RADIALEN

```

10 'omzetten van graden naar radialen
20 'voor interrupt met de f1 toets
30 ON KEY GOSUB 460
40 'voor blauw veld en kader
50 'en witte letters
60 SCREEN 0: WIDTH(40): COLOR 15,4,4
70 'voor het printen van de titel
80 GOSUB 420
90 PI=4*ATN(1)
100 'selectie naar werkwijze
110 PRINT"Wilt u van graden naar ";
120 PRINT"radialen (G/R) of van ";
130 PRINT"radialen naar graden (R/G)";
140 INPUT K$: IF K$="G/R" THEN 330
150 'voor verkeerde invoer
160 IF K$<>"R/G" THEN 60
170 'invoer van gegeven
180 PRINT: PRINT: INPUT"Rad. ";R
190 'berekening radialen-graden
200 G=R/2/PI*360
210 'het printen van het resultaat
220 GOSUB 420: PRINT: PRINT: PRINT
230 PRINT R"rad. komt overeen met"
240 PRINT: PRINT G"graden"
250 PRINT: PRINT: PRINT: PRINT
260 PRINT "Gebbruik voor een volgende ";
270 PRINT "omzetting": PRINT"toets f1."
280 'schakel f1 toets in
290 KEY(1) ON
300 'wacht op interrupt
310 GOTO 310
320 'invoer van gegeven
330 PRINT: PRINT: INPUT"graden";G
340 'berekening graden-radialen
350 R=G/360*2*PI
360 'het printen van het resultaat
370 GOSUB 420: PRINT: PRINT: PRINT
380 PRINT G"graden komt overeen met"
390 PRINT: PRINT R"radialen."
400 GOTO 250
410 'subr. print de titel
420 CLS: PRINT TAB(12);
430 PRINT"Graden-radialen"
440 PRINT:PRINT:PRINT: RETURN
450 'start het programma opnieuw
460 KEY(1) OFF: RETURN 60

```

4.11. Berekening kegel en bol

Dit programma berekent de inhoud en de oppervlakte van een kegel als hiervan de straal van het grondvlak en de hoogte van de kegel worden ingevoerd. Ook kan de oppervlakte en de inhoud van een bol worden berekend. Wilt u de berekening van een kegel dan dient u na het starten van het programma het woord: kegel in te voeren. Daarna wordt u gevraagd om de hoogte van de kegel en de straal van het grondvlak. Heeft u voor de hoogte 120 en voor de straal 50 ingevoerd dan is het resultaat:

BEREKENING LICHAMEN

HOOGTE? 120
STRAAL GRONDVLAK? 50

DE KEGELINHOUD IS 314159.266

HET MANTELOPPERVLAK IS
20420.3523

GEbruik VOOR EEN VOLGENDE BEREKENING DE F1-TOETS

Voor de berekening van een bol moet u na het starten van het programma het woord bol invoeren en daarna de grootte van de straal. De computer geeft dan het resultaat van de berekening van de oppervlakte en de inhoud.

```
10 'berekening bol en kegel
20 'voor interrupt met de f1 toets
30 ON KEY GOSUB 500
40 'voor blauw veld en kader
50 'en witte letters
60 SCREEN 0: WIDTH(40): COLOR 15,4,4
70 'voor het printen van de titel
80 GOSUB 460
90 PI=4*ATN(1)
100 'selectie naar werkwijze
110 PRINT"Kies 'bol' of 'kegel'";
120 INPUT K$: IF K$="bol" THEN 350
130 'voor verkeerde invoer
140 IF K$<>"kegel" THEN 60
150 GOSUB 460
160 'invoer van gegeven
170 PRINT: PRINT: INPUT"Hoogte";H
180 INPUT"Straal grondvlak";R
190 'berekening van de kegelinhoud
200 V=R^2*PI*H/3
210 'berekening van het manteloppervlak
220 A=SQR(R^2+H^2)*R*PI
230 'print het resultaat
240 PRINT: PRINT: PRINT
250 PRINT"De kegelinhoud is"V: PRINT
260 PRINT"Het manteloppervlak is"A
270 PRINT: PRINT: PRINT
280 PRINT"Gebbruik voor een volgende ";
290 PRINT"berekening": PRINT"toets f1."
300 'schakel f1 toets in
310 KEY(1) ON
320 'wacht op een interrupt
330 GOTO 330
340 'bol, invoer van gegeven
350 GOSUB 460: PRINT: PRINT
360 INPUT"Straal";R
370 'berekening balinhoud
```

```
380 V=R^3*PI*4/3
390 'berekening boloppervlak
400 A=R^2*PI*4
410 'print het resultaat
420 PRINT: PRINT: PRINT
430 PRINT"De balinhoud is"V: PRINT
440 PRINT"De oppervlakte is"A
450 GOTO 270
460 CLS: PRINT TAB(8);
470 PRINT"Berekening bol en kegel"
480 PRINT:PRINT:PRINT: RETURN
490 'start het programma opnieuw
500 KEY(1) OFF: RETURN 60
```

4.12. Berekening van een kogelbaan

Een berekening waar we een hele kluit aan zullen hebben, is die van de kogelbaan, dat wil zeggen: het berekenen van de hoek waaronder een projectiel moet worden afgevuurd om een doel op een bepaalde afstand te treffen. De computer heeft er echter geen moeite mee als we hem het volgende programma laten afwerken. Overigens is de berekening die hierbij wordt uitgevoerd niet in overeenstemming met de werkelijkheid omdat we aannemen dat het geheel in het luchtledige plaatsvindt. Er zijn bij het afvuren van projectielen twee mogelijkheden: vlakbaan, waarbij wordt afgevuurd onder een zo klein mogelijke hoek en krombaan, waarbij juist wordt afgevuurd onder een zo groot mogelijke hoek, om ook doelen te kunnen treffen die achter een obstakel verborgen zijn. Voor beide gevallen is het programma geschikt. U kunt kiezen voor vlakbaan door na het starten van het programma VB in te toetsen. Toetst u KB in dan kiest u voor krombaan. Daarna moet u de aanvangssnelheid van het projectiel en de afstand tot het doel invoeren. Kiest u een te lage aanvangssnelheid om de afstand te kunnen overbruggen dan wordt u netjes verteld hoe groot de aanvangssnelheid minstens moet zijn. Kiest u voor vlakbaan en voert u een aanvangssnelheid van 500 m/s en een afstand van 5000 m in dan is het resultaat:

KIES VLAKBAAN (VB) OF
KROMBAAN (KB)
? VB

AANVANGSSNELHEID? 500

AFSTAND? 5000

DE HOEK WAARONDER MOET WORDEN GESCHOTEN IS
.205758422 RAD

DE GROOTSTE HOOGTE IS
521.780376

GEBRUIK VOOR DE VOLGENDE
BEREKENING TOETS F1

Zoals u ziet wordt ook door het programma berekend wat de grootste hoogte is die door het projectiel wordt bereikt. Na het bedienen van de f1-toets kunt u weer een nieuwe berekening laten maken. Als u nu KB kiest en dezelfde gegevens invoert dan wordt de hoek 1.3650379 radialen en de grootste hoogte 11978.2196 meter.

```

10 'kogelbaan
20 'voor interrupt met de f1 toets
30 ON KEY GOSUB 660
40 'voor blauw veld en kader
50 'en witte letters
60 SCREEN 0: WIDTH(40): COLOR 15,4,4
70 PI=4*ATN(1)
80 'voor het printen van de titel
90 GOSUB 620
100 'selectie naar werkwijze
110 PRINT "Kies vlakbaan (VB) of "
120 PRINT "krombaan (KB)";
130 INPUT G$
140 IF G$="KB" THEN A=1: GOTO 170
150 'opnieuw bij verkeerde invoer
160 A=0: IF G$("<"VB)" THEN 60
170 'invoer van de gegevens
180 GOSUB 620
190 INPUT "Aanvangssnelheid";V0
200 PRINT: INPUT "Afstand";S
210 P=9.82*S/V0^2: Q=4-4*P
220 'ga na of V0 te klein is voor S
230 IF P>.4909 THEN 520
240 'selectie VB of KB
250 IF A=1 THEN 470
260 'vlakbaan, berekening van de hoek
270 H=0
280 X=P-COS(H)*SIN(H)
290 IF ABS(X)<1E-07 THEN 320
300 H=H+X*PI/Q: GOTO 280
310 'berekening van de hoogte
320 T=V0*SIN(H)/10: HS!=4.91*T^2: H!=H
330 'het printen van het resultaat
340 PRINT: PRINT: PRINT
350 PRINT "De hoek waaronder moet ";
360 PRINT "worden geschoten"
370 PRINT "is "H!"radialen"
380 PRINT "De grootste hoogte is "HS!
```

```

390 PRINT: PRINT: PRINT
400 PRINT "Gebruik voor een volgende ";
410 PRINT "berekening toets f1."
420 'schakel f1 toets in
430 KEY(1) ON
440 'wacht op de interrupt
450 GOTO 450
460 'krombaan, berekening van de hoek
470 H=PI/2
480 X=P-COS(H)*SIN(H)
490 IF ABS(X)<1E-07 THEN 320
500 H=H-X*PI/Q: GOTO 480
510 'berekening min. V0 bij gegeven S
520 V0=SQR(9.82*S/.4909)
530 PRINT: PRINT: PRINT
540 PRINT "De aanvangssnelheid moet ";
550 PRINT "minstens gelijkzijn aan";
560 PRINT V0
570 PRINT: PRINT: PRINT
580 PRINT "Gebruik de spatiebalk."
590 G$=INKEY$: IF G$("<"CHR$(32)) THEN 590
600 GOTO 180
610 'subr. print de titel
620 CLS: PRINT TAB(15);
630 PRINT "Kogelbaan."
640 PRINT:PRINT:PRINT: RETURN
650 'start het programma opnieuw
660 KEY(1) OFF: RETURN 60
```

4.13. Vallende voorwerpen

Sommige vogels hebben de gewoonte om tijdens hun vlucht iets te laten vallen dat dan precies op iemands kleding of hoofd terecht komt. Het lijkt alsof deze dieren een ingebouwde computer hebben waarmee ze de juiste plaats kunnen berekenen waar ze het vuiltje los moeten laten. Met het volgende programma kan de computer dit ook. Dit programma berekent de afstand tot het doel waar de last moet worden losgelaten als de vlieghoogte en de vliegsnelheid worden ingevoerd. Voeren we als vlieghoogte 80 meter en als vliegsnelheid 5 m/s in dan is het resultaat:

VALLEND VOORWERP

VLIEGHOOGTE? 80

VLIEGSNELHEID? 5

DE AFSTAND IS 20

GEBRUIK VOOR DE VOLGENDE
BEREKENING DE SPATIEBALK

U ziet, de juiste afstand tot het doel is 20 meter. Ook nu kunt u voor een volgende berekening de fl-toets indrukken.

```

10 'vallend voorwerp
20 'voor blauw veld en kader
30 'en witte tekens
40 SCREEN 0: WIDTH(40): COLOR 15,4,4
50 'het printen van de titel
60 PRINT TAB(12)"Vallend voorwerp"
70 'invoeren van de gegevens
80 PRINT: PRINT: PRINT
90 INPUT"Vlieghoogte";H
100 INPUT"Vliegsnelheid";V
110 'bereken de afstand
120 S=V*SQR(H/5)
130 'print het resultaat
140 PRINT: PRINT
150 PRINT"De afstand is"S
160 PRINT: PRINT: PRINT
170 PRINT"Gebbruik voor een volgende";
180 PRINT" berekening"
190 PRINT"de spatiebalk."
200 G#=INKEY#: IF G#(<>CHR$(32)) THEN 200
210 GOTO 40

```

4.14. Statistiek

In een vorig hoofdstuk ben u al een programma tegegekomen dat wat met statistiek te maken heeft, namelijk het programma "functies" in paragraaf 4.6. Door dat programma wordt een lijn berekend naar aanleiding van een aantal ingevoerde punten. Daarbij moet een gemiddelde snelheid worden bepaald van punten die bijvoorbeeld door metingen zijn gevonden.

Ook bij het programma dat u in deze paragraaf wordt geboden wordt het gemiddelde bepaald (onder andere) van een aantal ingevoerde waarden. Deze kunnen bijvoorbeeld het resultaat zijn van een aantal steekproeven (monsters).

Direct na het starten van het programma kunt u achter het woord "monster" op het scherm elke waarde afzonderlijk invoeren. Na elk ingevoerd getal (ook al na het eerste) worden de volgende gegevens getoond:

- Het getal ingevoerde waarden (n).
- De som van de ingevoerde getallen (Σx).
- De gemiddelde waarde van de ingevoerde getallen, berekend met $\Sigma x / n$ en aangegeven met xg.

- De som van de kwadraten van de ingevoerde getallen (Σx^2).
- De standaardafwijking s, berekend met $s = \text{SQR}((\Sigma x^2 - n * xg^2) / (n - 1))$.
- De standaardafwijking o, berekend met $\sigma = \text{SQR}((\Sigma x^2 - n * xg^2) / n)$.

Heeft u achtereenvolgens de getallen 3, 4 en 5 ingevoerd dan is het resultaat na het invoeren van het getal 5:

AANTAL MONSTERS n	3
Σx	12
GEM. WAARDE xg	4
Σx^2	50
s	1
σ	.81649658

```

10 'statistiek
20 'voor blauw veld en kader
30 'en witte tekens
40 SCREEN 0: WIDTH(40): COLOR 15,4,4
50 'het printen van de titel
60 PRINT TAB(14)"Statistiek.": PRINT
70 'invoeren van de gegevens
80 PRINT
90 'wis de regel
100 PRINT STRING$(38,32):PRINT CHR$(30);
110 PRINT"Monster"SPC(9);: INPUT X
120 'bereken het aantal monsters
130 PRINT: N=N+1
140 PRINT"Aantal monsters n"N: PRINT
150 PRINT STRING$(38,32):PRINT CHR$(30);
160 'bereken het totaal v.d. monsters
170 ZX=ZX+X: PRINT CHR$(228)"x";
180 PRINT SPC(15)ZX: PRINT
190 PRINT STRING$(38,32):PRINT CHR$(30);
200 'bereken de gemiddelde waarde
210 XG=ZX/N: PRINT"Gem. waarde xg";
220 PRINT SPC(3)XG: PRINT
230 PRINT STRING$(38,32):PRINT CHR$(30);
240 'bereken de som v.d. kwadraten
250 Z2=Z2+X^2: PRINT CHR$(228)"x^2";
260 PRINT SPC(13)Z2: PRINT
270 PRINT STRING$(38,32):PRINT CHR$(30);
280 'bereken de standaardafwijking
290 D=Z2-N*XG^2
300 'als N=1 dan N-1=0
310 IF N=1 THEN 320 ELSE S=SQR(D/(N-1))
320 S1=SQR(D/N): PRINT"s"SPC(16)S: PRINT
330 PRINT STRING$(38,32):PRINT CHR$(30);
340 PRINT CHR$(229)SPC(16)S1
350 'voor het volgende monster
360 PRINT CHR$(11);: GOTO 60

```

5. Renteberekeningen

5.1. Berekening van het percentage

Waar we allemaal wel eens mee te maken krijgen, is het berekenen van rente bij geldbedragen. Het volgende programma maakt het mogelijk de rente van een bedrag te berekenen als het (netto)bedrag en het percentage van de rente worden ingevoerd. Ook het omgekeerde is mogelijk. Dan moet het netto- en het brutobedrag worden ingevoerd waarna de computer het percentage van de rente berekent. Voor het berekenen van het brutobedrag moet na het starten toets B worden ingedrukt. Daarna kan het nettobedrag en het rentepercentage worden ingevoerd. Wilt u dat het rentepercentage wordt berekend dan moet na het starten van het programma de toets R worden ingedrukt. Voert u daarna voor het nettobedrag 200 en voor het brutobedrag 230 in dan is het resultaat:

RENTEPERCENTAGE

NETTO? 200

BRUTO? 230

RENTE 15%

GEBRUIK VOOR EEN VOLGENDE
BEREKENING DE F1-TOETS.

```
10 'rentepercentage
20 'voor interrupt met de f1 toets
30 ON KEY GOSUB 420
40 'voor blauw veld en kader
50 'en witte tekens
60 SCREEN 0: WIDTH(40): COLOR 15,4,4
70 'voor het printen van de titel
80 GOSUB 380
90 'selectie naar werkwijze
100 PRINT"Wil u het brutobedrag (B) ";
110 PRINT"of het"
120 PRINT"rentepercentage (R)";
```

```
130 INPUT G$: IF G$="R" THEN 310
140 IF G$(">")"B" THEN 60
150 'bruto bedrag, input gegevens
160 GOSUB 380: INPUT"Netto";N
170 PRINT: INPUT"% rente";R: PRINT
180 'berekening bruto bedrag
190 B=(100+R)*N/100
200 'print het resultaat
210 PRINT"Bruto";B
220 PRINT: PRINT
230 PRINT"Gebbruik voor een volgende";
240 PRINT" berekening"
250 PRINT"toets f1."
260 'schakel f1 toets in
270 KEY(1) ON
280 'wacht op een interrupt
290 GOTO 290
300 'renteberek. invoer van gegevens
310 GOSUB 380: INPUT"Netto";N
320 PRINT: INPUT"Bruto";B: PRINT
330 'berekening van de rente
340 R=(B-N)/(N/100)
350 'print het resultaat
360 PRINT"Rente"R"%": GOTO 220
370 'subr. print de titel
380 CLS: PRINT TAB(12);
390 PRINT"Rentepercentage."
400 PRINT: PRINT: RETURN
410 'start het programma opnieuw
420 KEY(1) OFF: RUN
```

5.2. Sparen

Voor iemand die een spaarrekening heeft of er een wil openen, is het in het algemeen niet eenvoudig om uit te rekenen hoe groot het bedrag is dat hij aan het eind van het jaar kan verwachten. Nemen we aan dat steeds op de eerste van elke maand hetzelfde bedrag op de rekening wordt gestort dan moet ook elke maand opnieuw de rente worden berekend omdat het bedrag op de rekening regelmatig groter wordt.

Met het nu volgende programma kunt u het rekenwerk aan de computer overlaten. In werkelijkheid is de berekening heel wat ingewikkelder. Bij de banken wordt namelijk rekening gehouden met de datum waarop de storting binnenkomt en wordt met dagrente gerekend, waarbij rente wordt berekend over het aantal dagen van de betreffende maand waarover het gestorte bedrag rentegevend is. Ook loopt het boekjaar niet van 1 januari tot 31 december, maar bijvoorbeeld van 1 november

tot 31 oktober. Een jaar blijft hoe dan ook twaalf maanden lang. Als we aannemen dat de storting steeds op de eerste van de maand plaats heeft, zal het programma redelijk nauwkeurig de jaarlijkse rente uitrekenen. Het gebruik van het programma is eenvoudig. Na het starten kunt u het maandelijks te sparen bedrag, het rentepercentage en het aantal jaren invoeren dat u verwacht te kunnen sparen. Hierna verschijnt een tabel op uw scherm waarin u kunt aflezen hoe groot jaarlijks het rentebedrag en het totale spaarbedrag is. Ook wordt vermeld hoe groot het kapitaal is dat u in die jaren hebt gestort en hoe groot de rente is die in totaal is uitgekeerd. Overigens: deze rente wordt steeds aan het eind van elk jaar op uw spaartegoed bijgeschreven.

```

10 'sparen
20 'voor interrupt met de f1 toets
30 ON KEY GOSUB 440
40 'voor blauw veld en kader
50 'en witte tekens
60 SCREEN 0: WIDTH(40): COLOR 15,4,4
70 'voor het printen van de titel
80 GOSUB 400: KEY OFF
90 'invoer gegevens
100 PRINT"Hoeveel wilt u per maand ";
110 PRINT"sparen";: INPUT SP: PRINT
120 PRINT"Hoe groot is de rente";:
130 INPUT R: PRINT
140 PRINT"Hoeveel jaar wilt u sparen";
150 INPUT J: GOSUB 400
160 'print de kop van de tabel
170 PRINT"Jaar"TAB(9)"Rente"TAB(26);
180 PRINT"Bedrag": PRINT STRING$(36,195)
190 'berekening rente
200 FOR X=1 TO J: JR=0: FOR Y=1 TO 12
210 JR=JR+INT(B*R/12)/100: B=B+SP
220 T=T+SP: NEXT Y: B=B+JR
230 'print de bedragen
240 PRINT X TAB(8) JR TAB(25) B
250 'voor het volgende jaar
260 NEXT X
270 'bereken de totale rente
280 W=INT(100*(B-T))/100
290 'print de totalen
300 PRINT: PRINT
310 PRINT"Totaal is" T "ingelegd."
320 PRINT: PRINT"De totale rente is" W
330 PRINT: PRINT
340 PRINT"Gebbruik toets f1."
350 'schakel toets f1 in
360 KEY(1) ON
370 'wacht op interrupt
380 GOTO 380
390 'subr. print de titel
400 CLS: PRINT TAB(16);
410 PRINT"Sparen."
420 PRINT: PRINT: PRINT: RETURN
430 'start het programma opnieuw
440 KEY(1) OFF: RUN

```

U kunt het programma opnieuw starten door toets f1 te gebruiken.

5.3. De persoonlijke lening

Bij de leningen is een persoonlijke lening zeer populair. Bij deze lening krijgt u het bedrag dat u leent geheel in handen (zonder aftrek van kosten en rente) terwijl u eventueel de reeds afgeloste bedragen opnieuw kunt opnemen tot het maximale overeengekomen bedrag.

De verschuldigde rente wordt maandelijks berekend, rekening houdend met de dag van storting (dagrente), en ook maandelijks bij het nog verschuldigde bedrag opgeteld. Het kan zijn dat de grootte van het af te lossen bedrag afhankelijk is van het nog verschuldigde bedrag en daardoor elke maand kleiner wordt. Om zo snel mogelijk van een lening af te komen zal over het algemeen maandelijks een gelijk bedrag worden afgelost en dat hebben we in dit programma verwerkt.

Ook wordt ervan uitgegaan dat steeds op de eerste van de maand de aflossing wordt overgeschreven. Het is belangrijk om vooraf te weten hoeveel maanden er moet worden afgelost en hoe groot de totale kosten van de lening zijn. Op deze vragen geeft dit programma het antwoord.

Veronderstel dat de grootte van de lening f 5000 is en dat per maand f 250 wordt afgelost bij een rente van 1.1% per maand, dan is het resultaat:

PERSOONLIJKE LENING

HOE GROOT IS HET TE LENEN
BEDRAG? 5000

HOE GROOT IS DE RELATIEVE
RENTE? 1.1

HOEVEEL LOST U PER MAAND
AF? 250

DE LAATSTE AFLOSSING IS 178

DE TOTALE KOSTEN ZIJN 678

HET AANTAL MAANDEN IS 23

GEBRUIK VOOR EEN VOLGENDE BEREKENING DE TOETS F1

U ziet het, geld lenen is duurder dan u denkt!

```
10 'lenen
20 'voor interrupt met de f1 toets
30 ON KEY GOSUB 550
40 'voor blauw veld en kader
50 'en witte tekens
60 SCREEN 0: WIDTH(40): COLOR 15,4,4
70 'voor het printen van de titel
80 GOSUB 510: KEY OFF
90 'invoer gegevens
100 PRINT"Hoe groot is het te lenen ";
110 PRINT"bedrag"; INPUT B: PRINT: L=B
120 PRINT"Hoe groot is de relatieve ";
130 PRINT"rente"; INPUT R: PRINT
140 PRINT"Hoeveel lost u per maand af";
150 INPUT A: PRINT: PRINT: PRINT
160 'maandelijkse berekening
170 GOSUB 480
180 'de berekening gaat door tot het
190 'nog af te lossen bedrag kleiner
200 'is dan de aflossing per maand
210 IF L>A THEN 170
220 'over een bedrag kleiner dan
230 '25 wordt geen rente berekend
```

```
240 IF L<25 THEN X=X+1: GOTO 340
250 'voor het berekenen van de rente
260 'over het laatste bedrag
270 GOSUB 480
280 'Bepaal of het nog af te lossen
290 'bedrag kleiner is dan de
300 'maandelijkse aflossing
310 IF SGN(L)=1 THEN 240
320 L=L+A
330 'print het resultaat
340 PRINT"De laatste aflossing is"L
350 PRINT: PRINT"De totale kosten zijn";
360 PRINT ((X-1)*A+L)-B: PRINT
370 PRINT"Het aantal maanden is"X
380 PRINT: PRINT: PRINT
390 PRINT"Gebruik voor een volgende";
400 PRINT" berekening"
410 PRINT"toets f1."
420 'schakel f1 toets in
430 KEY(1) ON
440 'wacht op een interrupt
450 GOTO 450
460 'subr. berekent per maand het nog
470 'af te lossen bedrag L
480 L=INT(L*(100+R))/100-A
490 X=X+1: RETURN
500 'subr. print de titel
510 CLS: PRINT TAB(10);
520 PRINT"Persoonlijke lening."
530 PRINT: PRINT: PRINT: RETURN
540 'start het programma opnieuw
550 KEY(1) OFF: RUN
```

6. Complexe getallen

6.1. Complex rekenen

Het volgende programma maakt het mogelijk om via de computer berekeningen te maken met complexe getallen. De vier basisbewerkingen die we bij elke calculator tegenkomen zijn ook bij dit programma mogelijk: optellen, aftrekken, vermenigvuldigen en delen. Op gelijke wijze als met een rekenmachine is het mogelijk om meerdere bewerkingen na elkaar uit te voeren. Na invoeren en starten van het programma voeren we het reële en het imaginaire deel van het eerste getal in. Daarna volgt het bewerkingsteken en het reële deel en het imaginaire deel van het tweede getal. De computer voert nu de bewerking uit en toont het resultaat.

Nu kan het bewerkingsteken worden ingevoerd van de volgende bewerking met een nieuw getal. Dit gaat door totdat na de laatste bewerking, die nodig is, in plaats van een bewerkingsteken een = teken is ingevoerd. De computer toont dan het definitieve resultaat. Door op de f1-toets te drukken krijgen we de kans om een nieuwe opgave te berekenen. Het resultaat van de berekening

$$(-30 + i41)/(2 + i5)$$

is als volgt:

COMPLEXE GETALLEN

$$(-30 + i41) / (2 + i5) =$$

$$(5 + i8)$$

REEEL DEEL?

Het gebruik van dit programma komt geheel overeen met dat van "Bewerkingen met breuken".

```
10 'bewerkingen met breuken
20 'voor interrupt met de f1 toets
30 ON KEY GOSUB 860
40 'voor blauw veld en kader
50 'en witte tekens
60 SCREEN 0: WIDTH(40): COLOR 15,4,4
70 PRINT"Wilt u een aanwijzing,";
80 PRINT" ja of nee";: INPUT A$
90 IF A$="ja" THEN 640
100 IF A$(">")"nee" THEN 60
110 'voor het printen van de titel
120 S$="+-*/=": GOSUB 610
130 'invoer eerste getal
140 INPUT"Reeel deel";RR
150 INPUT"Imaginaire deel";IR: PRINT
160 'invoer bewerkingsteken
170 INPUT"Bewerking";B$: PRINT
180 FOR N=1 TO 5
190 IF MID$(S$,N,1)=B$ THEN 210
200 NEXT N: GOTO 170
210 IF B$="=" THEN GOSUB 550: GOTO 470
220 'invoer tweede getal
230 INPUT"Reeel deel";R
240 INPUT"Imaginaire deel";I
250 'selectie naar bewerking
260 ON N GOTO 280,380,400,430
270 'optelling
280 RT=RR+R: IT=IR+I
290 'print het resultaat
300 GOSUB 610: P=RR: Q=IR: GOSUB 550
310 PRINT B$;: P=R: Q=I
320 GOSUB 550: PRINT"="
330 PRINT: PRINT: P=RT: Q=IT: GOSUB 550
340 RR=RT: IR=IT: PRINT: PRINT
350 'voor de volgende bewerking
360 GOTO 170
370 'aftrekking
380 RT=RR-R: IT=IR-I: GOTO 300
390 'vermenigvuldiging
400 A=RR: B=IR: C=R: D=I: GOSUB 590
410 GOTO 300
420 'deling
430 A=RR: B=IR: C=R: D=-I: GOSUB 590
440 T=C^2+D^2: RT=RT/T: IT=IT/T
450 GOTO 300
460 'einde van de opgave
470 PRINT: PRINT: PRINT
480 PRINT"Gebbruik voor de volgende ";
490 PRINT"opgave de": PRINT"f1 toets."
500 'schakel toets f1 in
510 KEY(1) ON
520 'wacht op een interrupt
530 GOTO 530
540 'subr. print een complexgetal
550 PRINT" ("P;: R$="+
560 IF Q<0 THEN R$="-": Q=-Q
570 PRINT R$ " i"Q") ";: RETURN
580 'subr. berekent het product
590 RT=A*C-B*D: IT=A*D+B*C: RETURN
600 'subr. print de titel
610 CLS: PRINT TAB(12);
```

```

620 PRINT"Complex rekenen.";
630 PRINT: PRINT: PRINT: RETURN
640 CLS: PRINT"U voert eerst het re-ele ";
650 PRINT"en het imagi-"
660 PRINT"naire deel van het ";
670 PRINT"eerste getal in en daarna ";
680 PRINT"het bewerkingsteken +,-, ";
690 PRINT"* of /. Nu voert u ";
700 PRINT"het re-ele en het imaginaire";
710 PRINT"deel van het tweede getal ";
720 PRINT"in. De bewerking wordt nu ";
730 PRINT"uitgevoerd. Dan kunt u ";
740 PRINT"opnieuw een bewerkingsteken";
750 PRINT" invoeren, een getal enz. "
760 PRINT"Voert u inplaats van een ";
770 PRINT"bewerkingstekenen = ";
780 PRINT"teken in dan krijgt u ";
790 PRINT"het definitieve antwoord."
800 PRINT: PRINT"Gebruik voor het ";
810 PRINT"voortzetten van het "
820 PRINT"programma de spatiebalk."
830 A$=INKEY$: IF A$(<>CHR$(32)) THEN 830
840 GOTO 120
850 'start het programma opnieuw
860 KEY(1) OFF: RETURN 120

```

6.2. Notatie complexe getallen

Er zijn twee methoden om een complex getal te noteren. Ten eerste door middel van een reëel deel plus een imaginair deel ($A + iB$) en ten tweede door een modulus en het argument; in het volgende programma aangegeven met $R(H)$. Start u het programma dan kunt u om te beginnen kiezen in welke richting u wilt omzetten, van $A + iB$ naar $R(H)$ of van $R(H)$ naar $A + iB$. U kunt daarna het getal in de juiste notatie invoeren waarna de computer het getal weergeeft in de vorm die door u is gekozen. Heeft u gekozen voor de omzetting van $A + iB$ naar $R(H)$ en voert u het getal $8 + i15$ in dan is het resultaat:

COMPLEXE GETALLEN

WILT U VAN $A + iB$ NAAR $R(H)$ (R/P)
OF ANDEROM (P/R)
? R/P

REEEL DEEL? 8
IMAGINAIR DEEL? 15

MODULUS = 17
ARGUMENT = 1.080839

GEbruik voor de volgende omzetting de F1-toets.

Het argument is hier uitgedrukt in radialen.

```

10 'notatie complexe getallen
20 'voor interrupt met de f1 toets
30 ON KEY GOSUB 520
40 'voor blauw veld en kader
50 'en witte tekens
60 SCREEN 0: WIDTH(40): COLOR 15,4,4
70 'het printen van de titel
80 PRINT TAB(11)"Complexe getallen."
90 PRINT: PRINT: PRINT
100 'keuze van stelsel
110 PRINT"Wilt u van a+ib naar r(h) ";
120 PRINT"(R/P) of andersom (P/R)";
150 INPUT A$: IF A$="R/P" THEN 320
160 IF A$("<>P/R" THEN 60
170 'invoer van de gegevens
180 PRINT: INPUT"Modulus";R
185 INPUT"Argument";H
190 'conversie r(h) naar a+ib
200 X=R*COS(H): Y=R*SIN(H)
210 'print het resultaat
220 PRINT: PRINT"Reeel deel="X
225 PRINT"Imaginair deel="Y
230 PRINT: PRINT
240 PRINT"Gebruik voor een ";
250 PRINT"volgende omzetting"
260 PRINT"toets f1."
270 'schakel f1 toets in
280 KEY(1) ON
290 'wacht op interrupt
300 GOTO 300
310 'invoer gegevens
320 PRINT: INPUT"Reeel deel";X
325 INPUT"Imaginair deel";Y
330 'niet delen door 0
340 IF X=0 THEN 470
350 'berekening argument
360 H=ATN(Y/X)
370 'correctie betreffende kwadrant
380 IF Y<0 THEN H=H+4*ATN(1)
390 IF SGN(X)=-SGN(Y) THEN H=H+4*ATN(1)
400 IF (Y=0) AND (X<0) THEN H=4*ATN(1)
410 'berekening modulus
420 R=SQR(X^2+Y^2)
430 'print het resultaat
440 PRINT: PRINT"Modulus="R
450 PRINT"Argument="H"rad.": GOTO 230
460 'berekening argument voor x=0
470 IF Y>0 THEN H=2*ATN(1)
480 IF Y<0 THEN H=6*ATN(1)
490 'berekening modulus voor x=0
500 R=Y: GOTO 440
510 'start het programma opnieuw
520 KEY(1) OFF: RUN

```


7. Datum

7.1. Eeuwigdurende kalender

De indeling van de dagen, jaren en eeuwen zoals we die nu kennen, geschiedt volgens de Gregoriaanse kalender. Deze is zo genoemd omdat hij is ingesteld door paus Gregorius XIII op 15 oktober 1582, als verbetering van de Juliaanse kalender die daarvoor in gebruik was. Het is dus geen wonder dat de geboorte van Christus als begin van onze tijdrekening is genomen. Het is niet de enige tijdrekening waarvan het begin samenhangt met een godsdienstige gebeurtenis. Dat is ook het geval met de Islamitische tijdrekening, de Hegira of Hedsjera. Het begingpunt hiervan valt samen met Mohammeds vlucht. De Joodse tijdrekening valt samen met wat in Joods orthodoxe kringen wordt gezien als het tijdstip van de schepping. We houden ons maar bij de tijdrekening die hier in Nederland officieel in gebruik is. Lastig voor een eeuwigdurende kalender is de onregelmatige verdeling van de 365 dagen, die ons jaar in het algemeen kent, over de twaalf maanden. Dan hebben we nog de schrikkeljaren van 366 dagen waarin februari een extra dag krijgt. Ook de schrikkeljaren komen niet regelmatig voor. De regel is dat elk jaar waarvan het jaartal deelbaar is door 4 een schrikkeljaar is, met uitzondering van de eeuwjaar die niet deelbaar zijn door 400, zoals 1700, 1800 en 1900. Als we een jaar van 365 dagen in weken verdelen dan blijft er 1 dag over ($365/7 = 52$ rest 1). Als 1 januari van een bepaald jaar op zondag valt dan valt 1 januari van het jaar daarop op maandag. Daarom moeten we voor elk jaar 1 dag corrigeren. Voor een eeuw zijn dat dus 100 dagen. Verdelen we deze 100 dagen in weken dan zien we dat we voor elke eeuw 2 dagen moeten corrigeren en voor elk schrikkeljaar 1 dag extra.

Omdat in een schrikkeljaar de extra dag steeds aan de maand februari wordt toegevoegd, rekent het programma met jaren die lopen van 1 maart tot en met 28 (29) februari. Al de hiervoor genoemde regels zijn uiteraard in het programma verwezenlijkt. Na het starten van het programma kunnen we de gewenste datum invoeren. De computer vertelt dan

op welke dag die datum valt. Heeft u als datum 24-1-1984 ingevoerd dan is het resultaat:

EEUWIGDURENDE KALENDER

24 - 1 - 1984 VALT OP DINSDAG

GEBRUIK VOOR EEN VOLGENDE DATUM DE F1-TOETS.

U kunt elke datum na 1582 invoeren, in de volgorde dag - maand - jaar. Het jaartal moet steeds met vier cijfers worden ingetoetst, dus 1984, niet 84.

```
10 'eeuwigdurende kalender
20 'voor interrupt met de f1 toets
30 ON KEY GOSUB 480
40 'voor rood veld en kader
50 'en wite letters
60 SCREEN 0: WIDTH(40): COLOR 15,6
70 'voor het printen van de titel
80 GOSUB 440
90 FOR N=0 TO 6: READ A$(N): NEXT N
100 'invoer van de gegevens
110 INPUT"Dag";D: INPUT"Maand";M
120 INPUT"Jaar";J
130 'voor 1583 geen Greg. kalender
140 IF J<1583 THEN 400
150 MD=M-2: JR=J
160 'jaarindeling van 1 mrt-28 febr.
170 IF M<3 THEN MD=M+10: JR=J-1
180 'afschieden van de eeuwen
190 E=INT(JR/100): T=JR-E*100
200 'correctie voor de maanden (28-31)
210 A=INT(MD*2.59)
220 'corr. voor de schrikkeljaren
230 B=INT(JR/4)+INT(E/4)
240 'berekening v.h. nummer v.d. dag
250 X=A+B+E+T+D
260 Y=X-INT(X/7)*7
270 'het printen van het resultaat
280 GOSUB 440
290 PRINT D"-M"-J" valt op "A$(Y)
300 PRINT: PRINT: PRINT
310 PRINT"Gebbruik voor een volgende";
320 PRINT" datum de": PRINT"f1 toets."
```

```

330 'schakel toets f1 in
340 KEY(1) ON
350 'wacht op een interrupt
360 GOTO 360
370 DATA zondag,maandag,dinsdag
380 DATA woensdag,donderdag
390 DATA vrijdag,zaterdag
400 PRINT: PRINT: PRINT
410 PRINT"Geen data voor 1583."
420 FOR N=0 TO 5000: NEXT N: RUN
430 'subr. print de titel
440 CLS: PRINT TAB(8);
450 PRINT"Eeuwigdurende kalender."
460 PRINT: PRINT: PRINT: RETURN
470 'start het programma opnieuw
480 KEY(1) OFF: RUN

```

7.2. De kalender

Het volgende programma laat u de kalender zien van de maand die u wenst. U hoeft daarvoor na het invoeren van het programma en het starten daarvan slechts de maand en het betreffende jaar in te voeren.

Wilt u de kalender van de maand januari 2000, dan voert u voor de maand 1 en voor het jaar 2000 in. U moet dus het nummer van de maand invoeren. Het resultaat hiervan is dan:

KALENDER

zondag	2	9	16	23	30
maandag	3	10	17	24	31
dinsdag	4	11	18	25	
woensdag	5	12	19	26	
donderdag	6	13	20	27	
vrijdag	7	14	21	28	
zaterdag	1	8	15	22	29

U kunt geen datum kiezen vroeger dan 1583. Wilt u een nieuwe datum dan hoeft u slechts toets f1 in te drukken. De kalender verdwijnt dan van het scherm en u kunt gelijk een nieuwe datum invoeren voor een volgende kalender.

```

10 'kalender
20 'voor interrupt met de f1 toets
30 ON KEY GOSUB 860
40 'voor rood veld en kader
50 'en wite letters
60 SCREEN 0: WIDTH(40): COLOR 15,6
70 'voor het printen van de titel
80 GOSUB 820
90 KEY OFF
100 'invoer van de gegevens
110 'het maandnummer invoeren
120 INPUT"Maand";M
130 INPUT"Jaar";J
140 'voor 1583 geen Greg. kalender
150 IF J<1583 THEN 780
160 MD=M-2: JR=J
170 'jaarindeling van 1 mrt-28 febr.
180 IF M<3 THEN MD=M+10: JR=J-1
190 'afschieden van de eeuwen
200 E=INT(JR/100): T=JR-E*100
210 'correctie voor de maanden (28-31)
220 A=INT(MD*2.59)
230 'corr. voor de schrikkeljaren
240 B=INT(JR/4)+INT(E/4)
250 'bepaal op welke dag
260 'de eerste van de maand valt
270 X=A+B+E+T+1
280 Y=X-INT(X/7)*7
290 'bepaal het aant. dagen in febr.
300 IF M=2 THEN D=28: GOTO 720
310 'bepaal het aantal dagen per maand
320 D=28+INT((MD+1)*2.59)-A
330 'het printen van de kalender
340 GOSUB 820
350 'voor het printen v.d. maand
360 IF M>6 THEN M=M-6: GOTO 380
370 ON M GOTO 580,590,600,610,620,630
380 ON M GOTO 640,650,660,670,680,690
390 'print maand en jaar
400 PRINT TAB(8)M$ " "J
410 'print de dagen v.d. week
420 LOCATE 0,7
430 FOR N=0 TO 6: READ A$: PRINT A$
440 PRINT: NEXT N
450 'print de datum, voor 6 weken
460 R=13: FOR H=0 TO 5
470 'voor de datum per week
480 FOR N=Y TO 6
490 P=P+1: IF P>9 THEN R=12
500 IF P=D+1 THEN 540
510 LOCATE R+4*H,7+2*N
520 PRINT P: NEXT N
530 Y=0: NEXT H
540 'schakel de f1 toets in
550 KEY(1) ON
560 'wacht op een interrupt
570 GOTO 570
580 M$="Januari": GOTO 400
590 M$="Februari": GOTO 400
600 M$="Maart": GOTO 400
610 M$="April": GOTO 400
620 M$="Mei": GOTO 400
630 M$="Juni": GOTO 400
640 M$="Juli": GOTO 400
650 M$="Augustus": GOTO 400
660 M$="September": GOTO 400
670 M$="October": GOTO 400
680 M$="November": GOTO 400
690 M$="December": GOTO 400

```

```

700 'subr. bepaalt het aantal dagen
710 'in februari
720 JA=J: IF T+1=100 THEN JA=E+1
730 IF ABS(INT(JA/4)*4-JA)<.1 THEN D=29
740 GOTO 340
750 DATA zondag,maandag,dinsdag
760 DATA woensdag,donderdag
770 DATA vrijdag,zaterdag
780 PRINT: PRINT: PRINT
790 PRINT"Geen data voor 1583."
800 FOR N=0 TO 5000: NEXT N: RUN
810 'subr. print de titel
820 CLS: PRINT TAB(15);
830 PRINT"kalender."
840 PRINT:PRINT: PRINT: RETURN
850 'start het programma opnieuw
860 KEY(1) OFF: RUN

```

7.3. Feestdagen

Geen kalender is compleet zonder de aanduiding van de Christelijke feestdagen. Nu weten we allemaal wel dat de kerstdagen steeds op 25 en 26 december vallen. Anders is dat met Pasen en Pinksteren. Die wisselen per jaar en het enige wat we zeker weten is dat ze niet op één dag vallen. Deze feestdagen zijn namelijk afhankelijk van de maancyclus en deze loopt nu eenmaal niet synchroon met onze jaartelling.

Met het volgende programma kunt u voor elk jaar na 1982 bepalen op welke data deze dagen vallen. Na het starten van het programma hoeft u slechts het jaartal in te voeren van het jaar waarvan u de data van de feestdagen wilt weten en achtereenvolgens verschijnen op uw scherm de data van:

Carnaval,
 Aswoensdag,
 Palmzondag,
 Goede Vrijdag,
 Eerste Paasdag,
 Hemelvaartsdag en
 Eerste Pinksterdag.

```

10 'berekening feestdagen
20 'voor interrupt met de f1 toets
30 ON KEY GOSUB 760
40 KEY OFF
50 'voor rood veld en kader
60 'en witte tekens
70 SCREEN 0: WIDTH(40): COLOR 15,6
80 'print de titel
90 PRINT TAB(14)"Feestdagen."
100 PRINT: PRINT: PRINT
110 'invoer van het jaartal
120 INPUT"Jaar";J:PRINT
130 'geen jaartal kleiner dan 1583
140 IF J>1582 THEN 190

```

```

150 PRINT: PRINT: PRINT
160 PRINT"Geen datum voor 1583."
170 FOR N=1 TO 5000: NEXT: GOTO 70
180 'berekening kerkelijke maancyclus
190 DT=J: DL=19: GOSUB 560
200 A%=R%: DL=100: GOSUB 560: B%=Q%
210 C%=R%: DT=B%: DL=4: GOSUB 560
220 E%=Q%: F%=R%: DT=DT+8: DL=25
230 GOSUB 560: G%=Q%: DT=DT-G%-7
240 DL=3: GOSUB 560: H%=Q%
250 DT=19*A%+B%-E%-H%+15
260 DL=30: GOSUB 560: I%=R%: DT=C%
270 DL=4: GOSUB 560: K%=Q%: L%=R%
280 DT=2*(F%+K%)-I%-L%+32: DL=7
290 GOSUB 560: N%=R%: DL=451
300 DT=A%+11*I%+22*N%: GOSUB 560
310 P%=Q%: D1=I%+N%-7*P%+3: DL=31
320 'correctie voor schrikkeljaren
330 J%=J: GOSUB 590: JA=J/100
340 J%=JA: IF B=3 THEN 370
350 IF ABS(JA-J%)<.01 THEN GOSUB 590
360 'print het resultaat
370 A$="Begin carnaval": GOSUB 650
380 D1=D1+3: A$="Aswoensdag": GOSUB 650
390 D1=D1+39: A$="Palmzondag"
400 GOSUB 650: D1=D1+5
410 A$="Goede vrijdag": GOSUB 650
420 D1=D1+2: A$="Eerste paasdag"
430 GOSUB 650
440 D1=D1+39: A$="Hemelvaart"
450 GOSUB 650: D1=D1+10
460 A$="Eerste pinksterdag"
470 GOSUB 650: PRINT: PRINT
480 PRINT"Gebruik voor een volgend ";
490 PRINT"jaar toets f1."
500 'schakel toets f1 in
510 KEY(1) ON
520 'wacht op interrupt
530 GOTO 530
540 'subr. berekent quotient Q%
550 'en rest R%
560 Q%=DT/DL: R%= DT MOD DL: RETURN
570 'subr. berekent correctie voor
580 'februari in schrikkeljaren
590 B=2: IF (J%/4-INT(J%/4))>.1 THEN B=3
600 RETURN
610 'subroutine berekent en print
620 'de datum
630 'correctie voor het aantal
640 'dagen per maand
650 DT=D1: IF D1<31 THEN DT=D1-B
660 IF D1>91 THEN DT=D1+1
670 GOSUB 560: M=Q%: D=R%+1
680 IF M=0 THEN M$="februari."
690 IF M=1 THEN M$="maart."
700 IF M=2 THEN M$="april."
710 IF M=3 THEN M$="mei."
720 IF M=4 THEN M$="juni."
730 PRINT A$ " valt op"D1;M$: PRINT
740 RETURN
750 'start het programma opnieuw
760 KEY(1) OFF: RUN

```

7.4. Aantal dagen

Het volgende programma rekent voor u uit hoeveel dagen er liggen tussen twee data. Het enige wat u

hebt te doen, is het invoeren van de data. Na het starten van het programma kunt u de eerste datum invoeren in de volgorde dag, maand (maandnummer) en jaar. Daarna kunt u op gelijke wijze de tweede datum invoeren. Direct daarop geeft de computer het aantal dagen tussen de ingevoerde data. Wilt u het aantal dagen weten tussen 1 november 1983 en 15 augustus 1984 dan is na het intoetsen van 1 11 1983 en 15 8 1984 het resultaat:

AANTAL DAGEN

DATUM 1
DAG? 1
MAAND? 11
JAAR? 1983

DATUM 2
DAG? 15
MAAND? 8
JAAR? 1984

HET AANTAL DAGEN TUSSEN DE
DATA IS 288

GEBRUIK VOOR EEN VOLGENDE
BEREKENING DE F1-TOETS.

U kunt de data willekeurig invoeren, dat wil zeggen dat de vroegste datum niet beslist voorop moet

staan. Het getal dat wordt berekend geeft steeds het verschil aan en is dus positief.

```

10 'berekening aantal dagen
20 'voor interrupt met de f1 toets
30 ON KEY GOSUB 500
40 KEY OFF
50 'voor rood veld en kader
60 'en witte tekens
70 SCREEN 0: WIDTH(40): COLOR 15,6
80 'print de titel
90 PRINT TAB(13)"Aantal dagen."
100 PRINT: PRINT: PRINT
110 'invoer van de gegevens
120 PRINT"Datum 1": GOSUB 290: X1=X
130 PRINT: PRINT"Datum 2":GOSUB 290
140 X2=X: PRINT: PRINT: PRINT
150 PRINT"Het aantal dagen tussen ";
160 PRINT"de data is": X=X2-X1
170 IF X<0 THEN X=-X
180 PRINT X: PRINT: PRINT: PRINT
190 PRINT"Gebbruik voor een ";
200 PRINT"volgende berekening"
210 PRINT"toets f1."
220 'schakel de f1 toets in
230 KEY(1) ON
240 'wacht op een interrupt
250 GOTO 250
260 'subr. routine berekent aantal
270 'dagen vanaf denkbeeldig jaar 0
280 'invoer gegevens van 1 datum
290 INPUT"Dag";D: INPUT"Maand";M
300 INPUT"Jaar";J
310 'geen datum voor 1583
320 IF J>1582 THEN 380
330 PRINT: PRINT: PRINT
340 PRINT"Geen datum voor 1583."
350 FOR N=0 TO 5000: NEXT N
360 RUN
370 'jaar van 1 maart tot 28 febr.
380 MD=M-2: JR=J
390 IF M<3 THEN MD=M+10: JR=J-1
400 'berekening van het aantal
410 'dagen in het betreffende jaar
420 A=MD*28+INT(MD*2.59)-2
430 'correctie voor schrikkeljaar
440 E=INT(JR/100)
450 B=INT(JR/4)+INT(E/4)-E
460 'x is ant. dagen vanaf jr. 0
470 X=JR*365+A+B+D
480 RETURN
490 'start het programma opnieuw
500 KEY(1) OFF: RUN

```

8. Spelletjes

8.1. Het meten van de reactiesnelheid

Met het volgende programma kunt u uw reactiesnelheid testen. Na het starten van het programma kunt u eventueel een aanwijzing krijgen, maar als u het volgende goed doorleest, heeft u die aanwijzing niet nodig en kunt u nee invoeren. Direct daarna begint het spel en kunt u het wit oplichten van het veld op uw scherm verwachten. Dit doet zich voor als een korte flits en daarop moet u zo snel mogelijk reageren door op de spatiebalk te drukken (denk er om, dit kan ontaarden in het slaan op de toets en deze is nu eenmaal geen aambeel; hij kan kapot!). Nu is in uw computer een teller ingebouwd die precies elke 1/50 seconde wordt opgehoogd. Met deze klok gaan we de tijd meten die verlopen is tussen het begin van de flits en uw reactie met de toets. Eerst wordt de stand van de teller gelezen op het moment dat de flits verschijnt en daarna op het moment dat de toets wordt ingedrukt. Het verschil in standen van de teller wordt gedeeld door 50 zodat de reactietijd in seconden (en delen van seconden) bekend is. De nauwkeurigheid kan uiteraard niet groter zijn dan 1/50 seconde.

Nadat u de spatiebalk hebt ingedrukt, volgt een nieuwe flits en kunt u opnieuw reageren enzovoorts, totdat na tien flitsen de reactietijden op het scherm worden getoond. De flitsen komen overigens op willekeurige tijden na het indrukken van de spatiebalk, de ene keer vrij snel, een andere keer duurt het lang. U hoeft nog niet eens zo snel te reageren om de volgende resultaten te bereiken.

REACTIETIJD

REACTIESNELHEID IN SEC.:

FLITS 1: „216666667
FLITS 2: „266666667
FLITS 3: „166666667
FLITS 4: „236666667

FLITS 5: „236666667
FLITS 6: „266666667
FLITS 7: „2
FLITS 8: „216666667
FLITS 9: „183333333
FLITS 10: „183333333

GEBRUIK VOOR EEN VOLGENDE SERIE DE F1-TOETS.


U ziet, voor een volgende serie moet u de f1-toets indrukken. Direct daarna begint het spel van voren af aan.

```
10 'reactiesnelheid
20 'voor interrupt met de f1 toets
30 ON KEY GOSUB 560
40 'voor blauw scherm en kader
50 'en witte tekens
60 SCREEN 0: WIDTH(40): COLOR 15,4,4
70 PRINT"Druk op de toets SELECT."
80 A=RND(1): G#=INKEY$
90 IF G#<>CHR$(24) THEN 80
100 PRINT"Wilt u een aanwijzing, ";
110 INPUT"ja of nee";G$
120 IF G$="ja" THEN 570
130 IF G#<>"nee" THEN 60
140 'voor blauw veld, wit kader
150 'en witte tekens
160 SCREEN 1: WIDTH(32): COLOR 15,4,15
170 KEY OFF: TT=0
180 PRINT: PRINT: PRINT: PRINT
190 PRINT: PRINT: PRINT
200 PRINT TAB(9) "U kunt de flits"
210 PRINT: PRINT TAB(11) "verwachten."
220 'voor tien flitsen
230 FOR N=1 TO 10
240 'bepaal de vertragingstijd
250 G=RND(1)*2000+250
260 FOR T=0 TO G: NEXT T
270 'voor wit veld, blauw kader
280 'en blauwe tekens
290 COLOR 4,15,4
300 'start de timer
310 TIME=0
320 'tijdsduur van de flits
330 FOR P=1 TO 10: NEXT P
340 'terug naar blauw scherm
350 COLOR 15,4,15
```

```

360 'wacht op de reactie
370 G$=INKEY$: IF G$(<>CHR$(32)) THEN 370
380 'bepaal de reactietijd
390 TI(N)=TIME/50: NEXT N
400 'het printen van het resultaat
410 CLS: PRINT TAB(8)"Reactiesnelheid."
420 PRINT: PRINT
430 PRINT"Reactietijd in sec.": PRINT
440 FOR N=1 TO 10
450 PRINT TAB(5)"Flits"N": "TI(N)
460 TT=TT+TI(N): NEXT N: TT=TT/10
470 PRINT: PRINT"Gemiddelde tijd:";TT
480 PRINT: PRINT
490 PRINT"Gebruik voor een volgende ";
500 PRINT"serie": PRINT"toets f1."
510 'schakel toets f1 in
520 KEY(1) ON
530 'wacht op een interrupt
540 GOTO 540
550 'start het programma opnieuw
560 KEY(1) OFF: GOTO 160
570 CLS: PRINT"Als u zo dadelijk door";
580 PRINT" op een willekeu- rige toets";
590 PRINT" te drukken het programma ";
600 PRINT"ver-der laat afwerken, ";
610 PRINT"dan ziet u tien keer na ";
620 PRINT"elkaar het scherm wit ";
630 PRINT"opflitsen. De bedoeling ";
640 PRINT"is dat u na elke flits ";
650 PRINT"zo snelmogelijk de ";
660 PRINT"spatiebalk indrukt."
670 PRINT"Na de tiende flits wordt ";
680 PRINT"uw reactiesnel-heid ";
690 PRINT"afgedrukt."
700 G$=INKEY$: IF G$="" THEN 700
710 GOTO 160

```



daaronder de nummering. Deze nummering hoort bij de plaats, niet bij de kikker. Dat wil zeggen dat bij het springen van een kikker de nummering van de plaatsen niet verandert. U hoeft alleen maar het nummer van de plaats in te voeren waarop de kikker, die u wilt laten springen, staat. Deze gaat dan vanzelf naar de lege plaats, behalve als de sprong die hij moet maken te groot is. Dan gebeurt er gewoon niets. Alle kikkers kunnen zowel naar links als naar rechts springen. Staan aan het eind van het spel alle kikkers op de juiste plaats (de lege plaats weer in het midden) dan wordt automatisch de uitslag getoond met daarbij de waardering. Om een volgend spel te kunnen spelen dient u de f1-toets in te drukken.

8.2. Stuivertje verwisselen

Dit programma had eigenlijk beter kikkertje verwisselen kunnen heten. Het toont namelijk op het scherm een aantal kikkers die een bepaalde kleur hebben. Er zijn vijf groene kikkers en vijf rode kikkers. De vijf groene kikkers staan op vijf plaatsen links vanaf het midden en de vijf rode kikkers rechts vanaf het midden, allemaal op één rij. Tussen de twee kleuren is een plaats vrij gehouden. Er zijn dus in totaal elf plaatsen voor tien kikkers. De bedoeling is dat aan het eind van het spel de kikkers van plaats zijn verwisseld, zodat de groene kikkers rechts en de rode kikkers links staan. Daarvoor moeten ze dus over elkaar heen springen. De moeilijkheid is nu dat een kikker niet over meer dan één andere kikker kan springen. Het aantal stappen, dat wil zeggen: het aantal sprongen en verschuivingen, worden geteld. Men zegt dat het mogelijk is om de verwisseling in vijfendertig stappen uit te voeren. Na het starten van het programma worden de kikkers op het scherm getekend met

```

10 'stuivertje verwisselen
20 'voor de interrupt met toets f1
30 ON KEY GOSUB 930
40 'voor blauw veld en kader
50 'en witte tekens
60 SCREEN 0: WIDTH(40): COLOR 15,4,4
70 KEY OFF
80 PRINT"Wilt u een aanwijzing, ja ";
90 INPUT"of nee";G$
100 IF G$="ja" THEN 990
110 IF G$(">"nee") THEN 40
120 'voor een geel veld en kader
130 'en zwarte tekens
140 SCREEN 1: WIDTH(32): COLOR 1,10,10
150 'het invoeren van de karakters
160 FOR A=0 TO 3: FOR B=0 TO 7
170 READ D: FOR C=0 TO 16 STEP 8
180 VPOKE (232+A+C)*8+B,D: NEXT C
190 NEXT B: NEXT A
200 'het invoeren van de kleuren
210 VPOKE 8221,202: 'groene voorgrond
220 VPOKE 8222,170: 'gele voorgrond
230 VPOKE 8223,106: 'rode voorgrond
240 'het printen van de groene kikkers
250 FOR A=1 TO 5: C=232

```

```

260 GOSUB 870: NEXT A
270 'het printen van een gele kikker
280 A=6: C=240: GOSUB 870
290 'het printen van de rode kikkers
300 FOR A=7 TO 11: C=248
310 GOSUB 870: NEXT A
320 'voor de nummers van de kikkers
330 FOR A=1 TO 11: LOCATE 3*(A-1),14
340 PRINT USING"###";A: NEXT A
350 'L geeft de lege plaats aan
360 'het aantal verplaatsingen is T
370 L=6: T=0
380 'invoer v.h. nummer v.d. kikker
390 PRINT CHR$(11): PRINT
400 PRINT"Welke kikker";
410 'het wissen van het oude nummer
420 PRINT STRING$(4,32);
430 'cursor op de juiste plaats
440 FOR B=0 TO 3: PRINT CHR$(29);
450 NEXT B
460 INPUT N
470 'over niet meer dan 1 kikker
480 D=ABS(N-L): IF D>2 OR D=0 THEN 390
490 T=T+1
500 'bepaal de kleur v.d. gekozen kikker
510 'D bevat informatie over de kleur
520 GOSUB 850
530 'wis de kikker
540 A=N: C=240: GOSUB 870
550 'kikker op de lege plaats
560 A=L: C=D: GOSUB 870: L=N
570 'ga na of de kikkers op hun plaats
580 'staan
590 FOR N=1 TO 5: GOSUB 850
600 IF D<> 248 THEN 390
610 NEXT N: N=6: GOSUB 850
620 IF D<>240 THEN 390
630 'het printen van het resultaat
640 LOCATE 0,17: PRINT"Voor het spel ";
650 PRINT"waren"T"zetten": PRINT"nodig."
660 IF T>35 THEN 690
670 PRINT"Dit is het kleinste aantal ";
680 PRINT"dat mogelijk is.": GOTO 770
690 IF T>39 THEN 710
700 A$="zeer goed": GOTO 760
710 IF T>49 THEN 730
720 A$="redelijk": GOTO 760
730 IF T>59 THEN 750
740 A$="matig": GOTO 760
750 A$="slecht"
760 PRINT"Dit is een "A$" resultaat."
770 PRINT: PRINT"Gebruik voor een ";
780 PRINT"volgend spel de";
790 PRINT"toets f1."
800 'schakel toets f1 in
810 KEY(1) ON
820 'wacht op een interrupt
830 GOTO 830
840 'subr. bepaalt de kleur v.e. kikker
850 D=VPEEK(6496+(N-1)*3): RETURN
860 'subr. print 1 kikker
870 LOCATE 3*(A-1),11
880 PRINT CHR$(C);: PRINT CHR$(C+1)
890 LOCATE 3*(A-1),12
900 PRINT CHR$(C+2);: PRINT CHR$(C+3)
910 RETURN
920 'start het spel opnieuw
930 KEY(1) OFF: CLS: RETURN 250
940 'data voor de kikkers

```

```

950 DATA 0,227,63,7,7,7,7,3
960 DATA 0,135,252,224,224,224,224,192
970 DATA 3,15,24,24,24,24,240,0
980 DATA 192,240,24,24,24,24,15,0
990 CLS: PRINT"Er zijn rode en groene";
1000 PRINT" kikkers die aan het eind ";
1010 PRINT"van het spel van plaats ";
1020 PRINT"moeten zijn verwisseld. ";
1030 PRINT"Het nummer van de kik-";
1040 PRINT"ker die van plaats moet ";
1050 PRINT"veranderen kunt u ";
1060 PRINT" invoeren. Elke kikker ";
1070 PRINT"kan slechts over";
1080 PRINT"1 andere kikker heen ";
1090 PRINT"springen. Het spel moet ";
1100 PRINT"in 35 zetten te doen zijn."
1110 PRINT: PRINT"Druk om verder ";
1120 PRINT"te gaan met het programma";
1130 PRINT"een willekeurige toets in."
1140 G$=INKEY$: IF G$="" THEN 1140
1150 GOTO 130

```

8.3. Verplaatsen van een getallenrij

Het volgende spel laat u een getallenrij op een bepaalde manier verplaatsen. Er zijn drie regels: A, B, en C. Bij de aanvang van het spel staat op regel A een getallenrij. Deze getallenrij moet na afloop in dezelfde vorm op regel C staan. U mag daarbij slechts één getal per keer verplaatsen en wel steeds het laagste getal van de regel. Op de regel waarop het getal wordt geplaatst ontstaat daardoor weer een getallenrij en ook deze moet steeds aflopend zijn, van links naar rechts. Er kan daarom rechts naast een getal op een regel alleen een lager getal komen. Bij het verplaatsen kunt u ook regel B gebruiken. Het volgende voorbeeld moet dit verduidelijken.

Na het starten van het programma kunt u de moeilijkheidsgraad invoeren. Dit is een getal van 1 tot en met 5 en heeft betrekking op de lengte van de getallenrij. Hoe groter de moeilijkheidsgraad des te langer de rij. Daarna ziet u de drie regels met de getallenrij op regel A op het scherm geplaatst. Door een combinatie van een getal en een letter in te voeren, kunt u een getal verplaatsen. Zo verplaatst de combinatie 2B het getal 2 naar rij B, als dat is toegestaan tenminste. Een niet toegestane verplaatsing voert de computer namelijk niet uit. Laten we aannemen dat de rij uit drie getallen bestaat:

```

A 3 2 1
B
C

```

U voert nu 1C in waarvan het resultaat is:

A 3 2
B
C 1

Nu wordt 2B ingevoerd:

A 3
B 2
C 1

Vervolgens 1B.

A 3
B 2 1
C


We gaan verder met 3C:

A
B 2 1
C 3

Het verdere verloop is nu niet moeilijk meer:

A 1
B 2
C 3

A 1
B
C 3 2



A
B
C 3 2 1

Dit is het einde van het spel en daarna wordt u verteld in hoeveel zetten de rij is verplaatst. In werkelijkheid wordt het spel met een langere getallenrij gespeeld. Voor een volgend spel kunt u de f1-toets indrukken.

```
10 'getallenrij
20 'voor interrupt met toets f1
30 ON KEY GOSUB 1560
40 'voor blauw veld en kader
50 'en witte tekens
60 SCREEN 0: WIDTH(40): COLOR 15,4,4
70 KEY OFF
80 PRINT"Wilt u een aanwijzing, ja ";
90 INPUT"of nee";G$
100 IF G$="ja" THEN 1300
110 IF G$(">"nee" THEN 60
120 'voor het printen van de titel
130 GOSUB 830
140 'invoer van de moeilijkheidsgraad
150 LOCATE 0,6
160 INPUT"Welke moeilijkheidsgraad";M
170 'moeilijkheidsgr. niet groter dan 5
180 IF M>5 THEN 130
190 'voor het opbouwen van
200 'de getallenrij
210 M=4+M: FOR N=M TO 1 STEP -1
220 N$=STR$(N): A$(N)=RIGHT$(N$,1)
230 NEXT N
240 'voor het printen van de rijen
250 GOSUB 830: GOSUB 880
260 'wis de laatste zet
270 LOCATE 10,14: PRINT STRING$(5,32)
280 'invoer van de zet
290 LOCATE 0,14
300 INPUT"Welke zet";S$
310 'uitsplitsen van het getal
320 RY=ASC(RIGHT$(S$,1))-64
330 'voor A t.e.m. C is RY 1 t.e.m. 3
340 IF RY<1 OR RY>3 THEN 270
350 'bepaal het getal op de rij
360 NU$=LEFT$(S$,1): NU=ASC(NU$)-48
370 'bepaal of het getal voorkomt
380 IF NU<1 OR NU>M THEN 270
390 'selectie van rij A t.e.m. C
400 ON RY GOTO 430,550,620
410 'voor regel A
420 'zoek het kleinste getal van de rij
430 FOR N=1 TO M
440 IF VAL(A$(N))=0 THEN NEXT N
450 'de rij is leeg bij n=m+1
460 IF N=M+1 THEN 510
470 'opnieuw als het verkeerde getal
480 'is gekozen
490 IF VAL(A$(N))=(NU THEN 270
500 'bepaal de regel waarop het getal is
510 F=0: GOSUB 1020: IF F=1 THEN 270
520 'verwisel het getal
530 A$(N-1)=NU$: GOSUB 880: GOTO 690
540 'voor regel B
```



```

550 FOR N=1 TO M
560 IF VAL(B#(N))=0 THEN NEXT N
570 IF N=M+1 THEN 590
580 IF VAL(B#(N))=<NU THEN 270
590 F=0: GOSUB 1020: IF F=1 THEN 270
600 B#(N-1)=NU#: GOSUB 880: GOTO 690
610 'voor regel C
620 FOR N=1 TO M
630 IF VAL(C#(N))=0 THEN NEXT N
640 IF N=M+1 THEN 660
650 IF VAL(C#(N))=<NU THEN 270
660 F=0: GOSUB 1020: IF F=1 THEN 270
670 C#(N-1)=NU#: GOSUB 880
680 'verhoog de zettenteller
690 U=U+1
700 'ga na of rij C "vol" is
710 IF C#(1)<>"1" THEN 270
720 'einde van het spel
730 LOCATE 0,17
740 PRINT"Het resultaat is in"U;
750 PRINT"zetten bereikt.": PRINT
760 PRINT"Gebbruik voor een volgend ";
770 PRINT"spel": PRINT"toets f1."
780 'schakel toets f1 in
790 KEY(1) ON
800 'wacht op een interrupt
810 GOTO 810
820 'subr. print de titel
830 CLS: PRINT TAB(14);
840 PRINT"Getallenrij."
850 RETURN
860 'subr. print de getallenrijen
870 'rij A
880 LOCATE 0,6: PRINT TAB(3)"A"CHR$(32);
890 FOR X=M TO 1 STEP -1
900 PRINT CHR$(32)A$(X);: NEXT X: PRINT
910 'rij B
920 PRINT: PRINT TAB(3)"B"CHR$(32);
930 FOR X=M TO 1 STEP -1
940 PRINT CHR$(32)B$(X);: NEXT X: PRINT
950 'rij C
960 PRINT: PRINT TAB(3)"C"CHR$(32);
970 FOR X=M TO 1 STEP -1
980 PRINT CHR$(32)C$(X);: NEXT X: RETURN
990 'subr. zoekt in elke regel naar
1000 'het getal
1010 'voor regel A
1020 FOR T=1 TO M
1030 IF VAL(A$(T))=NU THEN 1160
1040 NEXT T
1050 'voor regel B
1060 FOR T=1 TO M
1070 IF VAL(B$(T))=NU THEN 1220
1080 NEXT T
1090 'voor regel C
1100 FOR T=1 TO M
1110 IF VAL(C$(T))=NU THEN 1260
1120 NEXT T
1130 'Ga na of het getal het eerste van
1140 'de regel is. Zo niet, nieuw getal
1150 'voor regel A
1160 IF VAL(A$(T-1))<>0 THEN 1290
1170 'het getal is het eerste v.d. regel
1180 'wis het getal
1190 A$(T)=CHR$(32): GOSUB 880
1200 A$(T)="": RETURN
1210 'voor regel B
1220 IF VAL(B$(T-1))<>0 THEN 1290
1230 B$(T)=CHR$(32): GOSUB 880

```

```


1240 B$(T)="": RETURN
1250 'voor regel C
1260 IF VAL(C$(T-1))<>0 THEN 1290
1270 C$(T)=CHR$(32): GOSUB 880
1280 C$(T)="": RETURN
1290 F=1: RETURN
1300 CLS: PRINT"U kunt kiezen voor ";
1310 PRINT"een moeilijkheids-"
1320 PRINT"graad 1-5. Daarna begint ";
1330 PRINT"het spel. A, B en C stellen";
1340 PRINT"regels voor. Op elke ";
1350 PRINT"regel kan slechts een ";
1360 PRINT"aflopende getallenrij."
1370 PRINT"De getallenrij op regel ";
1380 PRINT"A moet naar "
1390 PRINT"regel C worden ";
1400 PRINT"verplaatst. Alleen het "
1410 PRINT"laagste getal van de ";
1420 PRINT"regel kan worden "
1430 PRINT"verplaatst. Dat ";
1440 PRINT"gaat door het getal en"
1450 PRINT"de letter van de rij ";
1460 PRINT"waar het getal heenmoet. ";
1470 PRINT"gecombineerd in te ";
1480 PRINT"voeren, bij-"
1490 PRINT"voorbeeld: 2C.": PRINT
1500 PRINT"Gebbruik om het programma ";
1510 PRINT"te vervolgen een wille";
1520 PRINT"keurige toets."
1530 G$=INKEY$: IF G$="" THEN 1530
1540 GOTO 130
1550 'start het programma opnieuw
1560 KEY(1) OFF: RUN 130

```

8.4. Neem

Het volgende spelletje staat bekend onder de naam "neem" en wordt gespeeld door twee personen; of door één persoon tegen de computer. Het oorspronkelijke spel bestond uit een aantal rijen voorwerpen van verschillende lengte. Om de beurt nemen de spelers een aantal van die voorwerpen weg. Degene die het laatste voorwerp wegpakt, heeft gewonnen. De spelregels van het spel zijn als volgt. De spelers nemen beurtelings van slechts één rij, welke rij dat geeft niet, een aantal voorwerpen weg. Het aantal voorwerpen dat wordt weggenomen, staat ter keuze van de speler; een hele rij mag ook. Bij elke beurt moet echter tenminste één voorwerp worden weggehaald.

De bedoeling is dat door de spelers een zodanige tactiek wordt gevolgd dat ze de ander dwingen om het voorlaatste voorwerp te moeten wegnemen. Daartoe moeten ze bij de eerste beurt al een zodanige situatie voor de tegenspeler achterlaten dat deze onmogelijk kan winnen; tenzij de speler later een fout maakt. De speler die het spel mag beginnen, heeft dus de meeste kans om te winnen, tenzij



de situatie waarmee het spel aanvangt net ongunstig voor hem is.

Het volgende programma zorgt ervoor dat op het scherm een aantal rijen volle cirkels worden getoond die schijven moeten voorstellen. Het aantal rijen wordt bepaald door de moeilijkheidsgraad, 1 voor drie rijen en 2 voor vier rijen. De rijen zijn gemerkt met A, B, C en eventueel D. U kunt kiezen of u wilt spelen tegen de computer, of tegen een andere speler. De computer laat altijd de eerste beurt aan u. Wordt met twee spelers gespeeld dan wordt de eerste beurt door het toeval bepaald. Het aantal schijven dat u wilt wegnemen, kunt u invoeren door de combinatie van dat aantal en de letter van de rij in te toetsen. Wilt u drie schijven van rij C wegnemen dan voert u 3C in. De lengte van de rijen bij aanvang van het spel worden ook al weer door het toeval bepaald.

```

10 'neem
20 'voor interrupt met toets f1
30 ON KEY GOSUB 1450
40 'voor blauw veld en kader
50 'en witte tekens
60 SCREEN 0: WIDTH(40): COLOR 15,4,4
70 KEY OFF
80 'voor een echt rnd getal
90 A=RND(-TIME)
100 PRINT"Wilt u een aanwijzing, ja ";
110 INPUT"of nee";G$
120 IF G$="ja" THEN 1460
130 IF G$(">"nee" THEN 60
140 'voor blauw veld en kader
150 'en witte tekens
160 SCREEN 1: WIDTH(32): COLOR 15,4,4
170 'het invoeren van de karakters
180 'voor de schijven
190 FOR A=0 TO 7: READ C
200 FOR B=0 TO 32 STEP 8

```

```

210 VPOKE (224+B)*8+A,C: NEXT B: NEXT A
220 'het invoeren van de kleuren
230 'van de schijven
240 VPOKE 8220,20: 'zwarte schijven
250 VPOKE 8221,36: 'groene schijven
260 VPOKE 8222,116: 'cyaan schijven
270 VPOKE 8223,164: 'gele schijven
280 'moeilijkheidsgraad 3 of 4 rijen
290 PRINT"Welke moeilijkheidsgraad, "
300 INPUT"1 of 2";M: M=M+2: PRINT: PRINT
310 IF M>4 OR M<3 THEN 290
320 PRINT"Speelt u tegen de computer "
330 PRINT"(1) of tegen een andere "
340 PRINT"speler (2)";
350 INPUT G: IF G>2 OR G<1 THEN 320
360 'voor het printen van de titel
370 GOSUB 1180
380 'voor m rijen
390 FOR N=1 TO M
400 'bepaal de lengte van de rij
410 T(N)=INT(RND(1)*9+1): NEXT N
420 PRINT: PRINT: PRINT
430 INPUT"Hoe is de naam";N$(1)
440 N$=N$(1): IF G=2 THEN 1070
450 'voor het printen van de schijven
460 GOSUB 1160
470 'wis de vorige zet
480 GOSUB 1330
490 'invoer van de zet
500 PRINT N$" is aan zet";: INPUT A$
510 'bepaal de ingevoerde rij
520 K$=RIGHT$(A$,1): K=ASC(K$)-64
530 IF K<1 OR K>M THEN 480
540 'bepaal het aantal schijven die
550 'moeten verdwijnen
560 S=ASC(LEFT$(A$,1))-48
570 IF S<1 OR S>T(K) THEN 480
580 'bereken het aantal schijven dat op
590 'de rij overblijft en print de rij
600 T(K)=T(K)-S: GOSUB 1230
610 'bepaal of alle rijen leeg zijn
620 GOSUB 1370: IF B=0 THEN 960
630 'volgende zet voor tegenspeler
640 IF G=2 THEN 1130
650 'wis de zet
660 GOSUB 1330
670 'volgende zet voor de computer
680 PRINT"Ik ben aan zet";
690 'wacht
700 FOR N=1 TO 2000: NEXT N
710 'bij gunstige situatie voor de
720 'speler een noodzet
730 GOSUB 1410: IF B=0 THEN 930
740 'bepaal de eerste rij met schijven
750 FOR K=1 TO M: A(K)=T(K)
760 IF T(K)=0 THEN NEXT K
770 'neem steeds 1 schijf weg tot de
780 'situatie gunstig is voor winnen
790 FOR N=1 TO A(K): T(K)=T(K)-1
800 'ga na of de situatie gunstig is
810 'voor winnen
820 GOSUB 1410: IF B=0 THEN 880
830 NEXT N
840 'probeer de volgende rij
850 T(K)=A(K): NEXT K
860 'print de nieuwe situatie en
870 'ga na of de comp. gewonnen heeft
880 GOSUB 1230: GOSUB 1370
890 IF B=0 THEN 990

```

```

900 GOTO 480
910 'neem van de eerste rij met schijven
920 'een schijf weg als noodzet
930 FOR K=1 TO M: IF T(K)=0 THEN NEXT K
940 T(K)=T(K)-1: GOTO 880
950 'speler heeft gewonnen
960 GOSUB 1330: PRINT N$ " heeft ";
970 PRINT "gewonnen.": GOTO 1000
980 'de computer heeft gewonnen
990 GOSUB 1330: PRINT "Ik heb gewonnen!"
1000 PRINT: PRINT: PRINT
1010 PRINT "Gebruik voor een volgend spel"
1020 PRINT "toets f1."
1030 'schakel toets f1 in
1040 KEY(1) ON
1050 'wacht op een interrupt
1060 GOTO 1060
1070 PRINT "Hoe is de naam van de"
1080 INPUT "tegenspeler"; N$(2)
1090 'bepaal welke speler begint
1100 R=INT(RND(1)*2+1)
1110 N$=N$(R): GOTO 460
1120 'het wisselen van speler
1130 IF N$=N$(1) THEN N$=N$(2): GOTO 480
1140 N$=N$(1): GOTO 480
1150 'subr. print de schijven
1160 GOSUB 1180: FOR K=1 TO M
1170 GOSUB 1230: NEXT K: RETURN
1180 'subr. print de titel
1190 CLS: PRINT TAB(7);
1200 PRINT "Neem de schijven.": RETURN
1210 'sub. print 1 rij met schijven
1220 'voor de plaats van de cursor
1230 LOCATE 5,6+2*K
1240 'print de letter
1250 PRINT CHR$(64+K)CHR$(32);
1260 'print de schijven
1270 FOR N=1 TO T(K): IF T(K)=0 THEN 1300
1280 PRINT CHR$(32)CHR$(216+K*8); NEXT N
1290 'verwijderen v.d. gekozen schijven
1300 PRINT STRING$(19-2*T(K),32)
1310 RETURN
1320 'subr. wist de regel
1330 LOCATE 0,16: PRINT STRING$(30,32)
1340 LOCATE 0,16: RETURN
1350 'subr. telt het totaal aantal
1360 'schijven
1370 B=T(1): FOR N=2 TO M
1380 B=B+T(N): NEXT N: RETURN
1390 'subr. bepaalt of de situatie
1400 'gunstig is voor winnen
1410 B=T(1): FOR R=2 TO M
1420 B=B AND NOT T(R) OR NOT B AND T(R)
1430 NEXT R: RETURN
1440 'start het programma opnieuw
1450 KEY(1) OFF: RESTORE: RETURN 160
1460 CLS:PRINT "U speelt het spel NEEM ";
1470 PRINT "tegen de computer of ";
1480 PRINT "tegen elkaar. Er worden ";
1490 PRINT "drie rijen schijven ";
1500 PRINT "getoond. De speler die ";
1510 PRINT "aan zet is moet van 1 rij ";
1520 PRINT "een aantal schijven weg";
1530 PRINT "nemen door het cijfer ";
1540 PRINT "van het aantal en de let";
1550 PRINT "ter van de rij (A, B of C)";
1560 PRINT "in te voeren (bijv. 3B).";
1570 PRINT "alle schijven van 1 rij ";
1580 PRINT "mogen worden weggenomen. ";

```

```

1590 PRINT "De speler die de laatste ";
1600 PRINT "schijf moet nemen heeft"
1610 PRINT "gewonnen."
1620 PRINT "U kunt kiezen uit twee ";
1630 PRINT "moeilijkheids- graden, ";
1640 PRINT "1 of 2.": PRINT: PRINT
1650 PRINT "Gebruik om het programma ";
1660 PRINT "te vervolgen een wille";
1670 PRINT "keurige toets."
1680 G$=INKEY$: IF G$="" THEN 1680
1690 GOTO 160
1700 DATA 60,126,255,255,255,255,126,60

```

8.5. Omkeren

Omkeren is een programma waarmee u een rij getallen die door elkaar staan in de goede volgorde moet zetten. Dit gaat dan uiteraard weer op een bijzondere manier: steeds wordt een gedeelte (of zo u wilt de gehele rij) van de getallen in de omgekeerde volgorde geplaatst. Stel we hebben de rij:

1 7 3 5 2 9 4 8 6

dan kunnen we de eerste zes getallen in tegenovergestelde richting opschrijven:

9 2 5 3 7 1 4 8 6

Met in tegenovergestelde richting wordt hier bedoeld: het in omgekeerde volgorde dan ze in de rij genoteerd staan, opschrijven van de gekozen getallen. Zoals u ziet, staat het getal 9 thans vooraan in de rij. Nu keren we de gehele rij om, hetgeen als resultaat heeft dat de 9 nu op zijn plaats staat:

6 8 4 1 7 3 5 2 9

Nu kunnen we de eerste twee getallen verwisselen:

8 6 4 1 7 3 5 2 9

Verwisselen we daarna de eerst acht getallen dan staat ook het getal acht op zijn plaats:

2 5 3 7 1 4 6 8 9

Bij wijze van voorbeeld voeren we alle verwisselingen uit die we bij deze methode nodig hebben om de getallen op de juiste volgorde te krijgen:

```

7 3 5 2 1 4 6 8 9
6 4 1 2 5 3 7 8 9
3 5 2 1 4 6 7 8 9
5 3 2 1 4 6 7 8 9
4 1 2 3 5 6 7 8 9
3 2 1 4 5 6 7 8 9
1 2 3 4 5 6 7 8 9

```

Zoals u ziet hebben we elf verwisselingen nodig gehad om de getallen in de juiste volgorde te krijgen. Met deze methode is het altijd mogelijk om de meest vreemde combinatie in de goede volgorde te krijgen. Het kan vaak met een minder aantal verwisselingen als het wat handiger wordt aangepakt. Wel moet steeds de getallenreeks worden verwisseld vanaf het begin van de rij, zoals in dit voorbeeld is gedaan. Dit wordt ook in het programma aangehouden. Direct na het starten hiervan krijgt u een rij van negen getallen in een volgorde die door het toeval is bepaald. Om te kunnen verwisselen voert u het getal in dat staat op de plaats tot waar u de getallen wilt verwisselen. Voor de rij:

```
7 1 5 2 4 6 3 8 9
```

voert u het getal 3 in om alle getallen tot en met de zevende plaats verwisseld te krijgen. Op het eind van het spel wordt vermeld hoe veel verwisselingen u nodig hebt gehad. U kunt daarna kiezen of u dezelfde reeks nog een keer gaat proberen dan wel een nieuwe reeks gaat ordenen.

```

10 'omkeren
20 'voor blauw veld en kader
30 'en witte tekens
40 SCREEN 0: WIDTH(40): COLOR 15,4,4
50 KEY OFF
60 PRINT"Wilt u een aanwijzing, ja ";
70 INPUT"of nee";G$: IF G$="ja" THEN 830
80 IF G$("<"nee" THEN 40
90 CLS: PRINT"Een ogenblikje."
100 'voor een echt toevalsgetal
110 A(1)=RND(-TIME)
120 'voor het genereren van de getallen
130 FOR N=1 TO 9

```

```

140 A(N)=INT(RND(1)*9+1)
150 'voor het voorkomen van dubbele
160 'getallen
170 FOR T=0 TO N-1
180 IF A(T)=A(N) THEN 140
190 NEXT T
200 'voor het bewaren van de reeks
210 B(N)=A(N): NEXT N
220 'voor blauw veld en kader
230 'en witte tekens
240 SCREEN 1: WIDTH(32): COLOR 15,4,4
250 'voor het printen van de titel
260 'en de reeks
270 GOSUB 670: GOSUB 800
280 'voor het wissen van de vorige
290 'invoer
300 LOCATE 11,11: PRINT STRING$(4,32)
310 'invoer van het gekozen getal
320 LOCATE 0,11
330 INPUT"Welk getal";G
340 'voorkomen van verkeerde invoer
350 IF G>9 OR G<1 THEN 300
360 'var. U geef het aantal
370 'verwisselingen
380 U=U+1
390 'verwissel de getallen en print
400 'de regel
410 GOSUB 710: GOSUB 800
420 'ga na of de volgorde goed is
430 'niet goed, dan volgende zet
440 FOR N=1 TO 9
450 IF A(N)<>N THEN 300
460 NEXT N
470 'het printen van het resultaat
480 LOCATE 0,15
490 PRINT"U hebt"U"verwisselingen ";
500 PRINT"nodig": PRINT"gehad."
510 'bij het eerste spel b=0
520 IF B=1 THEN 640 ELSE B=1: U1=U: U=0
530 PRINT: PRINT"Wilt u nogmaals ";
540 PRINT"dezelfde reeks,"
550 PRINT"gebruik toets 1."
560 PRINT"Wilt u een volgende reeks,"
570 PRINT"gebruik toets 2."
580 G$=INKEY$: IF G$="1" THEN 620
590 IF G$("<"2" THEN 580
600 RUN 90
610 'haal de oorspr. reeks weer op
620 FOR N=1 TO 9: A(N)=B(N): NEXT N
630 GOTO 270
640 PRINT"Bij het vorige spel was ";
650 PRINT"dat"U1: U1=U: U=0: GOTO 530
660 'subr. print de titel
670 CLS: PRINT TAB(10);
680 PRINT"Verwisselen.": RETURN
690 'sub. keert de gekozen reeks om
700 'tel het aantal te verw. getallen
710 FOR K=1 TO 9: IF A(K)=G THEN 730
720 NEXT K
730 N=1
740 'verwissel twee getallen
750 SWAP A(N),A(K)
760 'voor de volgende getallen
770 N=N+1: K=K-1: IF N<K THEN 750
780 RETURN
790 'subr. print de reeks
800 LOCATE 2,7: FOR N=1 TO 9
810 PRINT A(N);: NEXT N
820 RETURN

```

```

830 CLS: PRINT"Bij de aanvang van het";
840 PRINT" spel ziet u een reeks ";
850 PRINT"getallen in een willekeu";
860 PRINT"rige": PRINT"volgorde. ";
870 PRINT"De bedoeling is dat de ";
880 PRINT"getal- len in de normale ";
890 PRINT"volgorde komen te"
900 PRINT"staan. Door een getal van ";
910 PRINT"de reeks in tetoetsen ";
920 PRINT"wordt de volgorde van ";
930 PRINT"de getal- len tot en met ";
940 PRINT"de ingetoetste omgekeerd.";
950 PRINT"Al omkerend kunt u ze in ";
960 PRINT"de juiste volg-orde krijgen."
970 PRINT: PRINT"Gebruik voor het ";
980 PRINT"vervolgen van het pro-"
990 PRINT"gramma een willekeurige ";
1000 PRINT"toets."
1010 G#=INKEY$: IF G#="" THEN 1010
1020 GOTO 90

```

```

3 7 10 1
8 15 4 6
2 5 11 13
9 14 12

```

8.6. Schuiven

De volgende puzzle is opgebouwd uit vijftien vakjes. Elk vakje heeft zijn eigen nummer en kan schuiven in een rooster van vier bij vier. Er zijn dus zestien plaatsen in het rooster. Dit heeft tot gevolg dat er één plaats vrij is. Daardoor zijn de vakjes steeds langs een verticale of horizontale regel te verschuiven, voor zover die regel de lege plaats bevat. Bij de aanvang van het spel is de plaats van de vakjes willekeurig, dat wil zeggen dat de nummers door elkaar staan. De bedoeling is om de vakjes van links naar rechts en van boven naar beneden door middel van schuiven in volgorde van nummering te krijgen:

```

1 2 3 4
5 6 7 8
9 10 11 12
13 14 15

```

Na het starten van het volgende programma wordt een rooster van vier bij vier op uw scherm geplaatst dat vijftien vakjes bevat waarvan de nummering door het toeval wordt bepaald. Omdat elk nummer slechts één keer mag voorkomen, kan het wel enige tijd duren voordat alle vakjes zijn verschenen. Daarbij is de plaats van het lege vakje ook willekeurig:


Nu kunt u de vakjes laten verschuiven door een getal in te voeren. Voert u bijvoorbeeld 9 (RETURN) in dan zal het vakje dat met 9 is genummerd naar rechts verschuiven en op de lege plaats komen. Het vakje zelf laat dan weer een lege plaats achter. Had u het getal 5 ingevoerd dan was het vakje dat met 5 is genummerd een plaats naar onderen gegaan. U kunt ook een hele rij laten verschuiven door slechts één getal in te voeren. Het intoetsen van het getal 7 heeft het volgende resultaat:

```

3 10 1
8 7 4 6
2 15 11 13
9 5 14 12

```

U ziet het, er zijn drie vakjes in één keer naar beneden geschoven. Al schuivend is het mogelijk de vakjes op hun plaats te krijgen. Door daarna op de f1-toets te drukken (het lege vakje moet zich daarvoor in de rechter benedenhoek bevinden) krijgt u een nieuwe puzzle voorgeschoteld.



```

10 'Boss puzzle
20 'voor interrupt met toets f1
30 ON KEY GOSUB 1440
40 'voor blauw veld en kader
50 'en witte tekens
60 SCREEN 0: WIDTH(40): COLOR 15,4,4
70 PRINT"Wilt u een aanwijzing, ja ";
80 INPUT"of nee";G$
90 IF G$="ja" THEN 1190
100 IF G$("<"nee" THEN 60
110 'zorg voor een echt randomgetal
120 A=RND(-TIME)
130 'voor grijs veld en kader
140 'en voor blauwe tekens
150 COLOR 4,14,14: SCREEN 2
160 OPEN "grp:" FOR OUTPUT AS #1
170 'wacht op het tekenen v.d. puzzle
180 PRESET (0,0)
190 PRINT #1,"Een ogenblikje."
200 'genereer 16 getallen in een
210 'random volgorde
220 FOR Y=1 TO 4: FOR X=1 TO 4
230 G(X,Y)=INT(RND(1)*16+1)
240 'voorkom dubbele getallen
250 FOR N=1 TO 4: FOR T=1 TO 4
260 IF N=Y AND T=X THEN 280
270 IF G(X,Y)=G(T,N) THEN 230
280 NEXT T: NEXT N
290 'voor het volgende getal
300 NEXT X: NEXT Y
310 'print de titel op een schone regel
320 SCREEN 2: PRESET (80,0)
330 PRINT #1,"Boss puzzle."
340 'het printen van het kader
350 LINE (72,20)-(184,172),7,BF
360 'het printen van de vakjes
370 FOR Y=1 TO 4: FOR X=1 TO 4
380 GOSUB 1000: NEXT X: NEXT Y
390 'het invoeren van het getal
400 PRESET (0,184): G1$=""
410 PRINT #1,"Voer het getal in."
420 G$=INKEY$: IF G$="" THEN 420
430 'bij RETURN is getal ingevoerd
440 IF G$=CHR$(13) THEN 460
450 G1$=G1$+G$: GOTO 420
460 G=VAL(G1$)
470 'geen verkeerd getal invoeren
480 IF G<1 OR G>15 THEN 320
490 'zoek de plaats van het vakje
500 FOR Q=1 TO 4: FOR P=1 TO 4
510 IF G(P,Q)=G THEN 540
520 NEXT P: NEXT Q
530 'zoek de plaats van het lege vakje
540 FOR Y1=1 TO 4: FOR X1=1 TO 4
550 IF G(X1,Y1)=16 THEN 590
560 NEXT X1: NEXT Y1
570 'bepaal of de vakjes op een
580 'verticale rij liggen
590 IF P=X1 THEN 900
600 'verkerde invoer als de vakjes nu
610 'niet op een hor. rij liggen
620 Y=Q: IF Q<>Y1 THEN 320
630 'voor het schuiven van een hor. rij
640 'bepaal de richting van het schuiven
650 IF P>X1 THEN 860
660 'schuiven naar rechts
670 FOR X=X1 TO P+1 STEP -1
680 G(X,Y)=G(X-1,Y): GOSUB 1000: NEXT X
690 'het lege vakje op zijn plaats
700 X=P
710 G(X,Y)=16: GOSUB 1000: G=0
720 'ga na of nog meer geschoven
730 'moet worden
740 FOR N=1 TO 4: FOR T=1 TO 4: G=G+1
750 IF G(T,N)<>G THEN 400
760 NEXT T: NEXT N
770 'einde van het spel
780 PRESET (0,174)
790 PRINT #1,"Voor een volgend spel ";
800 PRINT #1,"toets f1."
810 'schakel toets f1 in
820 KEY(1) ON
830 'wacht op interrupt
840 GOTO 840
850 'schuiven naar links
860 FOR X=X1 TO P-1: G(X,Y)=G(X+1,Y)
870 GOSUB 1000: NEXT X: GOTO 700
880 'verticale rij
890 'bepaal de schuifrichting
900 X=P: IF Q>Y1 THEN 960
910 'schuiven naar onder
920 FOR Y=Y1 TO Q+1 STEP -1
930 G(X,Y)=G(X,Y-1): GOSUB 1000: NEXT Y
940 Y=Q: GOTO 710
950 'schuiven naar boven
960 FOR Y=Y1 TO Q-1: G(X,Y)=G(X,Y+1)
970 GOSUB 1000: NEXT Y: GOTO 940
980 'subr. print 1 vakje
990 'bepaal de kleur van het vakje
1000 K=9: IF G(X,Y)<5 THEN K=10
1010 IF G(X,Y)>8 AND G(X,Y)<13 THEN K=10
1020 'voor een leeg vakje
1030 IF G(X,Y)=16 THEN K=4
1040 'bereken de co-ordinaten
1050 X0=56+X*24: X1=X0+24
1060 Y0=-6+Y*34: Y1=Y0+34
1070 'print het vakje
1080 LINE (X0,Y0)-(X1,Y1),K,BF
1090 'voor de plaats van het getal
1100 'in het vakje
1110 X1=9: IF G(X,Y)>9 THEN X1=8
1120 PRESET (X0+X1,Y0+13),K
1130 'print het getal in het vakje
1140 G$=STR$(G(X,Y))
1150 G=1: IF G(X,Y)>9 THEN G=2
1160 A$=RIGHT$(G$,G)
1170 PRINT #1,A$
1180 RETURN
1190 CLS: PRINT"Deze puzzle ";
1200 PRINT"bestaat uit 15 vakjes ";
1210 PRINT"in een 4 bij 4 rooster. ";
1220 PRINT"De vakjes zijn genum-";
1230 PRINT"merd en kunnen schuiven. ";
1240 PRINT"Na afloop van het spel moe";
1250 PRINT"ten de vakjes van links ";
1260 PRINT"naarrechts en van boven ";
1270 PRINT"naar beneden op num-mer in ";
1280 PRINT"het rooster staan. U kunt ";
1290 PRINT"met de vakjes van de ";
1300 PRINT"verticale of de horizonta-";
1310 PRINT"le rij, waarin ook het le";
1320 PRINT"ge vakje is,": PRINT"schui";
1330 PRINT"ven. Daarvoor voert u het ";
1340 PRINT"nummer van het vakje ";
1350 PRINT"in waar de lege plaats ";
1360 PRINT"moet komen."
1370 PRINT: PRINT"Gebruik om terug ";
1380 PRINT"te komen in het pro-"

```


```

1390 PRINT"gramma een willekeurige ";
1400 PRINT"toets."
1410 G$=INKEY$: IF G$="" THEN 1410
1420 GOTO 120
1430 'start het programma opnieuw
1440 KEY(1) OFF: CLOSE: RUN 120

```

8.7. Getal raden

Het volgende programma laat u een getal raden. Hiertoe wordt in de computer een getal gegenereerd van vier cijfers. Deze cijfers zijn geheel willekeurig. In dat getal kunnen meerdere gelijke cijfers voorkomen. Ook kan het eerste cijfer van het getal een 0 zijn. De computer laat dat getal niet zien zodat u het niet kent. Nu moet u een getal invoeren van eveneens vier cijfers. De computer vertelt u nu of er in dat getal cijfers staan die gelijk zijn en op dezelfde plaats staan als in het te raden getal. Ook vermeldt de computer of in het getal cijfers voorkomen die ook in het te raden getal aanwezig zijn maar die niet op de juiste plaats staan. Door voortdurend nieuwe getallen in te voeren met andere cijfers of een andere plaatsing van de cijfers moet u er achter zien te komen welk getal het is. Stel dat het te raden getal is: 5349.



Voert u het getal 4329 in dan krijgt u als antwoord:

PLAATS = 2 GOED 1

Welke cijfers op de goede plaats staan, weet u nog niet (3 en 9) en u kent ook het andere goede cijfer nog niet (4). U voert daarom steeds andere cijfercombinaties in totdat u het goede getal heeft ge-

vonden. Als u het goede getal heeft, meldt de computer dat. Ook wordt het aantal keren vermeld dat u een getal hebt ingevoerd. Door ja in te toetsen kunt u eventueel een nieuw getal raden. Het kan ook zijn dat u er niet uitkomt. In plaats van een getal in te toetsen, typt u dan het woord "help". De computer toont dan het getal.

```

10 'getal raden
20 'rnd getal wordt toevalsgetal
30 CLS: KEY OFF
40 PRINT "druk op de toets SELECT"
50 G$=INKEY$: A=RND(1)
60 IF G$(<)CHR$(24) THEN 50
70 'voor blauw veld en kader
80 'en witte tekens
90 SCREEN 0: WIDTH(40): COLOR 15,4,4
100 PRINT"Wilt u een aanwijzing, ja ";
110 INPUT"of nee";G$
120 IF G$="ja" THEN 640
130 IF G$(<)"nee" THEN 70
140 CLS: T=0: G$=""
150 'stel het te raden getal samen
160 FOR X=1 TO 4: C$=STR$(INT(10*RND(1)))
170 G$=G$+C$: NEXT X
180 'print de titel
190 PRINT TAB(14)"Getal raden.": PRINT
200 PRINT TAB(5);
210 PRINT"Raad een getal van 4 cijfers."
220 PRINT
230 'voer het getal in
240 INPUT"Getal";R$: CL=CSRLIN
250 'cursor 1 regel terug
260 LOCATE 17,CL-1
270 P=0: G=0: Y=0
280 IF R$="help" THEN 530
290 IF LEN(R$)<>4 THEN 580
300 'tel het aantal pogingen T
310 T=T+1: FOR X=1 TO 4
320 'neem elk cijfer apart
330 C$(X)=MID$(G$,2*X,1)
340 RC$(X)=MID$(R$,X,1)
350 'bepaal of cijfers gelijk zijn
360 IF RC$(X)=C$(X) THEN 590
370 'getal gevonden bij p=4
380 NEXT X: IF P=4 THEN 490
390 'zoek naar goede cijfers
400 Y=Y+1: FOR X=1 TO 4
410 IF RC$(Y)=C$(X) THEN 620
420 NEXT X
430 'vergelijk het volgende cijfer
440 IF Y<4 THEN 400
450 'geef resultaat van de poging
460 PRINT "plaats=";P; TAB(33)"goed";G
470 'voor de volgende poging
480 GOTO 230
490 'geef het resultaat
500 PRINT: PRINT
510 PRINT "Getal geraden in";T;
520 PRINT "pogingen": PRINT: GOTO 540
530 PRINT: PRINT "Het getal is";G$
540 INPUT"Nog eens? ja of nee";J$
550 IF J$="ja" THEN 140
560 IF J$(<)"nee" THEN 540
570 END

```

```

580 PRINT: PRINT "4 cijfers!": GOTO 230
590 'tel het aantal juiste cijfers
600 P=P+1: C$(X)="": RC$(X)="A"
610 GOTO 370
620 'tel het aantal goede cijfers
630 G=G+1: C$(X)="": GOTO 430
640 CLS: PRINT"De computer heeft een ";
650 PRINT"getal van vier cijfers ";
660 PRINT"gegenereerd. Hierin ";
670 PRINT"kunnen meer-dere gelijke ";
680 PRINT"cijfers voorkomen en het ";
690 PRINT"eerste cijfer kan ook ";
700 PRINT"een 0 zijn. U moeten ";
710 PRINT"willekeurig getal van ";
720 PRINT"vier cijfers invoeren. ";
730 PRINT"De computer vertelt u ";
740 PRINT"hoeveel cijfers er ";
750 PRINT"goed zijn en op de ";
760 PRINT"juiste plaats staan en ";
770 PRINT"hoeveel cijfers er goed ";
780 PRINT"zijn maar niet op de juiste";
790 PRINT" plaats": PRINT"staan."
800 PRINT: PRINT"Gebruik een wille";
810 PRINT"keurige toets om in het";
820 PRINT"programma terug te komen."
830 G$=INKEY$: IF G$="" THEN 830
840 GOTO 140

```

8.8. Woord raden

Het volgende programma laat u een woord raden. De bedoeling is dat de computer daarvoor een woord kiest uit een groot aantal woorden die hem ter beschikking worden gesteld. Daarvoor gaan we eerst een aantal files met woorden samenstellen die naar een tape worden geschreven. De computer kiest dan eerst een file en daarna een woord uit die file. Voor het samenstellen van de files gebruiken we het volgende hulpprogramma:

```

10 'woorden in files naar de tape
20 'voor blauw veld en kader
30 'en witte tekens
40 SCREEN 0: WIDTH(40): COLOR 15,4,4
50 'het printen van de titel
60 PRINT TAB(8)"Samenstellen van files."
70 PRINT: PRINT
80 'invoeren van het aantal vd files
90 OPEN"cas:raden" FOR OUTPUT AS #1
100 INPUT"Aantal files";A$
110 'niet meer dan 10 files
120 A=VAL(A$): IF A<1 OR A>10 THEN 100
130 'aantal files nr. naar de tape
140 PRINT #1, A$
150 CLOSE
160 FOR X=1 TO A
170 'open de file met naam X
180 OPEN"cas:"+STR$(X) FOR OUTPUT AS #1
190 PRINT"File"X
200 'aantal woorden in de file
210 INPUT"Hoeveel woorden"; N$
220 'niet meer dan 10 woorden per file

```

```

230 IF VAL(N$)<1 OR VAL(N$)>10 THEN 210
240 'het getal v.h. aantal woorden
250 'naar de tape
260 PRINT #1, N$: FOR Y=1 TO VAL(N$)
270 'invoeren van de woorden
280 PRINT"Woord";Y;:INPUT B$
290 'woorden niet langer dan 10 letters
300 IF LEN(B$)>10 THEN 280
310 PRINT #1, B$: NEXT Y: CLOSE
320 'voor de volgende file
330 NEXT X

```

Na het starten van het programma wordt eerst de label met de filenaam "woord raden" naar de band geschreven. Daarna moet u het aantal files invoeren. Wilt u tien files van elk acht woorden samenstellen dan voert u hiervoor tien in. Dit getal wordt als string in de file "woord raden" naar de band geschreven. Dan volgt file 1 waarvoor u eerst het nummer van het aantal woorden (8) van die file invoert. Dit nummer wordt als eerste string in de betreffende file geplaatst. Nu voert u na elkaar de woorden voor die file in. Als deze file op de band staat kunt u file 2 samenstellen enzovoorts, tot en met de laatste file. Het maximum aantal files dat u kunt invoeren is 10. Het maximum aantal woorden per file is tien, terwijl de woorden niet meer dan tien letters mogen bevatten.

Staan alle woorden op de band dan kunt u het hoofdprogramma in de computer laden. De cassette-recorder moet dan al klaar staan en de tape teruggespoeld tot voor de file "woord laden". Nu kunt u het programma starten. U drukt play in op de cassette-recorder en u zult zien dat een file met woorden in de computer wordt geladen. Welke file wordt bij toeval bepaald. Dan meldt de computer uit hoeveel letters het woord dat u moet raden is opgebouwd. U voert nu een letter in. De computer meldt of, en indien ja, hoe vaak de letter in het woord voorkomt. De plaats in het woord wordt niet bekend gemaakt. Nu probeert u een volgende letter, tot u alle letters van het woord hebt gevonden. Ook dit wordt door de computer gemeld. Dan probeert u het woord samen te stellen en als u dat goed hebt gedaan kunt u eventueel een volgend woord raden. Lukt het u niet om een woord te raden dan typt u "help" waarop de computer het juiste woord geeft.

```

10 'woordraden
20 'voor blauw veld en kader
30 'en witte tekens

```



```

40 SCREEN 0: WIDTH(40): COLOR 15,4,4
50 'rnd getal wordt toevalsgetal
60 CLS: KEY OFF
70 PRINT "druk op toets SELECT"
80 G$=INKEY$: A=RND(1)
90 IF G$(<>CHR$(24)) THEN 80
100 'begin van het spel
110 CLS
120 'inlezen van het aantal vd files
130 PRINT"Een ogenblikje."
140 OPEN"cas:raden" FOR INPUT AS #1
150 INPUT #1, A$: CLOSE: A=VAL(A$)
160 'F is het volgnummer van de file
170 F=F+1
180 'kies de file
190 CLS: PRINT"De file wordt geladen."
200 B(F)=INT(A*RND(1))+1: T=0
210 'bepaal of de file als is gekozen
220 FOR C=1 TO F
230 IF B(F)=B(C) AND F(<>C) THEN 200
240 NEXT C
250 'haal de file van de tape
260 OPEN"cas:"+STR$(B(F))FOR INPUT AS #1
270 'inlezen van het aantal v.d. woorden
280 INPUT #1, C$: N=VAL(C$)
290 'inlezen van de woorden
300 FOR D=1 TO N
310 INPUT #1, W$(D): NEXT D: CLOSE
320 CLS: PRINT "Spoel de tape terug."
330 FOR D=0 TO 5: CL=10
340 COLOR CL: FOR C=0 TO 500: NEXT C
350 IF CL=10 THEN CL=7: GOTO 340
360 NEXT D: COLOR 15
370 'het printen van de titel
380 CLS: PRINT TAB(14) "Woord raden."
390 PRINT TAB(10) "Type bij nood help."
400 PRINT
410 'bepaal het volgnummer (1 tot N)
420 T=T+1
430 'kies het woord
440 P(T)=INT(N*RND(1))+1: FOR Y=1 TO T
450 'bepaal of het woord al is gekozen
460 IF P(T)=P(Y) AND T(<>Y) THEN 440
470 'X is het woordnummer
480 NEXT Y: X=P(T)
490 L=LEN(W$(X)):G$=""
500 PRINT "Het woord heeft";L;"letters."
510 PRINT: INPUT"Welke letter";L$:K=0
520 IF L$="help" THEN 770
530 'ga na of de letter al bekend is
540 FOR Q=1 TO LEN(G$)
550 IF L$=MID$(G$,Q,1) THEN 960
560 NEXT
570 'bepaal of de letter in het woord is
580 FOR R=1 TO L
590 IF L$(<>MID$(W$(X),R,1) THEN 650
600 'letter in het woord gevonden
610 'K telt het aantal gelijke letters
620 'voeg de letter bij de reeds
630 'gevonden letters
640 K=K+1: G$=G$+L$
650 NEXT: PRINT K"keer "L$
660 IF LEN(G$)=L THEN 700
670 ' voor de volgende letter
680 GOTO 510
690 'alle letters van het woord gevonden
700 PRINT: PRINT "U hebt alle letters."
710 PRINT "Uw letters zijn "G$: PRINT
720 INPUT"Welk woord";W$

```

```


730 IF W$="help" THEN 770
740 'ga na of het woord juist is
750 IF W$=W$(X) THEN 780
760 PRINT "Verkeerd woord.": GOTO 720
770 PRINT "Het woord is "W$(X): GOTO 800
780 PRINT "Goed geraden."
790 'bepaal of alle woorden zijn geweest
800 IF T=N THEN 860
810 'alle woorden nog niet gevonden
820 INPUT"Nog eens, ja of nee";J$
830 IF J$="ja" THEN 380
840 IF J$(<>"nee" THEN 820 ELSE 950
850 'alle woorden v.d. file gevonden
860 PRINT "U hebt alle woorden."
870 'ga na of alle files zijn geweest
880 IF F=A THEN 940
890 'nog niet alle files geraden
900 PRINT "Wilt u nog een serie, ";
910 INPUT"ja of nee";J$
920 IF J$="ja" THEN 170
930 IF J$(<>"nee" THEN 900 ELSE 950
940 PRINT "U hebt alle series."
950 END
960 PRINT"Letter al bekend.": GOTO 510

```

8.9. Colormind

Colormind is een spel dat u een kleurencombinatie laat raden. Daartoe stelt de computer zelf een kleurencombinatie samen die hij uiteraard niet laat zien. Daarna voert u een kleurencombinatie in. De computer vertelt hoeveel kleuren van uw combinatie in zijn combinatie voorkomen maar niet op de juiste plaats staan en hoeveel kleuren wel op de goede plaats staan. Het aantal kleuren waaruit kan worden gekozen is beperkt tot zes. De kleurencombinatie bestaat uit vier eenheden. Dat houdt in dat het aantal kleuren in de combinatie kleiner dan vier kan zijn. In dat geval komt in een combinatie een bepaalde kleur meer dan eens voor. Deze identieke kleuren kunnen ook naast elkaar staan, zoals in de combinatie groen, rood, rood, geel, de kleur rood twee keer voorkomt.

Had u de combinatie rood, blauw, cyaan, geel ingevoerd dan vermeldt de computer dat één kleur op de juiste plaats staat en dat één kleur wel in de combinatie zit, maar niet op de goede plaats. Welke kleuren dat zijn, wordt niet vermeld. Hierbij is ervan uitgegaan dat de eerst genoemde combinatie die van de computer is. Door steeds een nieuwe combinatie in te voeren moet u, gebruikmakend van de antwoorden door de computer, de gevraagde combinatie zien te vinden. U kunt echter niet meer dan tien combinaties invoeren. De kleuren waaruit gekozen kan worden zijn: rood, cyaan, magenta, groen, blauw en geel.



Na het invoeren en het starten van het programma kunt u uw combinatie invoeren door de eerste twee letters (of meer) van elke gekozen kleur in te toetsen. De op deze manier ingevoerde kleuren worden door de computer afgedrukt als een verticale balk. Onder deze balk wordt met zwarte merktekens aangegeven hoeveel kleuren uit uw combinatie hetzelfde zijn en op de goede plaats staan. Met blauwe merktekens wordt aangegeven hoeveel kleuren er verder nog overeenkomen met die in de gevraagde combinatie maar niet op de juiste plaats staan. Heeft u tien combinaties ingevoerd en de gevraagde combinatie niet gevonden dan wordt deze door de computer getoond. Heeft u de combinatie eerder gevonden dan wordt dat ook door de computer vermeld. Steeds kunt u na een spel aan een volgend beginnen door de f1-toets in te drukken.

```

10 'colormind
20 'voor interrupt met toets f1
30 ON KEY GOSUB 1100
40 'voor blauw veld en kader
50 'en witte tekens
60 SCREEN 0: WIDTH(40): COLOR 15,4,4
70 PRINT"wilt u een aanwijzing, ja ";
80 INPUT"of nee";G$
90 IF G$="ja" THEN 1110
100 IF G$(">")"nee" THEN 60
110 'voor wit veld en kader
120 'en blauwe tekens
130 SCREEN 1: WIDTH(32): COLOR5,15,15
140 'invoeren van de karakters
150 FOR P=0 TO 7: A=VPEEK(1616+P)
160 FOR N=208 TO 248 STEP 8
170 VPOKE N*8+P,A: NEXT N: NEXT P
180 FOR P=0 TO 7: READ A
190 FOR N=192 TO 200 STEP 8
200 VPOKE N*8+P,A: NEXT N: NEXT P
210 FOR P=0 TO 7: READ A
220 VPOKE 1544+P,A: NEXT P

```

```

230 'het invoeren van de kleuren
240 VPOKE 8216,31: 'zwart
250 VPOKE 8217,95: 'l.blauw
260 VPOKE 8218,79: 'blauw
270 VPOKE 8219,111: 'rood
280 VPOKE 8220,127: 'cyan
290 VPOKE 8221,207: 'groen
300 VPOKE 8222,223: 'magenta
310 VPOKE 8223,175: 'geel
320 'voor een echt random getal
330 A=RND(-TIME): CLS
340 'het vormen van de combinatie
350 FOR N=1 TO 4
360 K(N)=INT(RND(1)*6+1): NEXT N
370 'het printen van de titel
380 PRINT TAB(11) "Colormind."
390 'voor max. tien beurten
400 FOR T=1 TO 10
410 'wissen van de ingev. combinatie
420 FOR N=1 TO 4: LOCATE 0,2+N
430 PRINT STRING$(15,32): NEXT N
440 'invoeren van de kleuren
450 FOR N=1 TO 4: G(N)=0: LOCATE 0,2+N
460 INPUT"kleur";G$(N): NEXT N
470 'bepaal de kleurnummers
480 FOR N=1 TO 4
490 IF LEFT$(G$(N),2)="bl" THEN G(N)=1
500 IF LEFT$(G$(N),2)="ro" THEN G(N)=2
510 IF LEFT$(G$(N),2)="cy" THEN G(N)=3
520 IF LEFT$(G$(N),2)="gr" THEN G(N)=4
530 IF LEFT$(G$(N),2)="ma" THEN G(N)=5
540 IF LEFT$(G$(N),2)="ge" THEN G(N)=6
550 'alleen de toegelaten kleuren
560 IF G(N)=0 THEN 420
570 'het printen v.d. ingev. kleuren
580 NEXT N: GOSUB 1050
590 'zoek naar gelijke kleuren
600 'op gelijke plaats
610 PL=0: G=0: FOR Y=1 TO 4
620 FOR X=1 TO 4
630 IF G(X)=K(X) THEN 790
640 'zoek naar gelijke kleuren
650 'op andere plaats
660 IF G(X)=K(Y) THEN 840
670 NEXT X: NEXT Y
680 'voor het printen van het resultaat
690 GOSUB 960
700 NEXT T
710 'het printen v.d. juiste combinatie
720 FOR N=1 TO 4: G(N)=K(N): NEXT
730 GOSUB 1050
740 'print de pijl
750 LOCATE 15+T,3+8: PRINT CHR$(193)
760 LOCATE 0,17: PRINT"U hebt tien ";
770 PRINT"beurten gehad.": GOTO 880
780 'tel het aantal juiste plaatsen
790 PL=PL+1: G(X)=0
800 'ga na of de combinatie juist is
810 IF PL=4 THEN 850
820 GOTO 670
830 'tel het aantal juiste kleuren
840 G=G+1: G(X)=0: GOTO 670
850 LOCATE 0,17: PRINT"U hebt de ";
860 PRINT"juiste combinatie"
870 PRINT"gevonden."
880 PRINT"Gebruik voor een volgend ";
890 PRINT"spel": PRINT"toets f1."
900 'schakel toets f1 in
910 KEY(1) ON

```


```

920 'wacht op een interrupt
930 GOTO 930
940 'subr. print het resultaat
950 'print het aantal goede plaatsen
960 N=0: IF PL=0 THEN 1000
970 FOR N=1 TO PL: LOCATE 15+T,11+N
980 PRINT CHR$(192): NEXT N: N=N-1
990 'print het aantal goede kleuren
1000 IF G=0 THEN 1030
1010 FOR P=1 TO G: LOCATE 15+T,11+N+P
1020 PRINT CHR$(200): NEXT P
1030 RETURN
1040 'subr. print de kleuren
1050 Q=0: FOR N=1 TO 4: FOR P=0 TO 1
1060 LOCATE 15+T,3+Q: Q=Q+1
1070 PRINT CHR$(200+G(N)*8)
1080 NEXT P: NEXT N: RETURN
1090 'voor en volgend spel
1100 KEY OFF: CLS: RUN 330
1110 CLS: PRINT"Colormind is een spel ";
1120 PRINT"dat u een kleuren-combina";
1130 PRINT"tie laat raden. De computer";
1140 PRINT" kiest vier kleuren uit ";
1150 PRINT"zes mogelijkheden. Er";
1160 PRINT"kunnen meerdere gelijke ";
1170 PRINT"kleuren voorko- men."
1180 PRINT"U kunt zelf een kleurencom";
1190 PRINT"binatie invoe- ren door ach";
1200 PRINT"ter het woord kleur op het "
1210 PRINT"scherm de eerste twee ";
1220 PRINT"letters van elke gekozen ";
1230 PRINT"kleur in te voeren ";
1240 PRINT"(vier kleu- ren). Met een ";
1250 PRINT"blauw merkteken wordt"
1260 PRINT"aangegeven of een kleur ";
1270 PRINT"in de gevraagde combinatie ";
1280 PRINT"voorkomt en met een zwart "
1290 PRINT"merkteken of een kleur ";
1300 PRINT"voorkomt en op de juiste ";
1310 PRINT"plaats staat."
1320 PRINT"U kunt kiezen uit de ";
1330 PRINT"kleuren blauw,"
1340 PRINT"rood, cyaan, groen, magenta";
1350 PRINT" of geel."
1360 PRINT: PRINT"Gebruik om het ";
1370 PRINT"programma te vervolgen"
1380 PRINT"een willekeurige toets."
1390 G#=INKEY$: IF G#="" THEN 1390
1400 GOTO 130
1410 DATA 63,63,63,63,63,63,63,0
1420 DATA 4,14,31,4,4,4,4,0

```

8.10. Galgje

Het spel galgje vertoont overeenkomsten met het spel woord raden. Er zijn echter toch wel belangrijke verschillen. Bij dit spel wordt het woord getoond door het aantal letters aan te geven met een evenredig aantal punten: zoveel punten zoveel letters. Als u een letter raadt en het is er één die in het woord voorkomt dan verschijnt deze letter in de puntenrij op de plaats die hij in het woord heeft.



Komt de letter meer dan eens voor dan wordt hij ook evenzo vaak getoond, steeds op de juiste plaats. Zo ziet u het woord als het ware voor u ontstaan. Dit maakt het wat makkelijker om het woord bij voorbaat, dus nog voordat alle letters bekend zijn, te raden. Het aantal keren dat u een letter kunt raden is echter beperkt. Per woord kunt u acht letters verkeerd raden, meer niet. De woorden die u bij dit spel nodig hebt, kunt u op dezelfde manier invoeren als bij het spel woord raden, met het daarbij gegeven hulpprogramma. U zet hiermee de woorden in files op de cassetteband. Na het starten van het programma kiest dit een file en uit deze file een woord, zodat de speler niet kan weten welk woord is gekozen. Op het scherm verschijnt dan de puntenrij die het aantal letters van het woord aangeeft. Ook is op het scherm een mannetje getekend en een galg. Als u het woord niet op tijd raadt, hangt de man. Op een gegeven moment kondigt dat gevaar zich aan omdat het touw waaraan de man komt te hangen, naar beneden komt zakken. Aan het eind van ieder spel kunt u een nieuw woord raden. Zijn alle woorden van een file geraden dan wordt door het programma een nieuwe file van de band gehaald. Dit gaat door totdat alle woorden zijn geraden.

```

10 'galgje
20 'voor een echt toevalsgetal
30 A=RND(-TIME): KEY OFF
40 'voor een blauw veld en kader
50 'en witte tekens
60 SCREEN 0: WIDTH(40): COLOR 15,4,4
70 'voor een interrupt met toets f1
80 ON KEY GOSUB 1590
90 PRINT"Wilt u een aanwijzing, ja ";
100 INPUT"of nee";G#:

```

```

110 IF G$="ja" THEN 1610
120 IF G$<>"nee" THEN 60
130 'begin van het spel
140 CLS
150 'het inlezen van het aantal vd files
160 PRINT"Een ogenblikje."
170 OPEN"cas:raden" FOR INPUT AS #1
180 INPUT #1, A$: CLOSE: A=VAL(A$)
190 'F is het volgnummer van de file
200 F=F+1: T=0
210 'kies de file
220 SCREEN 0
230 PRINT"De file wordt geladen."
240 B(F)=INT(A*RND(1))+1
250 'bepaal of de file al is gekozen
260 FOR C=1 TO F
270 IF B(F)=B(C) AND F<>C THEN 240
280 NEXT C
290 'haal de file van de tape
300 OPEN"cas:"+STR$(B(F))FOR INPUT AS #1
310 'inlezen van het aantal v.d. woorden
320 INPUT #1, C$: N=VAL(C$)
330 'inlezen van de woorden
340 FOR D=1 TO N
350 INPUT #1, W$(D): NEXT D: CLOSE
360 CLS: PRINT"Spoel de tape terug."
370 FOR D=0 TO 8: CL=15
380 COLOR CL: FOR C=0 TO 250: NEXT C
390 IF CL=15 THEN CL=7: GOTO 380
400 NEXT D: COLOR 15
410 'voor blauw veld en kader
420 'en voor witte tekens
430 SCREEN 1,3: WIDTH(32): COLOR 15,4,4
440 'het invoeren van de spritepatronen
450 RESTORE: FOR D=0 TO 2: S$=""
460 FOR C=0 TO 31: READ E
470 S$=S$+CHR$(E): NEXT C
480 SPRITE$(D)=S$: NEXT D
490 'print de titel
500 PRINT TAB(12) "Galgie."
510 'print de hor. galgbalk
520 LOCATE 18,8: PRINT CHR$(202);
530 PRINT STRING$(4,223)
540 'print de vert. galgbalk
550 FOR Y=0 TO 10: LOCATE 23,8+Y
560 PRINT CHR$(221): NEXT Y
570 'print de bodemlijn
580 LOCATE 16,19
590 PRINT STRING$(9,195)
600 'print de man op de bodemlijn
610 PUT SPRITE 0, (132,119),10,0
620 'kies een woord
630 T=T+1
640 P(T)=INT(N*RND(1)+1)
650 'ga na of het woord al eerder
660 'is gekozen
670 FOR E=1 TO T
680 IF P(T)=P(E) AND T<>E THEN 640
690 NEXT E
700 'geef de lengte van het woord
710 WW$="": L=LEN(W$(P(T)))
720 FOR E=1 TO L: WW$(E)=". "
730 WW$=WW$+W$(E): NEXT E
740 'voor het printen van de stippen
750 GOSUB 1550
760 'maximum aantal beurten L+8
770 FOR P=1 TO L+8
780 'wis de laatste invoer
790 LOCATE 13,6
800 PRINT STRING$(15,32)
810 'terug naar de gewiste regel
820 LOCATE 0,6
830 'voer de gekozen letter in
840 INPUT"Welke letter";L$
850 'vergelijk de ingevoerde letter
860 'met de letters van het woord
870 FOR E=1 TO L
880 LL$=MID$(W$(P(T)),E,1)
890 'noteer de plaats van de letter
900 'indien deze in het woord voorkomt
910 IF L$=LL$ THEN WW$(E)=L$
920 NEXT E
930 'plaats de gevonden letters
940 'op de stippen
950 WW$="": FOR E=1 TO L
960 WW$=WW$+WW$(E): NEXT E
970 'print de gevonden letters
980 GOSUB 1550
990 'ga na of het woord is gevonden
1000 IF WW$=W$(P(T)) THEN 1210
1010 'bepaal of het touw moet zakken
1020 IF P>L+3 THEN 1380
1030 'voor de volgende beurt
1040 NEXT P
1050 'het woord is niet gevonden
1060 'print het juiste woord
1070 WW$=W$(P(T))
1080 GOSUB 1550
1090 LOCATE 0,6
1100 PRINT"Te laat, gebruik toets f1."
1110 'voor interrupt met toets f1
1120 KEY(1) ON
1130 'de man aan de galg
1140 FOR D=0 TO 2
1150 FOR C=0 TO 60: NEXT C
1160 PUT SPRITE 0, (132,100),8+D,D:NEXT D
1170 FOR C=0 TO 60: NEXT C
1180 PUT SPRITE 0, (132,100),9,1
1190 GOTO 1140
1200 'het woord is geraden
1210 LOCATE 0,21
1220 PRINT"U hebt het woord geraden."
1230 'als F=A dan zijn alle
1240 'files geraden
1250 IF F=A AND T=N THEN 1350
1260 PRINT"Wilt u een volgend spel,"
1270 INPUT"ja of nee";G$
1280 IF G$="nee" THEN END
1290 IF G$<>"ja" THEN 1210
1300 'als T=N dan zijn alle woorden
1310 'van de file geraden
1320 IF T<>N THEN 430
1330 'voor de volgende file
1340 GOTO 200
1350 PRINT"U hebt alle woorden gehad."
1360 END
1370 'het printen van het touw
1380 FOR C=0 TO P-L-4
1390 LOCATE 18,9+C
1400 PRINTCHR$(1)CHR$(86): NEXT C
1410 GOTO 1040
1420 'data voor spritepatroon 1
1430 DATA 3,7,3,1,3,7,11,19,3,3,1,1,1,1
1440 DATA 1,3,128,128,128,0,128,192,160
1450 DATA 144,128,128,0,0,0,0,0,0
1460 'data voor spritepatroon 2
1470 DATA 3,7,3,1,3,7,7,15,11,3,1,2,6,4
1480 DATA 12,0,128,128,128,0,128,192,192

```


```

1490 DATA 224,160,128,128,192,64,32,64,0
1500 'data voor spritepatroon 3
1510 DATA 3,7,3,1,3,3,3,7,7,3,2,4,40,16
1520 DATA 0,0,128,128,128,0,128,128,128
1530 DATA 192,192,128,128,64,32,16,32,0
1540 'subr. print de gevonden letters
1550 LOCATE 0,4
1560 PRINT"Het woord: ";WW$
1570 RETURN
1580 'terug naar het programma
1590 KEY(1) OFF: LOCATE 0,21
1600 RETURN 1260
1610 CLS: PRINT"Galgje is een spel dat";
1620 PRINT" u een woord laat raden. ";
1630 PRINT"Een aantal stippen geven ";
1640 PRINT"de leng-te van het woord. ";
1650 PRINT"U kunt een letter in-"
1660 PRINT"voeren en als deze in het ";
1670 PRINT"woord voorkomtwordt hij ";
1680 PRINT"op de juiste plaats ge";
1690 PRINT"toond. Door steeds ";
1700 PRINT"opnieuw een letter in te"
1710 PRINT"voeren moet u het woord ";
1720 PRINT"raden. U kunt"
1730 PRINT"een max. aantal letters ";
1740 PRINT"invooeren. Heeft u het dan ";
1750 PRINT"nog niet gevonden dan ";
1760 PRINT"hangt": PRINT"de man."
1770 PRINT: PRINT"Gebruik om naar het ";
1780 PRINT"programma terug te"
1790 PRINT"keren een willekeurige ";
1800 PRINT"toets."
1810 G$=INKEY$:IF G$="" THEN 1810
1820 GOTO 140

```

8.11. Ruimtecapsule

In het spel "ruimtecapsule" gaat het erom een ruimtecapsule heelhuids te laten landen op het oppervlak van een denkbeeldige planeet. Een hinderenis hierbij is dat niet al te ruim kan worden omgesprongen met de brandstof die nodig is voor de remraketten. Deze remraketten zijn hard nodig omdat de denkbeeldige planeet geen dampkring heeft die een remmende werking kan uitoefenen op de capsule. Bij de aanvang van het spel is de capsule op 500 000 m afstand van de planeet en heeft hij een daalsnelheid van 2500 m/s. U heeft 7000 kg brandstof en de valversnelling is 4 m/s^2 . Deze gegevens krijgt u op het scherm geprint direct na het starten van het programma. U ziet de hoogte boven het oppervlak, die in meter is gegeven, kleiner worden. Verder ziet u dat de vertraging, dat betekent de snelheidsvermindering van de capsule, -4 m/s^2 is. De eenheden zijn overigens niet achter de getallen vermeld. Een negatieve snelheidsvermindering houdt in dat de capsule steeds sneller zal vallen. De remraketten zullen de daalsnelheid moeten laten afnemen. Er is nog geen brandstof



verbruikt. Dat de snelheid toeneemt, blijkt uit het getal van de daalsnelheid. Dit wordt uiteraard veroorzaakt door de valversnelling ten gevolge van de aantrekkingskracht van de planeet. Als de hoogte is verminderd tot 400000 m ziet u aan de rechterbovenkant van het scherm de onderkant van de capsule verschijnen. De daalsnelheid van de capsule op uw scherm is in overeenstemming met het getal dat daarvoor wordt getoond. U kunt de remraketten in werking stellen door het indrukken van de CRSR-toets (CRSR UP). De vertragende werking daarvan ziet u aan het getal achter VERTRAGING, dat van -4 al snel verandert in een positief getal. U ziet de snelheid verminderen en tegelijkertijd de brandstofhoeveelheid afnemen. De bedoeling is dat de snelheid kleiner is dan 5 m/s op het moment dat de capsule de bodem raakt (hoogte is 0). Bij een grotere landingsnelheid wordt de capsule vernield. Heeft u zoveel brandstof verbruikt dat deze voor de landing op is, dan neemt de valsnelheid weer toe en zal de capsule te pletter vallen. Natuurlijk zult u de remwerking wel eens willen verminderen. Dat kan ook met een CRSR-toets (CRSR DOWN). Voor het brandstofverbruik is het erg ongunstig als u de vertraging zo groot maakt en de remmende werking zo lang laat duren dat de capsule zelfs weer gaat opstijgen. Deze situatie zult u in elk geval moeten zien te voorkomen.

```

10 'ruimtecapsule
20 'voor een interrupt met de f1 toets
30 ON KEY GOSUB 1000
40 'voor elke seconde een interrupt
50 ON INTERVAL=50 GOSUB 1020
60 'freq. getal ruis, kanaal 1 aan

```

```

70 SOUND 6,27: SOUND 7,54: SOUND 8,16
80 'instellen periodetijd omhullende
90 SOUND 11,49: SOUND 12,30: SOUND 0,55
100 'voor blauw veld en kader
110 'en witte tekens
120 SCREEN 0: WIDTH(40): COLOR 15,4,4
130 KEY OFF
140 PRINT"Wilt u een aanwijzing, ja ";
150 INPUT"of nee";G$
160 IF G$="ja" THEN 1030
170 IF G$(">")"nee" THEN 120
180 'het invoeren van de spritepatronen
190 RESTORE: FOR N=0 TO 31: READ A
200 S1$=S1$+CHR$(A): NEXT N
210 FOR N=0 TO 31: READ B
220 S2$=S2$+CHR$(B): NEXT N
230 'voor blauw veld en kader
240 'en witte tekens
250 SCREEN 1,3: WIDTH(32): COLOR 15,4,5
260 'stel de sprites samen
270 SPRITE$(0)=S1$
280 SPRITE$(1)=S2$
290 'print de bodemlijn
300 LOCATE 20,23
310 PRINT STRING$(8,192)
320 'voor resp. hoogte, brandstof
330 'versnelling en snelheid
340 H=500000!: B=7500: A1=0: V=2500
350 'zet de tijd (TI) op nul
360 TI=0: T1=0: T0=0: TIME=0: T=0: A1=0
370 'elke sec. correctie van de tijd TI
380 INTERVAL ON
390 'bereken de tijd- en de
400 'snelheidsverandering
410 TI=TI+TIME/51
420 T1=TI: T=T1-T0: T0=T1: A=4-A1
430 'berekening van de nieuwe hoogte
440 H=H-(V*T+A/2*T^2): IF H<0 THEN H=0
450 'bereken de Y co-ordinaat
460 'voor de sprite
470 Y=149-INT(H/2193)
480 IF Y<-50 THEN 520
490 'print de sprite
500 PUT SPRITE 0, (176,Y),10,0
510 'bereken de nieuwe snelheid
520 V=V+T*A
530 'berekening brandstof
540 B=B-6*A1*T/5
550 'het printen van de gegevens
560 LOCATE 0,0
570 PRINT"Hoogte"TAB(11);
580 PRINT USING"#####";INT(H)
590 PRINT: PRINT"Snelheid"TAB(11);
600 PRINT USING"#####";INT(V)
610 PRINT: PRINT"Vertraging"TAB(11);
620 PRINT USING"#####";INT(-A)
630 PRINT: PRINT"Brandstof"TAB(11);
640 PRINT USING"#####";INT(B)
650 'ga na of de capsule geland is
660 IF H=0 THEN 800
670 'bij B=0 brandstof op, vertraging -4
680 IF B<0 THEN A1=0: B=0: GOTO 410
690 'bepaal of een CURSOR toets in is
700 G$=INKEY$
710 'vergroting van de vertraging
720 IF G$=CHR$(30) THEN A1=A1+1
730 'vermindering van de vertraging
740 IF G$=CHR$(31) THEN A1=A1-1
750 'vertraging niet kleiner dan -4
760 IF A1<0 THEN A1=0
770 GOTO 410
780 'ga na of de landingssnelheid
790 'niet te groot is
800 IF V>5 THEN 930
810 'goede landing
820 LOCATE 0,8
830 PRINT"U heeft een goede landing ";
840 PRINT"gemaakt.": PRINT
850 PRINT"Gebruik voor een volgend spel"
860 PRINT"toets f1."
870 'schakel toets f1 in
880 KEY(1) ON
890 'wacht op een interrupt met toets f1
900 GOTO 900
910 'landingssnelheid te groot
920 'schakel geluid in
930 SOUND 1,0: SOUND 13,0
940 'voor spritepatroon 1
950 PUT SPRITE 0, (176,Y),10,1
960 LOCATE 0,8
970 PRINT"Uw landing is mislukt!"
980 PRINT: GOTO 850
990 'terug voor volgend spel
1000 KEY(1) OFF: RETURN 250
1010 'verhoog de klok met 1 sec.
1020 TI=TI+1: TIME=0: RETURN
1030 CLS: PRINT"U bevindt zich in een ";
1040 PRINT"ruimtecapsule op 500 km ";
1050 PRINT"hoogte van het oppervlak ";
1060 PRINT"van een planeet zonder ";
1070 PRINT"dampkring. Uw snelheid is";
1080 PRINT"2500 m/s en u heeft een ";
1090 PRINT"beperkte hoe-";
1100 PRINT"veelheid brandstof. De val";
1110 PRINT"versnelling is 4 m/s^2. ";
1120 PRINT"De remraketten kunt u in";
1130 PRINT"scha- kelen met de CURSOR";
1140 PRINT" UP toets en weer"
1150 PRINT"uitschakelen met de ";
1160 PRINT"CURSOR DOWN toets."
1170 PRINT"Uw landingssnelheid moet ";
1180 PRINT"kleiner zijn dan 5m/s."
1190 PRINT: PRINT"Gebruik om terug ";
1200 PRINT"te komen in het"
1210 PRINT"programma een ";
1220 PRINT"willekeurige toets."
1230 G$=INKEY$: IF G$="" THEN 1230
1240 GOTO 190
1250 'data voor spritepatroon 1
1260 DATA 1,3,7,13,12,13,13,15
1270 DATA 13,13,13,12,31,60,120,112
1280 DATA 128,192,224,176,176,48,176,240
1290 DATA 240,240,240,48,248,60,30,14
1300 'data voor spritepatroon 2
1310 DATA 0,0,0,0,0,0,0,7
1320 DATA 31,31,31,127,127,255,255
1330 DATA 0,0,0,0,0,0,0,128,240,240
1340 DATA 240,248,248,240,243,247

```

8.12. Paren zoeken

Bij het volgende spel zijn in veertig vakjes twintig letters verborgen. Het zijn de letters van A tot en met T en van elke letter precies twee. Het is de be-

doeling door het openen van de vakjes de gelijke letters bij elkaar te zoeken. Dit spel wordt gespeeld door meerdere spelers. De speler die de meeste paren letters heeft gevonden, heeft gewonnen. Elke speler mag twee vakjes openen. Bevatten deze twee vakjes verschillende letters dan worden ze weer gesloten en mag de volgende speler twee vakjes openen. Maakt een speler twee vakjes open waarin zich gelijke letters bevinden dan heeft hij een punt verdient en mag hij nogmaals twee vakjes openen. Dit gaat zo door totdat hij twee vakjes open maakt met verschillende letters. Dan gaat de beurt naar de volgende speler. Iedereen kan zien welke letters de geopende vakjes bevatten, uiteraard totdat ze weer worden gesloten. De bedoeling is dat iedere speler goed onthoudt welke letters de vakjes bevatten zodat hij later, als hij weer aan de beurt is en een bepaalde letter nodig heeft, weet waar hij die kan vinden. Bij dit spel is een goed geheugen dan ook belangrijker dan geluk. Door het vaak te spelen traint u dan ook uw geheugen.


Na het starten van het programma moet eerst aan de computer worden verteld hoeveel spelers er meedoen. Dan kunt u de namen van die spelers invoeren en de volgorde waarin dat geschiedt, bepaalt ook de volgorde waarin de spelers aan de beurt komen. De computer geeft met de naam van de speler aan wiens beurt het is. Na het invoeren van de namen wordt u verzocht om een ogenblikje te wachten. Dat "ogenblikje" zal wel even duren want de computer moet veertig vakjes vullen met letters en geen enkele letter mag meer dan twee keer voorkomen. Daarna worden met gekleurde blokjes de vakjes voorgesteld: vijf regels van acht kolommen. De regels zijn van 1 tot en met 5 ge-

nummerd en de kolommen met letters van A tot en met H. Nu geeft de computer aan welke speler aan de beurt is en deze kan door het invoeren van de letter van de kolom en het cijfer van de regel, in deze volgorde, bijvoorbeeld D3, een vakje openen. Dan verschijnt de letter in het vakje dat is geopend. Nu kan de speler nog een vakje openen. Zijn de letters die nu zichtbaar zijn gelijk dan blijven de vakjes open maar de letters worden gewist. De speler blijft aan de beurt. Zijn de letters niet gelijk dan wordt het vakje na korte tijd gesloten en de beurt gaat naar de volgende speler. Het is dus zaak de letter snel in u op te nemen. Zijn alle paren letters gevonden dan wordt door de computer de uitslag getoond. Het indrukken van de f1-toets is nodig om nog een volgend spel te kunnen spelen.

```

10 'zoek de paren
20 'voor blauw veld en kader
30 'en witte tekens
40 SCREEN 0: WIDTH(40): COLOR 15,4,4
50 KEY OFF
60 PRINT"Wilt u een aanwijzing, ja ";
70 INPUT"of nee";G$
80 IF G$="ja" THEN 1510
90 IF G$(">")"nee" THEN 40
100 'voor een echt randomgetal
110 A=RND(-TIME)
120 'het bij toeval bepalen van
130 'de letters voor de vakjes
140 CLS: PRINT"Een ogenblikje."
150 CLEAR 1000
160 FOR Y=1 TO 5: FOR X=1 TO 8
170 'rnd numers van 65-104
180 L(X,Y)=INT(RND(1)*40+65)
190 'het voorkomen van dubbele nummers
200 FOR N=1 TO 5: FOR M=1 TO 8
210 IF M=X AND N=Y THEN 230
220 IF L(X,Y)=L(M,N) THEN 180
230 NEXT M,N,X,Y
240 'voor dubbele nummers
250 'tussen 65 en 84
260 FOR Y=1 TO 5: FOR X=1 TO 8
270 IF L(X,Y)>84 THEN L(X,Y)=L(X,Y)-20
280 NEXT X,Y: BEEP
290 'het invoeren van de namen
300 'van de spelers
310 CLS: INPUT"Hoeveel spelers";S
320 DIM S$(S), T(S): PRINT
330 FOR N=1 TO S
340 PRINT"Hoe heet speler"N;
350 INPUT S$(N): PRINT: NEXT N
360 'voor blauw veld en kader
370 'en witte tekens
380 SCREEN 2: COLOR 15,4,5
390 'voor het printen van de letters
400 'voor de kolommen
410 OPEN"grp:" FOR OUTPUT AS #1
420 FOR X=65 TO 72
430 PRESET (55+(X-65)*20,0),4
440 PRINT #1, CHR$(X): NEXT X
450 'voor het printen van de cijfers

```



```

460 'voor de rijen
470 FOR Y=1 TO 5
480 PRESET(26,26+(Y-1)*28),4
490 PRINT #1,Y: NEXT Y
500 'voor het printen va het kader
510 LINE (44,12)-(215,160),7,BF
520 'voor het printen van de vakken
530 FOR Y=1 TO 5: FOR X=1 TO 8
540 'voor het printen van 1 vak
550 GOSUB 1350: NEXT X,Y
560 'voor het opvolgen v.d. spelers
570 FOR Q=1 TO S
580 'elke speler twee beurten
590 FOR P=1 TO 2
600 'wis de laatste keuze
610 LINE (0,156+P*8)-(255,190),4,BF
620 'terug naar de gewiste regel
630 PRESET (0,156+P*8),4
640 PRINT #1,S$(Q)+" speelt";
650 PRINT #1," vakje";P;
660 'invoer van de keuze
670 V$=INKEY$: IF V$="" THEN 670
680 X(P)=ASC(V$)-64
690 IF X(P)<1 OR X(P)>8 THEN 610
700 PRINT #1,V$;
710 V$=INKEY$: IF V$="" THEN 710
720 Y(P)=ASC(V$)-48
730 IF Y(P)<1 OR Y(P)>5 THEN 610
740 'niet twee keer hetzelfde vakje
750 IF X(P)=X(P-1)ANDY(P)=Y(P-1) THEN610
760 'niet valsspelen
770 IF L(X(P),Y(P))=0 THEN 610
780 PRINT #1,V$
790 'opent het gekozen vakje
800 K=4: X=X(P): Y=Y(P)
810 GOSUB 1390
820 'bepaal de letter in het vak
830 G(P)=L(X(P),Y(P))
840 'print de letter in het vakje
850 GOSUB 1450
860 'voor het volgende vakje
870 NEXT P
880 'ga na of de vakken een gelijke
890 'inhoud hebben
900 IF G(1)=G(2) THEN 1060
910 'wis de regels
920 LINE (0,164)-(255,190),4,BF
930 PRESET (0,172)
940 PRINT #1,"Misgeraden."
950 'laat de vakken nog even open
960 FOR N=0 TO 4000: NEXT N
970 'sluit de vakken
980 X=X(1): Y=Y(1): GOSUB 1350
990 X=X(2): Y=Y(2): GOSUB 1350
1000 'voor de volgende speler
1010 NEXT Q
1020 'herhaal de rij spelers
1030 GOTO 570
1040 'de vakken zijn gelijk
1050 'tel het aantal punten
1060 T(Q)=T(Q)+1
1070 LINE (0,164)-(255,190),4,BF
1080 PRESET (0,172)
1090 PRINT #1,"Goed geraden."
1100 FOR N=0 TO 2000: NEXT N
1110 'maak de vakken leeg
1120 X=X(1): Y=Y(1): K=4: GOSUB 1390
1130 X=X(2): Y=Y(2): K=4: GOSUB 1390
1140 L(X(1),Y(1))=0: L(X(2),Y(2))=0
1150 'ga na of alle paren gevonden zijn
1160 FOR Y=1 TO 5: FOR X=1 TO 8
1170 IF L(X,Y)>0 THEN 590
1180 NEXT X,Y
1190 'print het resultaat
1200 SCREEN 0
1210 PRINT"Alle paren gevonden."
1220 FOR Q=1 TO S: PRINT
1230 PRINT S$(Q)" heeft"T(Q)"paren."
1240 NEXT Q: PRINT: PRINT
1250 PRINT"Gebruik voor een volgend ";
1260 PRINT"spel toets f1."
1270 'voor interrupt met de f1 toets
1280 ON KEY GOSUB 1500
1290 'schakel toets f1 in
1300 KEY(1) ON
1310 'wacht op een interrupt
1320 GOTO 1320
1330 'subr. print 1 vak
1340 'voor het bepalen van de kleur
1350 B= X AND 1: B=-B
1360 IF (Y AND 1)=1 THEN B=B+1
1370 IF ABS(B)=1 THEN K=10 ELSE K=15
1380 'het berekenen van de plaats
1390 X2=48+(X-1)*20: X3=X2+20
1400 Y2=16+(Y-1)*28: Y3=Y2+28
1410 'het printen van het vak
1420 LINE(X2,Y2)-(X3,Y3),K,BF
1430 RETURN
1440 'subr. print een letter in het vak
1450 PRESET (X2+7,Y2+10),4
1460 L$=CHR$(G(P)): A$=RIGHT$(L$,1)
1470 PRINT #1,A$
1480 RETURN
1490 'start het programma opnieuw
1500 KEY(1) OFF: RUN 110
1510 CLS: PRINT"U speelt dit spel met ";
1520 PRINT"2 of meer spelers tegen ";
1530 PRINT"elkaar. Het scherm ";
1540 PRINT"toont 40 vakjeswaarachter ";
1550 PRINT"20 dubbele letters ";
1560 PRINT"verborgen zijn. Elke speler";
1570 PRINT" mag twee vakjes ope-";
1580 PRINT"nen. Treft hij gelijke ";
1590 PRINT"letters dan telt dat voor ";
1600 PRINT"1 punt en mag hij nog ";
1610 PRINT"twee vak-jes openen. Zijn ";
1620 PRINT"alle paren gevonden dan";
1630 PRINT"is het spel ten einde."
1640 PRINT"Een vakje is te openen ";
1650 PRINT"door het invoeren van de ";
1660 PRINT"kolomletter en het regel";
1670 PRINT"nummer, bijv. D4."
1680 PRINT: PRINT"Gebruik voor het ";
1690 PRINT"terugkeren naar het"
1700 PRINT"programma een ";
1710 PRINT"willekeurige toets."
1720 G$=INKEY$: IF G$="" THEN 1720
1730 GOTO 110

```



9. Gokken

9.1. Eenentwintig

Eenentwintig is een spel dat is afgeleid van het bekende kaartspel eenentwintigen. Erg veel verschil is er niet. Er zijn wat afwijkende spelregels. Ten eerste kan het slechts worden gespeeld door één persoon tegen de computer. Dat komt omdat de kaarten nu eenmaal voor iedereen duidelijk zichtbaar op het scherm staan. Ten tweede kan zowel door de speler als door de computer maar met één stapeltje kaarten worden gespeeld. De computer schudt de kaarten en deelt ze uit. Dat wil zeggen hij geeft eerst de speler een kaart en daarna zichzelf. Dan krijgt opnieuw de speler een kaart. Deze kan nu bij toeval hebben gewonnen. Is dat niet het geval dan neemt de computer zelf ook een tweede kaart. Hierna kan ieder voor zichzelf beslissen of hij nog meer kaarten wil. Dat gebeurt ook al weer om de beurt en kaart voor kaart. De speler kan daarvoor op de vraag: KAART? met JA antwoorden of hij nog een kaart wil en met NEE als hij past. Ook de computer kan passen. Of dat gebeurt is daar merkt de speler niets van. Hij kan met kaarten vragen doorgaan tot hij voldoende punten bij elkaar denkt te hebben. En daar zit dan het gok-element in: de speler moet meer punten hebben dan de computer maar mag niet meer punten hebben dan 21. Heeft hij er meer dan 21 dan heeft hij het spel verloren. Dat geldt ook voor de computer. De puntentelling wordt dan ook bij het uitrekenen van elke kaart door de computer bijgehouden en zodra de grens van 21 punten is gepasseerd wordt het spel gestopt. Het spel stopt ook indien de speler of de computer precies 21 punten heeft. Degene die die eer te beurt valt heeft het spel gewonnen. Er wordt met 32 kaarten gespeeld, van de vier kleuren: alle kaarten boven de 6. Voor de puntenwaarde van de kaarten geldt de boer voor 1 punt, de vrouw voor 2, de heer voor 3 en het aas voor 11 of 1. De andere kaarten hebben een waarde die gelijk is aan hun nummer. De waarde van een aas wordt op 11 gesteld, tenzij de speler een kaart krijgt waardoor zijn puntentotaal boven de 21 uitkomt. Dan wordt de waarde van het aas automatisch teruggebracht op 1. Heeft de speler twee azen dan wordt de waar-

de van een van de twee tot 1 teruggebracht als het totale puntenaantal meer wordt dan 21. Wordt daarna de totale puntenwaarde weer meer dan 21 dan wordt ook de tweede aas in waarde verminderd.

Na het starten van het programma krijgt u uw twee kaarten te zien. De kaarten van de computer krijgt u niet te zien. De computer is overigens eerlijk, hij kijkt ook niet in uw kaarten! Nu kunt u door JA in te voeren nog een kaart krijgen en daarna eventueel nog meer, net zo lang tot u past. Heeft de computer nog niet gepast dan gaat deze door tot ook deze past. Nu worden de kaarten van de computer getoond en wordt de uitslag bekend gemaakt, met het aantal punten van de speler en van de computer. Dat gebeurt ook als een van de twee voor die tijd 21 punten of meer dan 21 punten heeft. Door de f1-toets in te drukken kunt u een volgend spel spelen.



```
10 'eenentwintig
20 KEY OFF
30 'voor blauw veld en kader
40 'en witte tekens
50 SCREEN 0: WIDTH(40): COLOR 15,4,4
60 PRINT"Wilt u een aanwijzing, ja ";
70 INPUT"of nee";G$
80 IF G$="ja" THEN 1950
90 IF G$(">"nee" THEN 50
```

```

100 'voor interrupt met de f1 toets
110 ON KEY GOSUB 2200
120 'voor een groen veld en kader
130 'en blauwe tekens
140 SCREEN 1: WIDTH(32): COLOR 4,3,12
150 'voor een echt randomgetal
160 A=RND(-TIME)
170 'voor het overnemen van karakters
180 FOR A= 0 TO 3: FOR B=0 TO 7
190 VPOKE 1904+8*A+B,VPEEK(24+8*A+B)
200 NEXT B,A
210 'voor de kleuren van de karakters
220 VPOKE 8221,99: VPOKE 8222,19
230 'het printen van de titel
240 PRINT TAB(9)"Eenentwintig.": PRINT
250 PRINT"Jouw kaarten. ";
260 PRINTTAB(19)"Mijn kaarten."
270 S#=CHR$(238)+CHR$(239)
280 S#=S#+CHR$(240)+CHR$(241)
290 'voor de soort kaarten
300 K#="BVH789JAA"
310 'voor de waarde van de kaarten
320 'en vorm de stapels
330 FOR S=1 TO 5: FOR K=1 TO 9
340 READ A(S,K): NEXT K: RESTORE: NEXT S
350 'eerste kaart voor de speler
360 GOSUB 1250: N=1: J$(N)=R#: Q=3: YQ=4
370 'print de kaart op het scherm
380 GOSUB 1380
390 'eerste kaart voor de computer
400 GOSUB 1250: M$(N)=R#: GOSUB 1770
410 'tweede kaart voor de speler
420 GOSUB 1250: N=2: J$(N)=R#: YQ=6
430 'print de kaart, tel de punten
440 GOSUB 1380: GOSUB 1510
450 'VJ=1 bij groter dan 21
460 IF VJ=1 THEN 1000
470 'WJ=1 bij 21 punten
480 IF WJ=1 THEN 1000
490 'tweede kaart voor de computer
500 GOSUB 1250: M$(N)=R#
510 'tel de punten voor de computer
520 'VM=1 bij groter dan 21
530 GOSUB 1770: IF VM=1 THEN 1000
540 'WM=1 bij 21 punten
550 IF WM=1 THEN 1000
560 'als de speler heeft gepast (NJ=1)
570 'is de beurt voor de computer
580 N=N+1: IF NJ=1 THEN 730
590 'speler heeft nog niet gepast
600 'kan nog een kaart nemen
610 LOCATE 11,YQ: PRINT"Kaart";
620 'als speler past g#="nee"
630 INPUT G#: IF G#="nee" THEN 960
640 IF G#<>"ja" THEN 610
650 'de speler neemt nog een kaart
660 GOSUB 1250: J$(N)=R#: YQ=YQ+2
670 'print de kaart
680 GOSUB 1380
690 'tel de punten van de speler
700 GOSUB 1510: IF VJ=1 THEN 1000
710 IF WJ=1 THEN 1000
720 'doorgaan als computer past (NM=1)
730 IF NM=1 THEN 920
740 'ga na of de computer moet passen
750 IF TM>17 THEN 980
760 FOR T=1 TO N
770 K=VAL(RIGHT$(M$(T),1))
780 IF K=8 THEN 860
790 NEXT T: P=0: U=0: FOR S=1 TO 4
800 FOR K=1 TO 8: A=A(S,K)
810 IF A=0 THEN 840
820 P=P+1
830 IF (A<(21-TM)) OR (A=11) THEN U=U+1
840 NEXT K,S: IF U<(P/2) THEN 980
850 'de computer neemt nog en kaart
860 GOSUB 1250: M$(N)=R#
870 'tel de punten
880 GOSUB 1770
890 IF VM=1 THEN 1000
900 IF WM=1 THEN 1000
910 'als beide passen NJ=1 en NM=1
920 IF NJ=1 AND NM=1 THEN 1190
930 'voor de volgende ronde
940 GOTO 580
950 'speler heeft gepast
960 NJ=1: GOTO 730
970 'computer heeft gepast
980 NM=1: GOTO 920
990 'het printen van het resultaat
1000 W#="Jij bent"
1010 IF VJ=1 OR WM=1 THEN W#="Ik ben"
1020 'bewaars het laatste regelnummer
1030 YR=YQ
1040 'print de kaarten v.d. computer
1050 GOSUB 1900
1060 'bepaal het hoogste regelnummer
1070 IF YR>YQ THEN YQ=YR
1080 YQ=YQ+3: LOCATE 0,YQ
1090 PRINT"Jouw punten:"TAB(20);
1100 PRINT"Mijn punten:"
1110 PRINT TAB(2) TJ TAB(26) TM: PRINT
1120 PRINT TAB(7) W# "winnaar!"
1130 PRINT: PRINT"Nog een spel? ";
1140 PRINT"gebruik toets f1."
1150 'schakel toets f1 in
1160 KEY(1) ON
1170 'wacht op een interrupt
1180 GOTO 1180
1190 W#="Jij bent"
1200 IF TM>TJ THEN W#="Ik ben"
1210 IF TM=TJ THEN W#="geen"
1220 GOTO 1050
1230 'subr. geeft een kaart
1240 'bepaal de kaartkleur
1250 S=INT(RND(1)*4+1)
1260 'bepaal de kaartsoort
1270 K=INT(RND(1)*8+1)
1280 'opnieuw als de kaart al
1290 'is uitgegeven
1300 IF A(S,K)=0 THEN 1250
1310 'stel de kaart samen
1320 R#=STR$(S)+STR$(K)
1330 'haal de kaart van de stapel
1340 A(S,K)=0
1350 RETURN
1360 'subr. print 1 kaart op het scherm
1370 'bepaal het kleurnummer
1380 S=VAL(LEFT$(R#,2))
1390 'bepaal het kaartnummer
1400 K=VAL(RIGHT$(R#,1))
1410 'bepaal het kaartsymbool
1420 K1#=MID$(K#,K,1)
1430 IF K1#="J" THEN K1#="10"
1440 'bepaal het kleursymbool
1450 S1#=MID$(S#,S,1)
1460 'print de symbolen
1470 LOCATE 0,YQ: PRINT S1#;K1#

```

```

1480 RETURN
1490 'subr. telt de punten v.d. speler
1500 'voor het aantal uitgegeven kaarten
1510 TJ=0: FOR T=1 TO N
1520 'bepaal de puntenwaarde
1530 'en tel de punten op
1540 K=VAL(RIGHT$(J$(T),1))
1550 T(T)=A(5,K): TJ=TJ+T(T)
1560 'ga na of het aantal punten
1570 'te groot is
1580 IF TJ>21 THEN 1660
1590 'ga na of het spel gewonnen is
1600 IF TJ=21 THEN WJ=1
1610 'voor de volgende kaart
1620 NEXT T
1630 RETURN
1640 'aant. punten te groot
1650 'zoek naar een aas
1660 FOR P=1 TO N
1670 K=VAL(RIGHT$(J$(P),1))
1680 IF K=8 THEN 1730
1690 NEXT P
1700 'geen aas gevonden, verloren
1710 VJ=1: RETURN
1720 'een aas gevonden, maak de waarde 1
1730 J$(P)=LEFT$(J$(P),2)+"9"
1740 TJ=TJ-10: GOTO 1620
1750 'subr. berekent de punten
1760 'van de computer
1770 TM=0: FOR T=1 TO N
1780 K=VAL(RIGHT$(M$(T),1))
1790 T(T)=A(5,K): TM=TM+T(T)
1800 IF TM>21 THEN 1830
1810 IF TM=21 THEN WM=1
1820 NEXT T: RETURN
1830 FOR P=1 TO N
1840 K=VAL(RIGHT$(M$(P),1))
1850 IF K=8 THEN 1870
1860 NEXT P: VM=1: RETURN
1870 M$(P)=LEFT$(M$(P),2)+"9"
1880 TM=TM-10: GOTO 1820
1890 'subr. print de kaarten v.d. comp.
1900 Q=27: FOR U=1 TO N: YQ=2+2*U
1910 R$(U)=M$(U)
1920 IF M$(U)="" THEN RETURN
1930 GOSUB 1380: NEXT U: RETURN
1940 DATA 1,2,3,7,8,9,10,11,1
1950 CLS: PRINT"Eenentwintig is een ";
1960 PRINT"kaartspel dat 1 speler ";
1970 PRINT"met de computer speelt. ";
1980 PRINT"De computer schudt en ";
1990 PRINT"geeft de kaarten, eerst de"
2000 PRINT"speler en dan zichzelf 1 ";
2010 PRINT"kaart. Na de": PRINT"tweede";
2020 PRINT"kaart kan men passen ";
2030 PRINT"met nee, of nog een nemen ";
2040 PRINT"met ja. 21 punten wint,"
2050 PRINT"meer dan 21 punten ";
2060 PRINT"verliest, anders wint";
2070 PRINT"het grootste aantal punten.";
2080 PRINT"Alleen de kaarten hoger ";
2090 PRINT"dan 6 worden gebruikt."
2100 PRINT"Voor de punten: boer (B)=1,";
2110 PRINT"vrouw (V)=2,heer (H)=3 ";
2120 PRINT"en aas (A)=1 of 11."
2130 PRINT: PRINT"Gebruik voor het ";
2140 PRINT"terugkeren naar het"
2150 PRINT"programma een ";
2160 PRINT"willekeurige toets."

```

```

2170 G$=INKEY$: IF G$="" THEN 2170
2180 GOTO 110
2190 'start het programma opnieuw
2200 KEY(1) OFF: RUN 110

```

9.2. Het lottoformulier

Zelf bepalen welke nummers u op het lottoformulier zult aankruisen, dat betekent tevens dat u het lot in eigen hand neemt. Wilt u dat niet dan kunt u door de computer laten bepalen welke getallen u moet aankruisen. U kunt het volgende programma daarvoor gebruiken.

Na het starten van het programma dient u in te voeren hoeveel kolommen u wilt spelen. Daarna geeft de computer op het scherm evenzoveel regels van zes getallen tussen 1 en 41. Hiermee kunt u dan uw lottoformulier invullen.

```

10 'lotto
20 'voor interrupt met de toets f1
30 ON KEY GOSUB 420
40 'voor een echt randomgetal
50 A=RND(-TIME): KEY OFF
60 'voor blauw veld en kader
70 'en witte tekens
80 SCREEN 0: WIDTH(40): COLOR 15,4,4
90 'het printen van de titel
100 PRINT TAB(16)"Lotto."
110 PRINT: PRINT: PRINT
120 'invoer aantal kolommen
130 INPUT"Hoeveel kolommen wilt u";K
140 'voor aantal kolommen
150 FOR T=1 TO K
160 'voor zes getallen per kolom
170 FOR N=1 TO 6
180 'randomgetallen van 1 tot 41
190 G(N)=INT(RND(1)*41+1)
200 'geen dubbele getallen
210 FOR M=1 TO N
220 IF G(M)=G(N) AND M<>N THEN 190
230 NEXT M,N
240 'sorteer de getallen
250 FOR N=0 TO 4: FOR M=1 TO 5-N
260 IF G(M)<G(M+1) THEN 280
270 SWAP G(M),G(M+1)
280 NEXT M,N
290 'print 1 kolom
300 PRINT: FOR N=1 TO 6
310 PRINT TAB(3+4*N)G(N);
320 NEXT N: PRINT: PRINT
330 'voor de volgende kolom
340 NEXT T: PRINT
350 PRINT"Gebruik voor een volgende ";
360 PRINT"serie toets f1"
370 'schakel toets f1 in
380 KEY(1) ON
390 'wacht op een interrupt
400 GOTO 400
410 'start het programma opnieuw
420 KEY(1) OFF: RUN

```

10. Morse cursus

Het leren ontcijferen van de morsecode, dus het leren verstaan van de morsecode, en het in verstaanbare taal omzetten van een in morsecode uitgezonden bericht, is vooral een kwestie van routine: van zeer vaak doen dus. Opleidingsinstituten beschikken over een schat aan lesmateriaal waarmee de cursisten uitgebreid kunnen oefenen. Dat materiaal heeft iemand die het zichzelf wil leren, zoals veel radioamateurs, niet. Voor de MSX-bezitters is er een oplossing, de morse-cursus, die u in dit hoofdstuk wordt aangeboden. Deze cursus omvat het leren van de code en het oefenen in het ontcijferen hiervan. Hij is daarom uit twee gedeelten opgebouwd: morse leren en morse oefenen.

10.1. Morse leren

Dit gedeelte van de cursus omvat een zestal lessen. Deze hebben tot doel om u de morsecodetekens te leren herkennen, ook al worden ze met grote seinsnelheid ten gehore gebracht. Het gebruik van dit programma is eenvoudig: na het starten kunt u de seinsnelheid invoeren en het lesnummer kiezen. Uiteraard kiest u om te beginnen een lage seinsnelheid, bijvoorbeeld 5 woorden per minuut door een 5 in te voeren en natuurlijk les 1. Hierna worden vier letters en een cijfer op het scherm geschreven: e, i, h, s en de 5 met de daarbij behorende morsecodes (enkel punten). Nu hoort u uit de luidspreker van uw tv de morsecode van één van deze tekens. Het enige wat u hebt te doen, is de toets met de betreffende letter of het cijfer indrukken in de SHIFT-mode. Het beste is het om hiervoor de CAPS LOCK toets in te drukken. De computer vertelt of u het goed hebt gedaan. Heeft u een fout gemaakt dan meldt hij dat, met vermelding van het teken dat u wel had moeten gebruiken. Pas na het indrukken van de toets krijgt u het volgende code-teken te horen waarvoor u weer een toets moet indrukken (de goede) enzovoorts. Dit herhaalt zich twintig keer. Dan verschijnen er weer een aantal tekens, onder de voorgaande op uw scherm (t, m, o en 0). Hiervan zijn de morsecodetekens alleen maar opgebouwd uit strepen. Van deze nieuwe tekens krijgt u wederom twintig keer een morse-

codeteken te horen. Dit is het tweede gedeelte van de les.

Het derde gedeelte van de les combineert het eerste en het tweede gedeelte. Het vierde gedeelte is voor de eerste les hetzelfde als het derde gedeelte. Nu wordt vanzelf overgegaan naar les 2. Deze gaat overeenkomstig les 1, met uitzondering van het vierde gedeelte. Dit betreft er tevens de tekens van de eerste les bij. Ook nu wordt automatisch overgegaan naar de volgende les. In het vierde gedeelte worden steeds alle tekens van de voorgaande lessen herhaald. Mocht u merken dat u teveel fouten gaat maken dan moet u stoppen en op een wat later tijdstip weer beginnen met les 1. Het vierde gedeelte van de lessen is bij elke volgende les langer omdat er steeds meer tekens bij betrokken raken. Heeft u de zesde les geheel doorlopen en heeft u daarbij geen of slechts een enkele fout gemaakt, dan begint u opnieuw, echter met een hogere seinsnelheid. Dit herhaalt u net zo lang totdat u de laatste les feilloos doorkomt bij een seinsnelheid van 16 woorden per minuut. Op deze seinsnelheid kom ik nog terug bij de beschrijving van de laatste les.

De volgende codetekens zijn hier gebruikt:

A	. —
B	— . . .
C	— . — .
D	— . .
E	.
F	. . — .
G	— — .
H
I	. .
J	. — — —
K	— . —
L	. — . .
M	— —
N	— .
O	— — —
P	. — — .
Q	— — . —
R	. — .
S	. . .

T	—
U	. . —
V	. . . —
W	. — —
X	— . . —
Y	— . — —
Z	— — . .
1	. — — — —
2	. . — — —
3	. . . — —
4 —
5
6	—
7	— — . . .
8	— — — . .
9	— — — — .
0	— — — — —
,	— — — . . — —
/	— . . . —
.	. — . — . —
:	— — — . . .
—	— —
=	— . . . —
spatie	. — . . —
(— . — — . —
;	— . — . — .
'	. — — — — .

De volgende tekens zijn niet in het programma opgenomen omdat hiervoor geen toetsen beschikbaar zijn:


begin	— . — . —
eind	. — . — .
sluiten	. . . — . —
wacht	. — . . .

Voor de goede orde: een gedeelte van een les wordt door de computer herhaald als u meer dan 1/3 deel van de tekens fout hebt ingevoerd!

```

10 'morse leren
20 ON KEY GOSUB 1040
30 A=RND(-TIME): KEY OFF
40 SCREEN 0: WIDTH(40): COLOR 15,4,4
50 SOUND 0,135
60 SOUND 1,0
70 SOUND 7,62
80 SOUND 8,15
90 FOR L=1 TO 30: NEXT L
100 SOUND 8,0
110 FOR Y=1 TO 6: FOR X=1 TO 9

```



```

120 READ A(X,Y): NEXT X,Y
130 FOR Y=1 TO 6: FOR X=1 TO 9
140 READ B(X,Y): NEXT X,Y
150 PRINT: INPUT "Welke les";LS
160 IF LS>6 THEN CLS: GOTO 150
170 INPUT "Welke seinsnelheid";W
180 T=INT(510.54/W-7.138): T3=3*T+20
190 Q=20: P=5: M=0: QL=LS
200 QP=1: LP=1: P1=5
210 CLS
220 PRINT TAB(14)"Morse leren."
230 PRINT: PRINT "Les"LS
240 PRINT
250 FOR X=1 TO P1: TN=A(X,LS): T$=""
260 IF TN=0 THEN 360
270 FOR N=1 TO 7: TN=INT(TN/2)
280 IF TN=1 THEN 330
290 IF (TN AND 1)=1 THEN 310
300 T$(N)=CHR$(45): GOTO 320
310 T$(N)=CHR$(46)
320 NEXT N
330 FOR U=N-1 TO 1 STEP -1
340 T$=T$+T$(U): NEXT U
350 PRINT: PRINT CHR$(B(X,LS))" "T$
360 NEXT X
370 PU=0: FOR V=1 TO Q
380 FOR Z=1 TO 500: NEXT Z
390 X=INT(RND(1)*P+QP)
400 Y=INT(RND(1)*LP+QL)
410 TN=A(X,Y): IF TN=0 THEN 390
420 LOCATE 20,10
430 GOSUB 720
440 BI$=INKEY$: IF BI$="" THEN 440
450 IF BI$=CHR$(B(X,Y)) THEN 470
460 PRINT "Fout, "CHR$(B(X,Y)): GOTO 480
470 PRINT "Goed. "": PU=PU+1
480 NEXT V
490 FOR Z=0 TO 1000: NEXT Z
500 IF PU<2*Q/3 THEN 690
510 M=M+1: QP=6 : P1=9: P=4
520 IF LS=6 THEN 540
530 IF M=1 THEN 210
540 QP=1: P=9: IF LS=6 THEN 680
550 IF M=2 THEN 210
560 Q=Q*LS: QL=1: LP=LS
570 IF M=3 THEN 210
580 LS=LS+1: IF LS>6 THEN 600
590 IF M>3 THEN 190

```

```

600 CLS: PRINT"U hebt alle ";
610 PRINT"lessen gehad. Probeer "
620 PRINT"eens opnieuw met ";
630 PRINT"een hogere snelheid."
640 PRINT: PRINT"Gebruik hiervoor ";
650 PRINT"toets f1."
660 KEY(1) ON
670 GOTO 670
680 M=M+2: GOTO 560
690 CLS: PRINT"Slecht resultaat, ";
700 PRINT"opnieuw dezelfde les."
710 FOR N=1 TO 1000: NEXT: GOTO 210
720 FOR N=1 TO 7: TN=INT(TN/2)
730 IF TN=1 THEN 780
740 IF (TN AND 1)=1 THEN 760
750 T(N)=T3: GOTO 770
760 T(N)=T
770 NEXT N
780 ON N-1 GOTO 890,870,850,830,810,790
790 SOUND 8,15: FOR L=0 TO T(6): NEXT L
800 SOUND 8,0: FOR L=0 TO T: NEXT L
810 SOUND 8,15: FOR L=0 TO T(5): NEXT L
820 SOUND 8,0: FOR L=0 TO T: NEXT L
830 SOUND 8,15: FOR L=0 TO T(4): NEXT L
840 SOUND 8,0: FOR L=0 TO T: NEXT L
850 SOUND 8,15: FOR L=0 TO T(3): NEXT L
860 SOUND 8,0: FOR L=0 TO T: NEXT L
870 SOUND 8,15: FOR L=0 TO T(2): NEXT L
880 SOUND 8,0: FOR L=0 TO T: NEXT L
890 SOUND 8,15: FOR L=0 TO T(1): NEXT L
900 SOUND 8,0: FOR L=0 TO T: NEXT L
910 RETURN
920 DATA 6,14,30,62,126,4,8,16,64
930 DATA 12,28,60,124,0,10,22,46,94
940 DATA 58,54,0,26,50,20,44,92,188
950 DATA 24,48,96,112,120,18,38,78,70
960 DATA 36,40,42,66,152,90,212,170,142
970 DATA 194,164,108,230,0,0,0,0,0
980 DATA 69,73,83,72,53,84,77,79,48
990 DATA 65,85,86,52,0,78,68,66,54
1000 DATA 70,76,0,82,80,75,88,61,45
1010 DATA 87,74,49,50,51,71,90,55,56
1020 DATA 81,89,67,57,44,47,46,59,58
1030 DATA 39,40,32,63,0,0,0,0,0
1040 KEY(1) OFF: RUN

```

10.2. Morse oefenen

Met dit deel kunt u vaardigheid verkrijgen in het ontcijferen van morsecodetekens. Als u met het vorige programma de tekens feilloos hebt leren herkennen, ook bij hogere snelheden, dan kunt u zich aan dit gedeelte van de cursus wagen. Na het starten van het programma kunt u de gewenste seinsnelheid invoeren. Begin uiteraard weer met een lage snelheid! Direct daarna hoort u uit de luidspreker van uw tv de morsecodetekens van letters, cijfers en leestekens in een willekeurige volgorde, steeds in groepjes van vijf. Deze groepjes worden woorden genoemd, ook al leveren deze woorden geen verstaanbare taal op. De bedoeling is dat u direct na het horen van een morsecode-

teken de toets met het bijbehorende schriftteken indrukt (denk erom: CAPS LOCK toets ingedrukt!). In elk geval moet u alle tekens van een woord hebben ingevoerd voordat het volgende woord te horen is. Maakt u hierbij een fout dan wordt deze geteld. Maakt u in een woord meerdere fouten dan worden deze als één fout aangemerkt. Steeds wordt het door u ingetypte woord op het scherm zichtbaar gemaakt direct vóór het begin van het volgende woord. De woorden zijn te herkennen door de langere pauze tussen de groepjes. Het aantal woorden dat u in een serie krijgt, is afhankelijk van de seinsnelheid. Denkt u het sneller te kunnen dan verhoogt u de seinsnelheid. Deze seinsnelheid wordt uitgedrukt in het aantal woorden per seconde. Dit is in overeenstemming met wat officieel gebruikelijk is. Wat niet gebruikelijk is, is dat de woorden die worden gebruikt voor het bepalen van de seinsnelheid ook cijfers en leestekens bevatten. Voor het bepalen van de seinsnelheid is uitgegaan van het woord PARIS. Dit bevat 50 eenheden en elke eenheid heeft de tijdsduur van een punt. Omdat de woorden die u krijgt voorgescheteld ook leestekens en cijfers bevatten duurt een serie langer dan in overeenstemming is met de ingevoerde seinsnelheid. Haalt u met de computer een seinsnelheid van zestien woorden per seconde dan bent u in staat elk examen dat zestien woorden per minuut verlangt, zonder kleerscheuren door te komen (uiteraard alleen het opnemen).

```

10 'morse leren
20 ON KEY GOSUB 780
30 A=ROUND(-TIME): KEY OFF
40 SCREEN 0: WIDTH(40): COLOR 15,4,4
50 SOUND 0,135
60 SOUND 1,0
70 SOUND 7,62
80 SOUND 8,15
90 FOR L=1 TO 30: NEXT L
100 SOUND 8,0
110 FOR Y=1 TO 6: FOR X=1 TO 9
120 READ A(X,Y): NEXT X,Y
130 FOR Y=1 TO 6: FOR X=1 TO 9
140 READ B(X,Y): NEXT X,Y
150 INPUT"Welke seinsnelheid";W
160 PQ=INT(5*W): FOR N=0 TO 2000: NEXT
170 T=INT(510.54/W-7.138): T3=3*T+20
180 T4=T3-50: T7=7*T
190 CLS
200 PRINT TAB(14)"Morse leren."
210 FOR Q=1 TO PQ
220 TG$="": TN$="": FOR P=1 TO 5
230 X=INT(RND(1)*9+1)
240 Y=INT(RND(1)*6+1)
250 TN=A(X,Y): IF TN=0 THEN 230
260 TN$=TN$+CHR$(B(X,Y))

```

```

270 FOR L=0 TO T4: NEXT L
280 GOSUB 460: NEXT P
290 FOR L=1 TO T7
300 G#=INKEY#: IF G#<>" " THEN 440
310 NEXT L
320 IF TG#<>TN# THEN 450
330 PRINT TG#: NEXT @
340 QP=INT(PU*100/PQ)
350 PRINT: PRINT
360 PRINT"U heeft"PU"fouten in"PQ;
370 PRINT"woorden.": PRINT
380 PRINT"Dat is"QP"%."
390 PRINT: PRINT: PRINT
400 PRINT"Gebruik voor een volgende "
410 PRINT"serie de "; PRINT"f1 toets."
420 KEY(1) ON
430 GOTO 430
440 TG#=TG#+G#: GOTO 310
450 PU=PU+1: GOTO 330
460 FOR N=1 TO 7: TN=INT(TN/2)
470 IF TN=1 THEN 520
480 IF (TN AND 1)=1 THEN 500
490 T(N)=T3: GOTO 510
500 T(N)=T
510 NEXT N
520 ON N-1 GOTO 630,610,590,570,550,530

```

```

530 SOUND 8,15: FOR L=0 TO T(6): NEXT L
540 SOUND 8,0: FOR L=0 TO T: NEXT L
550 SOUND 8,15: FOR L=0 TO T(5): NEXT L
560 SOUND 8,0: FOR L=0 TO T: NEXT L
570 SOUND 8,15: FOR L=0 TO T(4): NEXT L
580 SOUND 8,0: FOR L=0 TO T: NEXT L
590 SOUND 8,15: FOR L=0 TO T(3): NEXT L
600 SOUND 8,0: FOR L=0 TO T: NEXT L
610 SOUND 8,15: FOR L=0 TO T(2): NEXT L
620 SOUND 8,0: FOR L=0 TO T: NEXT L
630 SOUND 8,15: FOR L=0 TO T(1): NEXT L
640 SOUND 8,0: FOR L=0 TO T: NEXT L
650 RETURN
660 DATA 6,14,30,62,126,4,8,16,64
670 DATA 12,28,60,124,0,10,22,46,94
680 DATA 58,54,0,26,50,20,44,92,188
690 DATA 24,48,96,112,120,18,38,78,70
700 DATA 36,40,42,66,152,90,212,170,142
710 DATA 194,164,108,230,0,0,0,0,0
720 DATA 69,73,83,72,53,84,77,79,48
730 DATA 65,85,86,52,0,78,68,66,54
740 DATA 70,76,0,82,80,75,88,61,45
750 DATA 87,74,49,50,51,71,90,55,56
760 DATA 81,89,67,57,44,47,46,59,58
770 DATA 39,40,32,63,0,0,0,0,0
780 KEY(1) OFF: RUN

```